SAS/QC® 13.2 User's Guide

The MACONTROL Procedure

This document is an individual chapter from *SAS/QC® 13.2 User's Guide*.

The correct bibliographic citation for the complete manual is as follows: SAS Institute Inc. 2014. *SAS/QC® 13.2 User's Guide*. Cary, NC: SAS Institute Inc.

SAS Institute Inc., SAS Campus Drive, Cary, North Carolina 27513.

August 2014



Chapter 9
The MACONTROL Procedure

Contents

Introduction: MACONTROL Procedure	**778**
Learning about the MACONTROL Procedure	780
PROC MACONTROL Statement	**780**
Overview: PROC MACONTROL Statement	780
Syntax: PROC MACONTROL Statement	781
BY Statement	783
Input and Output Data Sets: MACONTROL Procedure	784
EWMACHART Statement: MACONTROL Procedure	**785**
Overview: EWMACHART Statement	785
Getting Started: EWMACHART Statement	786
Creating EWMA Charts from Raw Data	786
Creating EWMA Charts from Subgroup Summary Data	789
Saving Summary Statistics	792
Saving Control Limit Parameters	793
Reading Preestablished Control Limit Parameters	795
Syntax: EWMACHART Statement	797
Summary of Options	798
Dictionary of Special Options	806
Details: EWMACHART Statement	809
Constructing EWMA Charts	809
Output Data Sets	815
ODS Tables	818
ODS Graphics	818
Input Data Sets	818
Methods for Estimating the Standard Deviation	822
Axis Labels	823
Missing Values	824
Examples: EWMACHART Statement	824
Example 9.1: Specifying Standard Values for the Process Mean and Process Standard Deviation	824
Example 9.2: Displaying Limits Based on Asymptotic Values	826
Example 9.3: Working with Unequal Subgroup Sample Sizes	828
Example 9.4: Displaying Individual Measurements on an EWMA Chart	833
Example 9.5: Computing Average Run Lengths	835
MACHART Statement: MACONTROL Procedure	**837**
Overview: MACHART Statement	837

Getting Started: MACHART Statement 838
Creating Moving Average Charts from Raw Data 839
Creating Moving Average Charts from Subgroup Summary Data 842
Saving Summary Statistics 844
Saving Control Limit Parameters 845
Reading Preestablished Control Limit Parameters 848
Syntax: MACHART Statement 850
Summary of Options 851
Dictionary of Special Options 859
Details: MACHART Statement 862
Constructing Uniformly Weighted Moving Average Charts 862
Output Data Sets 867
ODS Tables 870
ODS Graphics 870
Input Data Sets 871
Methods for Estimating the Standard Deviation 874
Axis Labels 876
Missing Values 876
Examples: MACHART Statement 877
Example 9.6: Specifying Standard Values for the Process Mean and Process Standard Deviation 877
Example 9.7: Annotating Average Run Lengths on the Chart 879
INSET Statement: MACONTROL Procedure 880
Overview: INSET Statement 880
Getting Started: INSET Statement 881
Syntax: INSET Statement 882
References 884

Introduction: MACONTROL Procedure

The MACONTROL procedure creates moving average control charts, which are tools for deciding whether a process is in a state of statistical control and for detecting shifts in a process average. The procedure creates the following two types of charts:

- *uniformly weighted moving average charts* (commonly referred to as *moving average charts*). Each point on a moving average chart represents the average of the w most recent subgroup means, including the present subgroup mean. The next moving average is computed by dropping the oldest of the previous w subgroup means and including the newest subgroup mean.

  The constant w, often referred to as the *span* of the moving average, is a parameter of the moving average chart. There is an inverse relationship between w and the magnitude of the shift to be detected; larger values of w are used to guard against smaller shifts.

- *exponentially weighted moving average (EWMA) charts*, also referred to as *geometric moving average (GMA) charts*. Each point on an EWMA chart represents the weighted average of all the previous subgroup means, including the mean of the present subgroup sample. The weights decrease exponentially going backward in time.

  The weight r ($0 < r \leq 1$) assigned to the present subgroup sample mean is a parameter of the EWMA chart. Small values of r are used to guard against small shifts. If $r = 1$, the EWMA chart reduces to a Shewhart $\overline{X}$ chart.

In the MACONTROL procedure, the EWMACHART statement produces EWMA charts, and the MACHART statement produces uniformly weighted moving average charts.

In contrast to the Shewhart chart where each point is based on information from a single subgroup sample, each point on a moving average chart combines information from the current sample and past samples. Consequently, the moving average chart is more sensitive to small shifts in the process average. On the other hand, it is more difficult to interpret patterns of points on a moving average chart, since consecutive moving averages can be highly correlated, as pointed out by Nelson (1983).

You can use the MACONTROL procedure to

- read raw data (actual measurements) or summarized data (subgroup means and standard deviations) to create charts
- specify control limits as probability limits or in terms of a multiple of the standard error of the moving average
- adjust the control limits to compensate for unequal subgroup sample sizes
- accept numeric- or character-valued subgroup variables
- display subgroups with date and time formats
- estimate the process standard deviation σ using a variety of methods or specify a standard (known) value for σ
- analyze multiple process variables in the same chart statement
- provide multiple chart statements. If used with a BY statement, the procedure generates charts separately for BY groups of observations.
- tabulate the information displayed in the control chart
- save moving averages, control limits, and control limit parameters in output data sets
- superimpose plotted points with stars (polygons) whose vertices indicate the values of multivariate data related to the process
- display a trend chart below the moving average chart that plots a systematic or fitted trend in the data
- produce charts as traditional graphics, ODS Graphics output, or legacy line printer charts. Line printer charts can use special formatting characters that improve the appearance of the chart. Traditional graphics can be annotated, saved, and replayed.

Learning about the MACONTROL Procedure

If you are using the MACONTROL procedure for the first time, begin by reading "PROC MACONTROL Statement" on page 780 to learn about input data sets. Then read the section "Getting Started: EWMACHART Statement" on page 786 in "EWMACHART Statement: MACONTROL Procedure" on page 785 or the section "Getting Started: MACHART Statement" on page 838 in "MACHART Statement: MACONTROL Procedure" on page 837. These chapters also provide syntax information, computational details, and advanced examples.

PROC MACONTROL Statement

Overview: PROC MACONTROL Statement

The PROC MACONTROL statement starts the MACONTROL procedure and it identifies input data sets.

After the PROC MACONTROL statement, you provide either an EWMACHART or an MACHART statement that specifies the type of moving average chart you want to create and the variables in the input data set that you want to analyze. For example, the following statements request a uniformly weighted moving average chart:

```
proc macontrol data=values;
   machart weight*lot / mu0    = 8.10
                        sigma0 = 0.05
                        span   = 5;
run;
```

In this example, the DATA= option specifies an input data set named values that contains the *process* measurement variable weight and the *subgroup-variable* lot.

You can use options in the PROC MACONTROL statement to do the following:

- specify input data sets containing variables to be analyzed, parameters for calculating moving averages, or annotation information
- specify a graphics catalog for saving traditional graphics
- specify that charts be produced as traditional graphics or line printer charts
- define characters used for features on line printer charts

In addition to the chart statement, you can provide BY statements, ID statements, TITLE statements, and FOOTNOTE statements. If you are producing traditional graphics, you can also provide graphics enhancement statements, such as SYMBOL*n* statements, which are described in *SAS/GRAPH: Reference*.

See Chapter 3, "SAS/QC Graphics," for a detailed discussion of the alternatives available for producing charts with SAS/QC procedures.

NOTE: If you are using the MACONTROL procedure for the first time, you should also read the sections "Getting Started: EWMACHART Statement" on page 786 and "Getting Started: MACHART Statement" on page 838.

Syntax: PROC MACONTROL Statement

The syntax for the PROC MACONTROL statement is as follows:

PROC MACONTROL < *options* > **;**

The PROC MACONTROL statement starts the MACONTROL procedure, and it optionally identifies various data sets and requests line printer charts. You can specify the following options in the PROC MACONTROL statement.

ANNOTATE=*SAS-data-set*

ANNO=*SAS-data-set*

specifies an input data set that contains appropriate annotate variables, as described in *SAS/GRAPH: Reference*. The ANNOTATE= option enables you to add features to the moving average chart (for example, labels that explain out-of-control points). The ANNOTATE= data set is used only when the chart is created as traditional graphics; it is ignored if ODS Graphics is enabled or if you specify the LINEPRINTER option.

The data set specified with the ANNOTATE= option in the PROC MACONTROL statement is a "global" annotate data set in the sense that the information in this data set is displayed on every chart produced in the current run of the MACONTROL procedure.

ANNOTATE2=*SAS-data-set*

ANNO2=*SAS-data-set*

specifies an input data set that contains appropriate annotate variables that add features to the trend chart (secondary chart) produced with the TRENDVAR= option in the EWMACHART or MACHART statement. This option is ignored if ODS Graphics is enabled or if you specify the LINEPRINTER option.

DATA=*SAS-data-set*

names an input data set that contains raw data (measurements) as observations. If the values of the *subgroup-variable* are numeric, you need to sort the data set so that these values are in increasing order (within BY groups). The DATA= data set can contain more than one observation for each value of the *subgroup-variable*.

You cannot specify a DATA= data set with a HISTORY= or TABLE= data set. If you do not specify an input data set, PROC MACONTROL uses the most recently created data set as a DATA= data set. For more information, see "DATA= Data Set" in the appropriate chart statement chapter.

FORMCHAR(*index***)='***string***'**

defines characters used for features on line printer charts, where *index* is a list of numbers ranging from 1 to 17 and *string* is a character or hexadecimal string. This option applies only if you also specify the LINEPRINTER option.

The *index* identifies which features are controlled with the *string* characters, as described in the following table. If you specify the FORMCHAR= option and omit the *index*, the *string* controls all 17 features.

Value of *index*	Description of Character	Chart Feature
1	vertical bar	frame
2	horizontal bar	frame, central line
3	box character (upper left)	frame
4	box character (upper middle)	serifs, tick (horizontal axis)
5	box character (upper right)	frame
6	box character (middle left)	not used
7	box character (middle middle)	serifs
8	box character (middle right)	tick (vertical axis)
9	box character (lower left)	frame
10	box character (lower middle)	serifs
11	box character (lower right)	frame
12	vertical bar	control limits
13	horizontal bar	control limits
14	box character (upper right)	control limits
15	box character (lower left)	control limits
16	box character (lower right)	control limits
17	box character (upper left)	control limits

Not all printers can produce the characters in the preceding list. By default, the form character list specified by the SAS system FORMCHAR= option is used; otherwise, the default is FORMCHAR='|—-|+|—|====='. If you print to a PC screen or if your device supports the ASCII symbol set (1 or 2), the following is recommended:

```
formchar='B3,C4,DA,C2,BF,C3,C5,B4,C0,C1,D9,BA,CD,BB,C8,BC,D9'X
```

Note that you can use the FORMCHAR= option to temporarily override the values of the SAS system FORMCHAR= option. The values of the SAS system FORMCHAR= option are not altered by the FORMCHAR= option in the PROC MACONTROL statement.

GOUT=*graphics-catalog*

specifies the graphics catalog for traditional graphics output from PROC MACONTROL. This is useful if you want to save the output. The GOUT= option is used only when the chart is created using traditional graphics; it is ignored if ODS Graphics is enabled or if you specify the LINEPRINTER option.

HISTORY=*SAS-data-set*

HIST=*SAS-data-set*

names an input data set that contains subgroup summary statistics (means, standard deviations, and sample sizes). Typically, this data set is created as an OUTHISTORY= data set in a previous run of PROC MACONTROL or PROC SHEWHART, but it can also be created with a SAS summarization procedure such as PROC MEANS.

If the values of the *subgroup-variable* are numeric, you need to sort the data set so that these values are in increasing order (within BY groups). A HISTORY= data set can contain only one observation for each value for the *subgroup-variable*.

You cannot use a HISTORY= data set with a DATA= or TABLE= data set. If you do not specify an input data set, PROC MACONTROL uses the most recently created data set as a DATA= data set. For more information on HISTORY= data sets, see "HISTORY= Data Set" in the appropriate chart statement chapter.

LIMITS=*SAS-data-set*

names an input data set that contains the control limit parameters for the moving average chart. Each observation in a LIMITS= data set contains the parameters for a *process*.

For details about the variables needed in a LIMITS= data set, see "LIMITS= Data Set" in the appropriate chart statement chapter.

If you do not provide a LIMITS= data set, you must specify the parameters with options in the chart statement.

LINEPRINTER

requests that legacy line printer charts be produced.

TABLE=*SAS-data-set*

names an input data set that contains subgroup summary statistics and control limits. Each observation in a TABLE= data set provides information for a particular subgroup and *process*. Typically, this data set is created as an OUTTABLE= data set in a previous run of PROC MACONTROL.

You cannot use a TABLE= data set with a DATA= or HISTORY= data set. If you do not specify an input data set, PROC MACONTROL uses the most recently created data set as a DATA= data set. For more information, see the "TABLE= Data Set" section in the appropriate chart statement chapter.

BY Statement

BY *variables* **;**

You can specify a BY statement with PROC MACONTROL to obtain separate analyses of observations in groups that are defined by the BY variables. When a BY statement appears, the procedure expects the input data set to be sorted in order of the BY variables. If you specify more than one BY statement, only the last one specified is used.

If your input data set is not sorted in ascending order, use one of the following alternatives:

- Sort the data by using the SORT procedure with a similar BY statement.
- Specify the NOTSORTED or DESCENDING option in the BY statement for the MACONTROL procedure. The NOTSORTED option does not mean that the data are unsorted but rather that the data are arranged in groups (according to values of the BY variables) and that these groups are not necessarily in alphabetical or increasing numeric order.
- Create an index on the BY variables by using the DATASETS procedure (in Base SAS software).

For more information about BY-group processing, see the discussion in *SAS Language Reference: Concepts*. For more information about the DATASETS procedure, see the discussion in the *Base SAS Procedures Guide*.

Input and Output Data Sets: MACONTROL Procedure

Figure 9.1 summarizes the data sets used with the MACONTROL procedure.

Figure 9.1 Input and Output Data Sets in the MACONTROL Procedure



EWMACHART Statement: MACONTROL Procedure

Overview: EWMACHART Statement

The EWMACHART statement creates an exponentially weighted moving average (EWMA) control chart, which is used to determine whether a process is in a state of statistical control and to detect shifts in the process average.

You can use options in the EWMACHART statement to

- specify the weight assigned to the most recent subgroup mean in the computation of the EWMAs
- compute control limits from the data based on a multiple of the standard error of the plotted EWMAs or as probability limits
- tabulate the EWMAs, subgroup sample sizes, subgroup means, subgroup standard deviations, control limits, and other information
- save control limit parameters in an output data set
- save the EWMAs, subgroup sample sizes, subgroup means, and subgroup standard deviations in an output data set
- read control limit parameters from an input data set
- specify one of several methods for estimating the process standard deviation
- specify a known (standard) process mean and standard deviation for computing control limits
- display a secondary chart that plots a time trend removed from the data
- add block legends and symbol markers to reveal stratification in process data
- superimpose stars at points to represent related multivariate factors
- clip extreme points to make the chart more readable
- display vertical and horizontal reference lines
- control axis values and labels
- control layout and appearance of the chart

You have three alternatives for producing EWMA charts with the EWMACHART statement:

- ODS Graphics output is produced if ODS Graphics is enabled, for example by specifying the ODS GRAPHICS ON statement prior to the PROC statement.
- Otherwise, traditional graphics are produced by default if SAS/GRAPH® is licensed.
- Legacy line printer charts are produced when you specify the LINEPRINTER option in the PROC statement.

See Chapter 3, “SAS/QC Graphics,” for more information about producing these different kinds of graphs.

Getting Started: EWMACHART Statement

This section introduces the EWMACHART statement with simple examples that illustrate the most commonly used options. Complete syntax for the EWMACHART statement is presented in the section "Syntax: EWMACHART Statement" on page 797, and advanced examples are given in the section "Examples: EWMACHART Statement" on page 824.

Creating EWMA Charts from Raw Data

NOTE: See *Exponentially Weighted Moving Average Chart* in the SAS/QC Sample Library.

In the manufacture of a metal clip, the gap between the ends of the clip is a critical dimension. To monitor the process for a change in the average gap, subgroup samples of five clips are selected daily. The data are analyzed with an EWMA chart. The gaps recorded during the first twenty days are saved in a SAS data set named Clips1.

```
data Clips1;
   input Day @ ;
   do i=1 to 5;
      input Gap @ ;
      output;
   end;
   drop i;
   datalines;
 1  14.76  14.82  14.88  14.83  15.23
 2  14.95  14.91  15.09  14.99  15.13
 3  14.50  15.05  15.09  14.72  14.97
 4  14.91  14.87  15.46  15.01  14.99
 5  14.73  15.36  14.87  14.91  15.25
 6  15.09  15.19  15.07  15.30  14.98
 7  15.34  15.39  14.82  15.32  15.23
 8  14.80  14.94  15.15  14.69  14.93
 9  14.67  15.08  14.88  15.14  14.78
10  15.27  14.61  15.00  14.84  14.94
11  15.34  14.84  15.32  14.81  15.17
12  14.84  15.00  15.13  14.68  14.91
13  15.40  15.03  15.05  15.03  15.18
14  14.50  14.77  15.22  14.70  14.80
15  14.81  15.01  14.65  15.13  15.12
16  14.82  15.01  14.82  14.83  15.00
17  14.89  14.90  14.60  14.40  14.88
18  14.90  15.29  15.14  15.20  14.70
19  14.77  14.60  14.45  14.78  14.91
20  14.80  14.58  14.69  15.02  14.85
;
```

A partial listing of Clips1 is shown in Figure 9.2.

Figure 9.2 Partial Listing of the Data Set Clips1

The Data Set Clips1

Day	Gap
1	14.76
1	14.82
1	14.88
1	14.83
1	15.23
2	14.95
2	14.91
2	15.09
2	14.99
2	15.13
3	14.50
3	15.05
3	15.09
3	14.72
3	14.97

The data set Clips1 is said to be in "strung-out" form, since each observation contains the day and gap measurement of a single clip. The first five observations contain the gap measurements for the first day, the second five observations contain the gap measurements for the second day, and so on. Because the variable Day classifies the observations into rational subgroups, it is referred to as the *subgroup-variable*. The variable Gap contains the gap measurements and is referred to as the *process variable* (or *process* for short).

The within-subgroup variability of the gap measurements is known to be stable. You can use an EWMA chart to determine whether the mean level is in control. The following statements create the EWMA chart shown in Figure 9.3:

```
ods graphics off;
symbol h = 0.8;
title 'EWMA Chart for Gap Measurements';
proc macontrol data=Clips1;
   ewmachart Gap*Day / weight=0.3;
run;
```

This example illustrates the basic form of the EWMACHART statement. After the keyword EWMACHART, you specify the *process* to analyze (in this case, Gap) followed by an asterisk and the *subgroup-variable* (Day). The WEIGHT= option specifies the weight parameter used to compute the EWMAs. Options such as WEIGHT= are specified after the slash (/) in the EWMACHART statement. A complete list of options is presented in the section "Syntax: EWMACHART Statement" on page 797. You must provide the weight parameter to create an EWMA chart. As an alternative to specifying the WEIGHT= option, you can read the weight parameter from an input data set; see "Reading Preestablished Control Limit Parameters" on page 795.

The input data set is specified with the DATA= option in the PROC MACONTROL statement.

Figure 9.3 Exponentially Weighted Moving Average Chart



Each point on the chart represents the EWMA for a particular day. The EWMA E_1 plotted at Day=1 is the weighted average of the overall mean and the subgroup mean for Day=1. The EWMA E_2 plotted at Day=2 is the weighted average of the EWMA E_1 and the subgroup mean for Day=2.

$$
\begin{aligned}
E_1 &= 0.3(14.904) + 0.7(14.952) = 14.9376\text{mm} \\
E_2 &= 0.3(15.014) + 0.7(14.9376) = 14.9605\text{mm}
\end{aligned}
$$

For succeeding days, the EWMA is the weighted average of the previous EWMA and the present subgroup mean. In the example, a weight parameter of 0.3 is used (since WEIGHT=0.3 is specified in the EWMACHART statement).

Note that the EWMA for the 7th day lies above the upper control limit, signaling an out-of-control process.

By default, the control limits shown are 3σ limits estimated from the data; the formulas for the limits are given in Table 9.3.

For computational details, see "Constructing EWMA Charts" on page 809. For more details on reading from a DATA= data set, see "DATA= Data Set" on page 818.

Creating EWMA Charts from Subgroup Summary Data

NOTE: See *Exponentially Weighted Moving Average Chart* in the SAS/QC Sample Library.

The previous example illustrates how you can create EWMA charts using raw data (process measurements). However, in many applications the data are provided as subgroup summary statistics. This example illustrates how you can use the EWMACHART statement with data of this type.

The following data set (Clipsum) provides the data from the preceding example in summarized form:

```
data Clipsum;
   input Day GapX GapS;
   GapN=5;
   datalines;
 1  14.904  0.18716
 2  15.014  0.09317
 3  14.866  0.25006
 4  15.048  0.23732
 5  15.024  0.26792
 6  15.126  0.12260
 7  15.220  0.23098
 8  14.902  0.17254
 9  14.910  0.19824
10  14.932  0.24035
11  15.096  0.25618
12  14.912  0.16903
13  15.138  0.15928
14  14.798  0.26329
15  14.944  0.20876
16  14.896  0.09965
17  14.734  0.22512
18  15.046  0.24141
19  14.702  0.17880
20  14.788  0.16634
;
```

A partial listing of Clipsum is shown in Figure 9.4. There is exactly one observation for each subgroup (note that the subgroups are still indexed by Day). The variable GapX contains the subgroup means, the variable GapS contains the subgroup standard deviations, and the variable GapN contains the subgroup sample sizes (these are all five).

Figure 9.4 The Summary Data Set Clipsum

The Data Set Clipsum

Day	GapX	GapS	GapN
1	14.904	0.18716	5
2	15.014	0.09317	5
3	14.866	0.25006	5
4	15.048	0.23732	5
5	15.024	0.26792	5

You can read this data set by specifying it as a HISTORY= data set in the PROC MACONTROL statement, as follows:

```
options nogstyle;
goptions ftext=swiss;
symbol color=salmon h=0.8;
title 'EWMA Chart for Gap Measurements';
proc macontrol history=Clipsum;
   ewmachart Gap*Day / weight=0.3
                       cframe = vibg
                       cinfill = ligr
                       coutfill = yellow
                       cconnect = salmon;
run;
options gstyle;
```

The NOGSTYLE system option causes ODS styles not to affect traditional graphics. Instead, the GOPTIONS and SYMBOL statements and EWMACHART statement options control the appearance of the graph. The GSTYLE system option restores the use of ODS styles for traditional graphics produced subsequently. The resulting EWMA chart is shown in Figure 9.5.

Note that Gap is *not* the name of a SAS variable in the data set but is, instead, the common prefix for the names of the three SAS variables GapX, GapS, and GapN. The suffix characters *X*, *S*, and *N* indicate *mean*, *standard deviation*, and *sample size*, respectively. Thus, you can specify three subgroup summary variables in a HISTORY= data set with a single name (Gap), which is referred to as the *process*. The variables GapX, GapS, and GapN are all required. The name Day specified after the asterisk is the name of the *subgroup-variable*.

Figure 9.5 EWMA Chart from Summary Data



In general, a HISTORY= input data set used with the EWMACHART statement must contain the following variables:

- subgroup variable
- subgroup mean variable
- subgroup standard deviation variable
- subgroup sample size variable

Furthermore, the names of subgroup mean, standard deviation, and sample size variables must begin with the *process* name specified in the EWMACHART statement and end with the special suffix characters *X*, *S*, and *N*, respectively. If the names do not follow this convention, you can use the RENAME option in the PROC MACONTROL statement to rename the variables for the duration of the MACONTROL procedure step (see "Creating Charts for Means and Ranges from Summary Data" on page 1841 for an example of the RENAME option).

In summary, the interpretation of *process* depends on the input data set.

- If raw data are read using the DATA= option (as in the previous example), *process* is the name of the SAS variable containing the process measurements.
- If summary data are read using the HISTORY= option (as in this example), *process* is the common prefix for the names of the variables containing the summary statistics.

For more information, see “HISTORY= Data Set” on page 820.

Saving Summary Statistics

NOTE: See *Exponentially Weighted Moving Average Chart* in the SAS/QC Sample Library.

In this example, the EWMACHART statement is used to create a summary data set that can be read later by the MACONTROL procedure (as in the preceding example). The following statements read measurements from the data set Clips1 and create a summary data set named Cliphist:

```
title 'Summary Data Set for Gap Measurements';
proc macontrol data=Clips1;
   ewmachart Gap*Day / weight     = 0.3
                       outhistory = Cliphist
                       nochart;
run;
```

The OUTHISTORY= option names the output data set, and the NOCHART option suppresses the display of the chart, which would be identical to the chart in Figure 9.3.

Figure 9.6 contains a partial listing of Cliphist.

Figure 9.6 The Summary Data Set Cliphist

Summary Data Set for Gap Measurements

Day	GapX	GapS	GapE	GapN
1	14.904	0.18716	14.9362	5
2	15.014	0.09317	14.9595	5
3	14.866	0.25006	14.9315	5
4	15.048	0.23732	14.9664	5
5	15.024	0.26792	14.9837	5

There are five variables in the data set Cliphist.

- Day contains the subgroup index.
- GapX contains the subgroup means.
- GapS contains the subgroup standard deviations.
- GapE contains the subgroup exponentially weighted moving averages.
- GapN contains the subgroup sample sizes.

Note that the summary statistic variables are named by adding the suffix characters *X*, *S*, *E*, and *N* to the *process* Gap specified in the EWMACHART statement. In other words, the variable naming convention for OUTHISTORY= data sets is the same as that for HISTORY= data sets.

For more information, see "OUTHISTORY= Data Set" on page 816.

Saving Control Limit Parameters

NOTE: See *Exponentially Weighted Moving Average Chart* in the SAS/QC Sample Library.

You can save the control limit parameters for an EWMA chart in a SAS data set; this enables you to use these parameters with future data (see "Reading Preestablished Control Limit Parameters" on page 795) or modify the parameters with a DATA step program.

The following statements read measurements from the data set Clips1 (see "Creating EWMA Charts from Raw Data" on page 786) and save the control limit parameters in a data set named Cliplim:

```
title 'Control Limit Parameters';
proc macontrol data=Clips1;
   ewmachart Gap*Day / weight    = 0.3
                       outlimits = Cliplim
                       nochart;
run;
```

The OUTLIMITS= option names the data set containing the control limit parameters, and the NOCHART option suppresses the display of the chart. The data set Cliplim is listed in Figure 9.7.

Figure 9.7 The Data Set Cliplim Containing Control Limit Information

Control Limit Parameters

VAR	_SUBGRP_	_TYPE_	_LIMITN_	_ALPHA_	_SIGMAS_	_MEAN_	_STDDEV_	_WEIGHT_
Gap	Day	ESTIMATE	5	.002699796	3	14.95	0.21108	0.3

Note that the data set Cliplim does not contain the actual control limits but rather the parameters required to compute the limits.

The data set contains one observation with the parameters for *process* Gap. The variable _WEIGHT_ contains the weight parameter used to compute the EWMAs. The value of _MEAN_ is an estimate of the process mean, and the value of _STDDEV_ is an estimate of the process standard deviation σ. The value of _LIMITN_ is the nominal sample size associated with the control limits, and the value of _SIGMAS_ is the multiple of σ associated with the control limits. The variables _VAR_ and _SUBGRP_ are bookkeeping variables that save the *process* and *subgroup-variable*. The variable _TYPE_ is a bookkeeping variable that indicates that the values of _MEAN_ and _STDDEV_ are estimates rather than standard values. For more information, see "OUTLIMITS= Data Set" on page 815.

You can create an output data set containing the control limits and summary statistics with the OUTTABLE= option, as illustrated by the following statements:

```
title 'Summary Statistics and Control Limits';
proc macontrol data=Clips1;
   ewmachart Gap*Day / weight   = 0.3
                       outtable = Cliptab
                       nochart;
run;
```

The data set Cliptab is listed in Figure 9.8.

Figure 9.8 The OUTTABLE= Data Set Cliptab

Summary Statistics and Control Limits

VAR	Day	_SIGMAS_	_LIMITN_	_WEIGHT_	_SUBN_	_SUBX_	_SUBS_	_LCLE_	_EWMA_
Gap	1	3	5	0.3	5	14.904	0.18716	14.8650	14.9362
Gap	2	3	5	0.3	5	15.014	0.09317	14.8463	14.9595
Gap	3	3	5	0.3	5	14.866	0.25006	14.8383	14.9315
Gap	4	3	5	0.3	5	15.048	0.23732	14.8345	14.9664
Gap	5	3	5	0.3	5	15.024	0.26792	14.8327	14.9837
Gap	6	3	5	0.3	5	15.126	0.12260	14.8319	15.0264
Gap	7	3	5	0.3	5	15.220	0.23098	14.8314	15.0845
Gap	8	3	5	0.3	5	14.902	0.17254	14.8312	15.0297
Gap	9	3	5	0.3	5	14.910	0.19824	14.8311	14.9938
Gap	10	3	5	0.3	5	14.932	0.24035	14.8311	14.9753
Gap	11	3	5	0.3	5	15.096	0.25618	14.8311	15.0115
Gap	12	3	5	0.3	5	14.912	0.16903	14.8310	14.9816
Gap	13	3	5	0.3	5	15.138	0.15928	14.8310	15.0285
Gap	14	3	5	0.3	5	14.798	0.26329	14.8310	14.9594
Gap	15	3	5	0.3	5	14.944	0.20876	14.8310	14.9548
Gap	16	3	5	0.3	5	14.896	0.09965	14.8310	14.9371
Gap	17	3	5	0.3	5	14.734	0.22512	14.8310	14.8762
Gap	18	3	5	0.3	5	15.046	0.24141	14.8310	14.9271
Gap	19	3	5	0.3	5	14.702	0.17880	14.8310	14.8596
Gap	20	3	5	0.3	5	14.788	0.16634	14.8310	14.8381

MEAN	_UCLE_	_STDDEV_	_EXLIM_
14.95	15.0350	0.21108	
14.95	15.0537	0.21108	
14.95	15.0617	0.21108	
14.95	15.0655	0.21108	
14.95	15.0673	0.21108	
14.95	15.0681	0.21108	
14.95	15.0686	0.21108	UPPER
14.95	15.0688	0.21108	
14.95	15.0689	0.21108	
14.95	15.0689	0.21108	
14.95	15.0689	0.21108	
14.95	15.0690	0.21108	
14.95	15.0690	0.21108	
14.95	15.0690	0.21108	
14.95	15.0690	0.21108	
14.95	15.0690	0.21108	
14.95	15.0690	0.21108	
14.95	15.0690	0.21108	
14.95	15.0690	0.21108	
14.95	15.0690	0.21108	

This data set contains one observation for each subgroup sample. The variable _EWMA_ contains the EWMAs. The variables _SUBX_, _SUBS_, and _SUBN_ contain the subgroup means, subgroup standard deviations, and subgroup sample sizes, respectively. The variables _LCLE_ and _UCLE_ contain the lower and upper control limits, and the variable _MEAN_ contains the central line. The variables _VAR_ and Day contain the *process* name and values of the *subgroup-variable*, respectively. For more information, see "OUTTABLE= Data Set" on page 817.

An OUTTABLE= data set can be read later as a TABLE= data set. For example, the following statements read Cliptab and display a EWMA chart (not shown here) identical to Figure 9.3:

```
title 'EWMA Chart for Gap Measurements';
proc macontrol table=Cliptab;
   ewmachart Gap*Day ;
run;
```

For more information, see "TABLE= Data Set" on page 821.

Reading Preestablished Control Limit Parameters

NOTE: See *Exponentially Weighted Moving Average Chart* in the SAS/QC Sample Library.

In the previous example, the OUTLIMITS= data set saved the control limit parameters in the data set Cliplim. This example shows how to apply these parameters to new data provided in the following data set:

```
data Clips1a;
   label Gap='Gap Measurement (mm)';
   input Day @;
   do i=1 to 5;
      input Gap @;
      output;
   end;
   drop i;
   datalines;
21  14.86 15.01 14.67 14.67 15.07
22  14.93 14.53 15.07 15.10 14.98
23  15.27 14.90 15.12 15.10 14.80
24  15.02 15.21 14.93 15.11 15.20
25  14.90 14.81 15.26 14.57 14.94
26  14.78 15.29 15.13 14.62 14.54
27  14.78 15.15 14.61 14.92 15.07
28  14.92 15.31 14.82 14.74 15.26
29  15.11 15.04 14.61 15.09 14.68
30  15.00 15.04 14.36 15.20 14.65
31  14.99 14.76 15.18 15.04 14.82
32  14.90 14.78 15.19 15.06 15.06
33  14.95 15.10 14.86 15.27 15.22
34  15.03 14.71 14.75 14.99 15.02
35  15.38 14.94 14.68 14.77 14.83
36  14.95 15.43 14.87 14.90 15.34
37  15.18 14.94 15.32 14.74 15.29
38  14.91 15.15 15.06 14.78 15.42
39  15.34 15.34 15.41 15.36 14.96
40  15.12 14.75 15.05 14.70 14.74
;
```

The following statements create an EWMA chart for the data in Clips1a using the control limit parameters in Cliplim:

```
ods graphics on;
title 'EWMA Chart for Second Set of Gap Measurements';
proc macontrol data=Clips1a limits=Cliplim;
   ewmachart Gap*Day / odstitle=title markers;
run;
```

The ODS GRAPHICS ON statement specified before the PROC MACONTROL statement enables ODS Graphics, so the EWMA chart is created using ODS Graphics instead of traditional graphics. The chart is shown in Figure 9.9.

Figure 9.9 EWMA Chart Using Preestablished Control Limit Parameters



The LIMITS= option in the PROC MACONTROL statement specifies the data set containing the control limit parameters. By default, this information is read from the first observation in the LIMITS= data set for which

- the value of _VAR_ matches the *process* name Gap
- the value of _SUBGRP_ matches the *subgroup-variable* name Day

Note that the EWMA plotted for the 39th day lies above the upper control limit, signalling an out-of-control process.

In this example, the LIMITS= data set was created in a previous run of the MACONTROL procedure. You can also create a LIMITS= data set with the DATA step. See "LIMITS= Data Set" on page 819 for details concerning the variables that you must provide, and see Example 9.1 for an illustration.

Syntax: EWMACHART Statement

The basic syntax for the EWMACHART statement is as follows:

EWMACHART *process* ∗ *subgroup-variable* / **WEIGHT=***value* < *options* > **;**

The general form of this syntax is as follows:

EWMACHART *processes* ∗ *subgroup-variable* < **(***block-variables***)** >
< **=***symbol-variable* | **=***'character'* > / **WEIGHT=***value* < *options* > **;**

Note that the WEIGHT= option is required unless its *value* is read from a LIMITS= data set. You can use any number of EWMACHART statements in the MACONTROL procedure. The components of the EWMACHART statement are described as follows.

process

processes

identify one or more processes to be analyzed. The specification of *process* depends on the input data set specified in the PROC MACONTROL statement.

- If raw data are read from a DATA= data set, *process* must be the name of the variable containing the raw measurements. For an example, see "Creating EWMA Charts from Raw Data" on page 786.
- If summary data are read from a HISTORY= data set, *process* must be the common prefix of the summary variables in the HISTORY= data set. For an example, see "Creating EWMA Charts from Subgroup Summary Data" on page 789.
- If summary data and control limits are read from a TABLE= data set, *process* must be the value of the variable _VAR_ in the TABLE= data set. For an example, see "Saving Control Limit Parameters" on page 793.

A *process* is required. If more than one *process* is specified, enclose the list in parentheses. For example, the following statements request distinct EWMA charts (each using a weight parameter of 0.3) for Weight, Length, and Width:

```
proc macontrol data=Measures;
   ewmachart (Weight Length Width)*Day / weight=0.3;
run;
```

subgroup-variable

is the variable that classifies the data into subgroups. The *subgroup-variable* is required. In the preceding EWMACHART statement, Day is the subgroup variable. For details, see "Subgroup Variables" on page 1871.

block-variables

are optional variables that group the data into blocks of consecutive subgroups. The blocks are labeled in a legend, and each *block-variable* provides one level of labels in the legend. See "Displaying Stratification in Blocks of Observations" on page 2026 for an example.

symbol-variable

is an optional variable whose levels (unique values) determine the symbol marker or plotting character used to plot the EWMAs.

- If you produce a line printer chart, an 'A' is displayed for the points corresponding to the first level of the *symbol-variable*, a 'B' is displayed for the points corresponding to the second level, and so on.
- If you produce traditional graphics, distinct symbol markers are displayed for points corresponding to the various levels of the *symbol-variable*. You can specify the symbol markers with SYMBOL*n* statements. See "Displaying Stratification in Levels of a Classification Variable" on page 2025 for an example.

character

specifies a plotting character for line printer charts. For example, the following statements create an EWMA chart using an asterisk (*) to plot the points:

```
proc macontrol data=Values lineprinter;
   ewmachart Length*Hour='*' / weight=0.3;
run;
```

options

specify chart parameters, enhance the appearance of the chart, request additional analyses, save results in data sets, and so on. The section "Summary of Options" on page 798, which follows, lists all options by function.

Summary of Options

The following tables list the EWMACHART statement options by function. Options unique to the MACONTROL procedure are listed in Table 9.1, and are described in detail in the section "Dictionary of Special Options" on page 806. Options that are common to both the MACONTROL and SHEWHART procedures are listed in Table 9.2. They are described in detail in "Dictionary of Options: SHEWHART Procedure" on page 1946.

Table 9.1 EWMACHART Statement Special Options

Option	Description
Options for Specifying Exponentially Weighted Moving Average Charts	
ALPHA=	requests probability limits for control charts
ASYMPTOTIC	requests constant control limits based on asymptotic expressions
LIMITN=	specifies either a fixed nominal sample size (*n*) for control limits or allows the control limits to vary with subgroup sample size

Table 9.1 *continued*

Option	Description
MU0=	specifies a standard (known) value μ_0 for the process mean
NOREADLIMITS	specifies that control limit parameters are not to be read from a LIMITS= data set
READALPHA	reads _ALPHA_ instead of _SIGMAS_ from the LIMITS= data set when both variables are available
READINDEX=	reads control limit parameters from the first observation in the LIMITS= data set where the variable _INDEX_ equals *value*
READLIMITS	reads control limit parameters from a LIMITS= data set (SAS 6.09 and earlier releases)
RESET	requests that the value of the EWMA be reset after each out-of-control point
SIGMA0=	specifies standard (known) value σ_0 for process standard deviation
SIGMAS=	specifies width of control limits in terms of multiple of standard error of plotted EWMAs
WEIGHT=	specifies weight assigned to the most recent subgroup mean in the computation of the EWMA
Options for Plotting Subgroup Means	
CMEANSYMBOL=	specifies color for MEANSYMBOL= symbol
MEANCHAR=	specifies *character* to plot subgroup means on line printer charts
MEANSYMBOL=	specifies symbol to plot subgroup means in traditional graphics

Table 9.2 EWMACHART Statement General Options

Option	Description
Options for Displaying Control Limits	
CINFILL=	specifies color for area inside control limits
CLIMITS=	specifies color of control limits, central line, and related labels
LCLLABEL=	specifies label for lower control limit
LIMLABSUBCHAR=	specifies a substitution character for labels provided as quoted strings; the character is replaced with the value of the control limit
LLIMITS=	specifies line type for control limits
NDECIMAL=	specifies number of digits to right of decimal place in default labels for control limits and central line
NOCTL	suppresses display of central line
NOLCL	suppresses display of lower control limit
NOLIMITLABEL	suppresses labels for control limits and central line
NOLIMITS	suppresses display of control limits
NOLIMITSFRAME	suppresses default frame around control limit information when multiple sets of control limits are read from a LIMITS= data set
NOLIMITSLEGEND	suppresses legend for control limits

Table 9.2 *continued*

Option	Description
NOUCL	suppresses display of upper control limit
UCLLABEL=	specifies label for upper control limit
WLIMITS=	specifies width for control limits and central line
XSYMBOL=	specifies label for central line
Process Mean and Standard Deviation Options	
SMETHOD=	specifies method for estimating process standard deviation σ
TYPE=	identifies parameters as estimates or standard values and specifies value of _TYPE_ in the OUTLIMITS= data set
Options for Plotting and Labeling Points	
ALLLABEL=	labels every point on EWMA chart
ALLLABEL2=	labels every point on trend chart
CLABEL=	specifies color for labels
CCONNECT=	specifies color for line segments that connect points on chart
CFRAMELAB=	specifies fill color for frame around labeled points
CNEEDLES=	specifies color for needles that connect points to central line
COUT=	specifies color for portions of line segments that connect points outside control limits
COUTFILL=	specifies color for shading areas between the connected points and control limits outside the limits
LABELANGLE=	specifies angle at which labels are drawn
LABELFONT=	specifies software font for labels (alias for the TESTFONT= option)
LABELHEIGHT=	specifies height of labels (alias for the TESTHEIGHT= option)
NEEDLES	connects points to central line with vertical needles
NOCONNECT	suppresses line segments that connect points on chart
NOTRENDCONNECT	suppresses line segments that connect points on trend chart
OUTLABEL=	labels points outside control limits
SYMBOLLEGEND=	specifies LEGEND statement for levels of *symbol-variable*
SYMBOLORDER=	specifies order in which symbols are assigned for levels of *symbol-variable*
TURNALL|TURNOUT	turns point labels so that they are strung out vertically
WNEEDLES=	specifies width of needles
Axis and Axis Label Options	
CAXIS=	specifies color for axis lines and tick marks
CFRAME=	specifies fill colors for frame for plot area
CTEXT=	specifies color for tick mark values and axis labels
DISCRETE	produces horizontal axis for discrete numeric group values

Table 9.2 *continued*

Option	Description
HAXIS=	specifies major tick mark values for horizontal axis
HEIGHT=	specifies height of axis label and axis legend text
HMINOR=	specifies number of minor tick marks between major tick marks on horizontal axis
HOFFSET=	specifies length of offset at both ends of horizontal axis
INTSTART=	specifies first major tick mark value on horizontal axis when a date, time, or datetime format is associated with numeric subgroup variable
NOHLABEL	suppresses label for horizontal axis
NOTICKREP	specifies that only the first occurrence of repeated, adjacent subgroup values is to be labeled on horizontal axis
NOVANGLE	requests vertical axis labels that are strung out vertically
SKIPHLABELS=	specifies thinning factor for tick mark labels on horizontal axis
SPLIT=	specifies splitting character for axis labels
TURNHLABELS	requests horizontal axis labels that are strung out vertically
VAXIS=	specifies major tick mark values for vertical axis of EWMA chart
VAXIS2=	specifies major tick mark values for vertical axis of trend chart
VFORMAT=	specifies format for primary vertical axis tick mark labels
VFORMAT2=	specifies format for secondary vertical axis tick mark labels
VMINOR=	specifies number of minor tick marks between major tick marks on vertical axis
VOFFSET=	specifies length of offset at both ends of vertical axis
VZERO	forces origin to be included in vertical axis for primary chart
VZERO2	forces origin to be included in vertical axis for secondary chart
WAXIS=	specifies width of axis lines
Plot Layout Options	
ALLN	plots means for all subgroups
BILEVEL	creates control charts using half-screens and half-pages
EXCHART	creates control charts for a process only when exceptions occur
INTERVAL=	natural time interval between consecutive subgroup positions when time, date, or datetime format is associated with a numeric subgroup variable
MAXPANELS=	maximum number of pages or screens for chart
NMARKERS	requests special markers for points corresponding to sample sizes not equal to nominal sample size for fixed control limits

Table 9.2 *continued*

Option	Description
NOCHART	suppresses creation of chart
NOFRAME	suppresses frame for plot area
NOLEGEND	suppresses legend for subgroup sample sizes
NPANELPOS=	specifies number of subgroup positions per panel on each chart
REPEAT	repeats last subgroup position on panel as first subgroup position of next panel
TOTPANELS=	specifies number of pages or screens to be used to display chart
TRENDVAR=	specifies list of trend variables
YPCT1=	specifies length of vertical axis on EWMA chart as a percentage of sum of lengths of vertical axes for EWMA and trend charts
ZEROSTD	displays EWMA chart regardless of whether $\hat{\sigma} = 0$
Reference Line Options	
CHREF=	specifies color for lines requested by HREF= and HREF2= options
CVREF=	specifies color for lines requested by VREF= and VREF2= options
HREF=	specifies position of reference lines perpendicular to horizontal axis on EWMA chart
HREF2=	specifies position of reference lines perpendicular to horizontal axis on trend chart
HREFDATA=	specifies position of reference lines perpendicular to horizontal axis on EWMA chart
HREF2DATA=	specifies position of reference lines perpendicular to horizontal axis on trend chart
HREFLABELS=	specifies labels for HREF= lines
HREF2LABELS=	specifies labels for HREF2= lines
HREFLABPOS=	specifies position of HREFLABELS= and HREF2LABELS= labels
LHREF=	specifies line type for HREF= and HREF2= lines
LVREF=	specifies line type for VREF= and VREF2= lines
NOBYREF	specifies that reference line information in a data set applies uniformly to charts created for all BY groups
VREF=	specifies position of reference lines perpendicular to vertical axis on EWMA chart
VREF2=	specifies position of reference lines perpendicular to vertical axis on trend chart
VREFLABELS=	specifies labels for VREF= lines
VREF2LABELS=	specifies labels for VREF2= lines
VREFLABPOS=	position of VREFLABELS= and VREF2LABELS= labels

Table 9.2 *continued*

Option	Description
Grid Options	
CGRID=	specifies color for grid requested with GRID or ENDGRID option
ENDGRID	adds grid after last plotted point
GRID	adds grid to control chart
LENDGRID=	specifies line type for grid requested with the ENDGRID option
LGRID=	specifies line type for grid requested with the GRID option
WGRID=	specifies width of grid lines
Clipping Options	
CCLIP=	specifies color for plot symbol for clipped points
CLIPFACTOR=	determines extent to which extreme points are clipped
CLIPLEGEND=	specifies text for clipping legend
CLIPLEGPOS=	specifies position of clipping legend
CLIPSUBCHAR=	specifies substitution character for CLIPLEGEND= text
CLIPSYMBOL=	specifies plot symbol for clipped points
CLIPSYMBOLHT=	specifies symbol marker height for clipped points
Graphical Enhancement Options	
ANNOTATE=	specifies annotate data set that adds features to EWMA chart
ANNOTATE2=	specifies annotate data set that adds features to trend chart
DESCRIPTION=	specifies description of EWMA chart's GRSEG catalog entry
FONT=	specifies software font for labels and legends on charts
NAME=	specifies name of EWMA chart's GRSEG catalog entry
PAGENUM=	specifies the form of the label used in pagination
PAGENUMPOS=	specifies the position of the page number requested with the PAGENUM= option
WTREND=	specifies width of line segments connecting points on trend chart
Options for Producing Graphs Using ODS Styles	
BLOCKVAR=	specifies one or more variables whose values define colors for filling background of *block-variable* legend
CFRAMELAB	draws a frame around labeled points
COUT	draw portions of line segments that connect points outside control limits in a contrasting color
CSTAROUT	specifies that portions of stars exceeding inner or outer circles are drawn using a different color
OUTFILL	shades areas between control limits and connected points lying outside the limits
STARFILL=	specifies a variable identfying groups of stars filled with different colors

Table 9.2 *continued*

Option	Description
STARS=	specifies a variable identfying groups of stars whose outlines are drawn with different colors
Options for ODS Graphics	
BLOCKREFTRANSPARENCY=	specifies the wall fill transparency for blocks and phases
INFILLTRANSPARENCY=	specifies the control limit infill transparency
MARKERS	plots subgroup points with markers
NOBLOCKREF	suppresses block and phase reference lines
NOBLOCKREFFILL	suppresses block and phase wall fills
NOFILLLEGEND	suppresses legend for levels of a STARFILL= variable
NOPHASEREF	suppresses block and phase reference lines
NOPHASEREFFILL	suppresses block and phase wall fills
NOREF	suppresses block and phase reference lines
NOREFFILL	suppresses block and phase wall fills
NOSTARFILLLEGEND	suppresses legend for levels of a STARFILL= variable
NOTRANSPARENCY	disables transparency in ODS Graphics output
ODSFOOTNOTE=	specifies a graph footnote
ODSFOOTNOTE2=	specifies a secondary graph footnote
ODSLEGENDEXPAND	specifies that legend entries contain all levels observed in the data
ODSTITLE=	specifies a graph title
ODSTITLE2=	specifies a secondary graph title
OUTFILLTRANSPARENCY=	specifies control limit outfill transparency
OVERLAYURL=	specifies URLs to associate with overlay points
OVERLAY2URL=	specifies URLs to associate with overlay points on secondary chart
PHASEPOS=	specifies vertical position of phase legend
PHASEREFLEVEL=	associates phase and block reference lines with either innermost or the outermost level
PHASEREFTRANSPARENCY=	specifies the wall fill transparency for blocks and phases
REFFILLTRANSPARENCY=	specifies the wall fill transparency for blocks and phases
SIMULATEQCFONT	draws central line labels using a simulated software font
STARTRANSPARENCY=	specifies star fill transparency
URL=	specifies a variable whose values are URLs to be associated with subgroups
URL2=	specifies a variable whose values are URLs to be associated with subgroups on secondary chart
Input Data Set Options	
MISSBREAK	specifies that observations with missing values are not to be processed
Output Data Set Options	
OUTHISTORY=	creates output data set containing subgroup summary statistics
OUTINDEX=	specifies value of _INDEX_ in the OUTLIMITS= data set

Table 9.2 *continued*

Option	Description
OUTLIMITS=	creates output data set containing control limits
OUTTABLE=	creates output data set containing subgroup summary statistics and control limits
Tabulation Options	
NOTE: specifying (EXCEPTIONS) after a tabulation option creates a table for exceptional points only.	
TABLE	creates a basic table of subgroup means, subgroup sample sizes, and control limits
TABLEALL	is equivalent to the options TABLE, TABLECENTRAL, TABLEID, TABLELEGEND, TABLEOUTLIM, and TABLETESTS
TABLECENTRAL	augments basic table with values of central lines
TABLEID	augments basic table with columns for ID variables
TABLEOUTLIM	augments basic table with columns indicating control limits exceeded
Block Variable Legend Options	
BLOCKLABELPOS=	specifies position of label for *block-variable* legend
BLOCKLABTYPE=	specifies text size of *block-variable* legend
BLOCKPOS=	specifies vertical position of *block-variable* legend
BLOCKREP	repeats identical consecutive labels in *block-variable* legend
CBLOCKLAB=	specifies fill colors for frames enclosing variable labels in *block-variable* legend
CBLOCKVAR=	specifies one or more variables whose values are colors for filling background of *block-variable* legend
Phase Options	
CPHASELEG=	specifies text color for *phase* legend
OUTPHASE=	specifies value of _PHASE_ in the OUTHISTORY= data set
PHASEBREAK	disconnects last point in a *phase* from first point in next *phase*
PHASELABTYPE=	specifies text size of *phase* legend
PHASELEGEND	displays *phase* labels in a legend across top of chart
PHASELIMITS	labels control limits for each phase, provided they are constant within that phase
PHASEREF	delineates *phases* with vertical reference lines
READPHASES=	specifies *phases* to be read from an input data set
Star Options	
CSTARCIRCLES=	specifies color for STARCIRCLES= circles
CSTARFILL=	specifies color for filling stars
CSTAROUT=	specifies outline color for stars exceeding inner or outer circles
CSTARS=	specifies color for outlines of stars
LSTARCIRCLES=	specifies line types for STARCIRCLES= circles

Table 9.2 *continued*

Option	Description
LSTARS=	specifies line types for outlines of STARVERTICES= stars
STARBDRADIUS=	specifies radius of outer bound circle for vertices of stars
STARCIRCLES=	specifies reference circles for stars
STARINRADIUS=	specifies inner radius of stars
STARLABEL=	specifies vertices to be labeled
STARLEGEND=	specifies style of legend for star vertices
STARLEGENDLAB=	specifies label for STARLEGEND= legend
STAROUTRADIUS=	specifies outer radius of stars
STARSPECS=	specifies method used to standardize vertex variables
STARSTART=	specifies angle for first vertex
STARTYPE=	specifies graphical style of star
STARVERTICES=	superimposes star at each point on EWMA chart
WSTARCIRCLES=	specifies width of STARCIRCLES= circles
WSTARS=	specifies width of STARVERTICES= stars
Options for Interactive Control Charts	
HTML=	specifies a variable whose values create links to be associated with subgroups
HTML2=	specifies variable whose values create links to be associated with subgroups on secondary chart
HTML_LEGEND=	specifies a variable whose values create links to be associated with symbols in the symbol legend
WEBOUT=	creates an OUTTABLE= data set with additional graphics coordinate data
Options for Line Printer Charts	
CLIPCHAR=	specifies plot character for clipped points
CONNECTCHAR=	specifies character used to form line segments that connect points on chart
HREFCHAR=	specifies line character for HREF= and HREF2= lines
SYMBOLCHARS=	specifies characters indicating *symbol-variable*
VREFCHAR=	specifies line character for VREF= and VREF2= lines

Dictionary of Special Options

ALPHA=*value*

requests *probability limits*. If you specify ALPHA=α, the control limits are computed so that the probability is α that a single EWMA exceeds its control limits. The value of α can range between 0 and 1. This assumes that the process is in statistical control and that the data follow a normal distribution. For the equations used to compute probability limits, see “Control Limits” on page 810.

Note the following:

- As an alternative to specifying ALPHA=α, you can read α from the variable _ALPHA_ in a LIMITS= data set by specifying the READALPHA option.

- As an alternative to specifying ALPHA=α (or reading _ALPHA_ from a LIMITS= data set), you can request "$k\sigma$ control limits" by specifying SIGMAS=k (or reading _SIGMAS_ from a LIMITS= data set).

If you specify neither the ALPHA= option nor the SIGMAS= option, the procedure computes 3σ control limits by default.

ASYMPTOTIC

requests constant upper and lower control limits based on the following asymptotic expressions:

$$\text{LCL} = \overline{\overline{X}} - k\hat{\sigma}\sqrt{r/n(2-r)}$$

$$\text{UCL} = \overline{\overline{X}} + k\hat{\sigma}\sqrt{r/n(2-r)}$$

Here r is the weight parameter ($0 < r \leq 1$), and n is the nominal sample size associated with the control limits. Substitute $\Phi^{-1}(1-\alpha/2)$ for k if you specify probability limits with the ALPHA= option. When you do not specify the ASYMPTOTIC option, the control limits are computed using the exact formulas in Table 9.3. Use the ASYMPTOTIC option only if all the subgroup sample sizes are the same or if you specify LIMITN=n. See Example 9.2.

CMEANSYMBOL=*color*

specifies the *color* used for the symbol requested with the MEANSYMBOL= option in traditional graphics. This option is ignored unless you are producing traditional graphics.

LIMITN=*n*

LIMITN=VARYING

specifies either a fixed or varying nominal sample size for the control limits.

If you specify LIMITN=n, EWMAs are calculated and displayed only for those subgroups with a sample size equal to n, unless you also specify the ALLN option, which causes all the EWMAs to be calculated and displayed. By default (or if you specify LIMITN=VARYING), EWMAs are calculated and displayed for all subgroups, regardless of sample size.

MEANCHAR='*character*'

specifies a *character* used in legacy line printer charts to plot the subgroup mean for each subgroup. By default, subgroup means are not plotted. This option is ignored unless you specify the LINEPRINTER option in the PROC MACONTROL statement.

MEANSYMBOL=*keyword*

specifies a symbol used to plot the subgroup mean for each subgroup in traditional graphics. By default, subgroup means are not plotted. This option is ignored unless you are producing traditional graphics.

MU0=*value*

specifies a known (standard) value μ_0 for the process mean μ. By default, μ is estimated from the data. See Example 9.1.

NOTE: As an alternative to specifying MU0=μ_0, you can read a predetermined value for μ_0 from the variable _MEAN_ in a LIMITS= data set.

NOREADLIMITS

specifies that control limit parameters for each *process* listed in the EWMACHART statement are *not* to be read from the LIMITS= data set specified in the PROC MACONTROL statement.

The following example illustrates the NOREADLIMITS option:

```
proc macontrol data=Pistons limits=Diamlim;
   ewmachart Diameter*Hour;
   ewmachart Diameter*Hour / noreadlimits weight=0.3;
run;
```

The first EWMACHART statement reads the control limits from the first observation in the data set Diamlim for which the variable _VAR_ is equal to 'Diameter' and the variable _SUBGRP_ is equal to 'Hour'. The second EWMACHART statement computes estimates of the process mean and standard deviation for the control limits from the measurements in the data set Pistons. Note that the second EWMACHART statement is equivalent to the following statements, which would be more commonly used:

```
proc macontrol data=Pistons;
   ewmachart Diameter*Hour / weight=0.3;
run;
```

For more information about reading control limit parameters from a LIMITS= data set, see the READLIMITS option later in this list.

READALPHA

specifies that the variable _ALPHA_, rather than the variable _SIGMAS_, is to be read from a LIMITS= data set when both variables are available in the data set. Thus the limits displayed are probability limits. If you do not specify the READALPHA option, then _SIGMAS_ is read by default.

READINDEX=*'value'*

reads control limit parameters from a LIMITS= data set (specified in the PROC MACONTROL statement) for each *process* listed in the EWMACHART statement.

The control limit parameters for a particular *process* are read from the first observation in the LIMITS= data set for which

- the value of _VAR_ matches *process*
- the value of _SUBGRP_ matches the *subgroup-variable*
- the value of _INDEX_ matches *value*

The *value* can be up to 48 characters and must be enclosed in quotes.

READLIMITS

specifies that control limit parameters are to be read from a LIMITS= data set specified in the PROC MACONTROL statement. The parameters for a particular *process* are read from the first observation in the LIMITS= data set for which

- the value of _VAR_ matches *process*
- the value of _SUBGRP_ matches the *subgroup variable*

NOTE: In SAS 6.10 and later releases, the READLIMITS option is not necessary.

RESET

requests that the value of the EWMA be reset after each out-of-control point. Specifically, when a point exceeds the control limits, the EWMA for the next subgroup is computed as the weighted average of the subgroup mean and the overall mean. By default, the EWMAs are not reset.

SIGMA0=*value*

specifies a known (standard) value σ_0 for the process standard deviation σ. The *value* must be positive. By default, the MACONTROL procedure estimates σ from the data using the formulas given in "Methods for Estimating the Standard Deviation" on page 822.

NOTE: As an alternative to specifying SIGMA0=σ_0, you can read a predetermined value for σ_0 from the variable _STDDEV_ in a LIMITS= data set.

SIGMAS=*value*

specifies the width of the control limits in terms of the multiple k of the standard error of the plotted EWMAs on the chart. The value of k must be positive. By default, k = 3 and the control limits are 3σ limits.

WEIGHT=*value*

specifies the weight r assigned to the most recent subgroup mean in the computation of the EWMA ($0 < r \leq 1$). The WEIGHT= option is required unless you read control limit parameters from a LIMITS= data set or a TABLE= data set. See the section "Choosing the Value of the Weight Parameter" on page 812 for details.

Details: EWMACHART Statement

Constructing EWMA Charts

The following notation is used in this section:

E_i	exponentially weighted moving average for the ith subgroup
r	EWMA weight parameter ($0 < r \leq 1$)
μ	process mean (expected value of the population of measurements)
σ	process standard deviation (standard deviation of the population of measurements)
x_{ij}	jth measurement in ith subgroup, with j =1, 2, 3, ..., n_i
n_i	sample size of ith subgroup
$\overline{X}_i$	mean of measurements in ith subgroup. If $n_i = 1$, then the subgroup mean reduces to the single observation in the subgroup
$\overline{\overline{X}}$	weighted average of subgroup means
$\Phi^{-1}(\cdot)$	inverse standard normal function

Plotted Points

Each point on the chart indicates the value of the exponentially weighted moving average (EWMA) for that subgroup. The EWMA for the ith subgroup (E_i) is defined recursively as

$$E_i = r\overline{X}_i + (1 - r)E_{i-1}, \quad i > 0$$

where r is a weight parameter ($0 < r \leq 1$). Some authors (for example, Hunter 1986 and Crowder 1987a,b) use the symbol λ instead of r for the weight. You can specify the weight with the WEIGHT= option in the

EWMACHART statement or with the variable _WEIGHT_ in a LIMITS= data set. If you specify a known value (μ_0) for μ, $E_0 = \mu_0$; otherwise, $E_0 = \overline{\overline{X}}$.

The preceding equation can be rewritten as

$$E_i = E_{i-1} + r(\overline{X}_i - E_{i-1})$$

which expresses the current EWMA as the previous EWMA plus the weighted error in the prediction of the current mean based on the previous EWMA.

The EWMA for the ith subgroup can also be written as

$$E_i = r\sum_{j=0}^{i-1}(1-r)^j \overline{X}_{i-j} + (1-r)^i E_0$$

which expresses the EWMA as a weighted average of past subgroup means, where the weights decline exponentially, and the heaviest weight is assigned to the most recent subgroup mean.

Central Line

By default, the central line on an EWMA chart indicates an estimate for μ, which is computed as

$$\hat{\mu} = \overline{\overline{X}} = \frac{n_1\bar{X}_1 + \cdots + n_N\bar{X}_N}{n_1 + \cdots + n_N}$$

If you specify a known value (μ_0) for μ, the central line indicates the value of μ_0.

Control Limits

You can compute the limits in the following ways:

- as a specified multiple (k) of the standard error of E_i above and below the central line. The default limits are computed with $k = 3$ (these are referred to as 3σ *limits*).
- as probability limits defined in terms of α, a specified probability that E_i exceeds the limits

The following table presents the formulas for the limits:

Table 9.3 Limits for an EWMA Chart

Control Limits
LCL = lower limit = $\overline{\overline{X}} - k\hat{\sigma}r\sqrt{\sum_{j=0}^{i-1}(1-r)^{2j}/n_{i-j}}$
UCL = upper limit = $\overline{\overline{X}} + k\hat{\sigma}r\sqrt{\sum_{j=0}^{i-1}(1-r)^{2j}/n_{i-j}}$

Probability Limits
LCL = lower limit = $\overline{\overline{X}} - \Phi^{-1}(1-\alpha/2)\hat{\sigma}r\sqrt{\sum_{j=0}^{i-1}(1-r)^{2j}/n_{i-j}}$
UCL = upper limit = $\overline{\overline{X}} + \Phi^{-1}(1-\alpha/2)\hat{\sigma}r\sqrt{\sum_{j=0}^{i-1}(1-r)^{2j}/n_{i-j}}$

These formulas assume that the data are normally distributed. If standard values μ_0 and σ_0 are available for μ and σ, respectively, replace $\overline{\overline{X}}$ with μ_0 and $\hat{\sigma}$ with σ_0 in Table 9.3. Note that the limits vary with both n_i and i.

If the subgroup sample sizes are constant ($n_i = n$), the formulas for the control limits simplify to

$$\text{LCL} = \overline{\overline{X}} - k\hat{\sigma}\sqrt{r(1-(1-r)^{2i})/n(2-r)}$$

$$\text{UCL} = \overline{\overline{X}} + k\hat{\sigma}\sqrt{r(1-(1-r)^{2i})/n(2-r)}$$

Consequently, when the subgroup sample sizes are constant, the width of the control limits increases monotonically with i. For probability limits, replace k with $\Phi^{-1}(1-\alpha/2)$ in the previous equations. Refer to Roberts (1959) and Montgomery (1996).

As i becomes large, the upper and lower control limits approach constant values:

$$\text{LCL} = \overline{\overline{X}} - k\hat{\sigma}\sqrt{r/n(2-r)}$$

$$\text{UCL} = \overline{\overline{X}} + k\hat{\sigma}\sqrt{r/n(2-r)}$$

Some authors base the control limits for EWMA charts on the asymptotic expressions in the two previous equations. For asymptotic probability limits, replace k with $\Phi^{-1}(1-\alpha/2)$ in these equations. You can display asymptotic limits by specifying the ASYMPTOTIC option.

Uniformly weighted moving average charts and exponentially weighted moving average charts have similar properties, and their asymptotic control limits are identical provided that

$$r = 2/(w+1)$$

where w is the weight factor for uniformly weighted moving average charts. Refer to Wadsworth, Stephens, and Godfrey (1986) and the American Society for Quality Control (1983).

You can specify parameters for the EWMA limits as follows:

- Specify k with the SIGMAS= option or with the variable _SIGMAS_ in a LIMITS= data set.
- Specify α with the ALPHA= option or with the variable _ALPHA_ in a LIMITS= data set.
- Specify a constant nominal sample size $n_i \equiv n$ for the control limits with the LIMITN= option or with the variable _LIMITN_ in a LIMITS= data set.
- Specify r with the WEIGHT= option or with the variable _WEIGHT_ in a LIMITS= data set.
- Specify μ_0 with the MU0= option or with the variable _MEAN_ in a LIMITS= data set.
- Specify σ_0 with the SIGMA0= option or with the variable _STDDEV_ in a LIMITS= data set.

Choosing the Value of the Weight Parameter

Various approaches have been proposed for choosing the value of r.

- Hunter (1986) states that the choice “can be left to the judgment of the quality control analyst” and points out that the smaller the value of r, “the greater the influence of the historical data.”
- Hunter (1986) also discusses a least squares procedure for estimating r from the data, **assuming an exponentially weighted moving average model for the data**. In this context, the fitted EWMA model provides a forecast of the process that is the basis for dynamic process control. You can use the ARIMA procedure in SAS/ETS® software to compute the least squares estimate of r. (Refer to *SAS/ETS User’s Guide* for information about PROC ARIMA.) Also see “Autocorrelation in Process Data” on page 2096.
- A number of authors have studied the design of EWMA control schemes based on average run length (ARL) computations. The ARL is the expected number of points plotted before a shift is detected. Ideally, the ARL should be short when a shift occurs, and it should be long when there is no shift (the process is in control.) The effect of r on the ARL was described by Roberts (1959), who used simulation methods. The ARL function was approximated and tabulated by Robinson and Ho (1978), and a more general method for studying run-length distributions of EWMA charts was given by Crowder (1987a, b). Unlike Hunter (1986), these authors assume the data are independent and identically distributed; typically the normal distribution is assumed for the data, although the methods extend to nonnormal distributions. A more detailed discussion of the ARL approach follows.

Average run lengths for two-sided EWMA charts are shown in Table 9.4, which is patterned after Table 1 of Crowder (1987a, b). The ARLs were computed using the EWMAARL DATA step function (see “EWMAARL Function” on page 2180 for details on the EWMAARL function). Note that Crowder (1987a, b). uses the notation L in place of k and the notation λ in place of r.

You can use Table 9.4 to find a combination of k and r that yields a desired ARL for an in-control process ($\delta = 0$) and for a specified shift of δ. Note that δ is assumed to be standardized; in other words, if a shift of Δ is to be detected in the process mean μ, and if σ is the process standard deviation, you should select the table entry with

$$\delta = \Delta / (\sigma / \sqrt{n})$$

where n is the subgroup sample size. Thus, δ can be regarded as the shift in the sampling distribution of the subgroup mean.

For example, suppose you want to construct an EWMA scheme with an in-control ARL of 90 and an ARL of 9 for detecting a shift of $\delta = 1$. Table 9.4 shows that the combination $r = 0.5$ and $k = 2.5$ yields an in-control ARL of 91.17 and an ARL of 8.27 for $\delta = 1$.

Crowder (1987a, b) cautions that setting the in-control ARL at a desired level does not guarantee that the probability of an early false signal is acceptable. For further details concerning the distribution of the ARL, refer to Crowder (1987a, b).

In addition to using Table 9.4 or the EWMAARL DATA step function to choose a EWMA scheme with desired average run length properties, you can use them to evaluate an existing EWMA scheme. For example, the “Getting Started” section of this chapter contains EWMA schemes with $r = 0.3$ and $k = 3$. The following statements use the EWMAARL function to compute the in-control ARL and the ARLs for shifts of $\delta = 0.25$ and $\delta = 0.5$:

```
data arlewma;
   arlin = ewmaarl(  0,0.3,3.0);
   arl1  = ewmaarl(.25,0.3,3.0);
   arl2  = ewmaarl(.50,0.3,3.0);
run;
```

The in-control ARL is 465.553, the ARL for $\delta = .25$ is 178.741, and the ARL for $\delta = .5$ is 53.1603. See Example 9.5 for an illustration of how to use the EWMAARL function to compute average run lengths for various EWMA schemes and shifts.

Table 9.4 Average Run Lengths for Two-Sided EWMA Charts

		r (weight parameter)					
k	δ	**0.05**	**0.10**	**0.25**	**0.50**	**0.75**	**1.00**
2.0	0.00	127.53	73.28	38.56	26.45	22.88	21.98
2.0	0.25	43.94	34.49	24.83	20.12	18.86	19.13
2.0	0.50	18.97	15.53	12.74	11.89	12.34	13.70
2.0	0.75	11.64	9.36	7.62	7.29	7.86	9.21
2.0	1.00	8.38	6.62	5.24	4.91	5.26	6.25
2.0	1.25	6.56	5.13	3.96	3.59	3.76	4.40
2.0	1.50	5.41	4.20	3.19	2.80	2.84	3.24
2.0	1.75	4.62	3.57	2.68	2.29	2.26	2.49
2.0	2.00	4.04	3.12	2.32	1.95	1.88	2.00
2.0	2.25	3.61	2.78	2.06	1.70	1.61	1.67
2.0	2.50	3.26	2.52	1.85	1.51	1.42	1.45
2.0	2.75	2.99	2.32	1.69	1.37	1.29	1.29
2.0	3.00	2.76	2.16	1.55	1.26	1.19	1.19
2.0	3.25	2.56	2.03	1.43	1.18	1.13	1.12
2.0	3.50	2.39	1.93	1.32	1.12	1.08	1.07
2.0	3.75	2.26	1.83	1.24	1.08	1.05	1.04
2.0	4.00	2.15	1.73	1.17	1.05	1.03	1.02
2.5	0.00	379.09	223.35	124.18	91.17	82.49	80.52
2.5	0.25	73.98	66.59	59.66	58.33	61.07	65.77
2.5	0.50	26.63	23.63	23.28	27.16	33.26	41.49
2.5	0.75	15.41	12.95	11.96	13.96	18.05	24.61
2.5	1.00	10.79	8.75	7.52	8.27	10.57	14.92
2.5	1.25	8.31	6.60	5.39	5.52	6.75	9.46
2.5	1.50	6.78	5.31	4.18	4.03	4.65	6.30
2.5	1.75	5.75	4.46	3.43	3.14	3.43	4.41
2.5	2.00	5.00	3.86	2.92	2.57	2.67	3.24
2.5	2.25	4.43	3.42	2.56	2.18	2.17	2.49
2.5	2.50	4.00	3.07	2.29	1.90	1.83	2.00
2.5	2.75	3.64	2.80	2.08	1.69	1.59	1.67
2.5	3.00	3.36	2.57	1.91	1.52	1.41	1.45
2.5	3.25	3.12	2.39	1.77	1.39	1.29	1.29
2.5	3.50	2.92	2.24	1.64	1.28	1.19	1.19
2.5	3.75	2.74	2.13	1.52	1.20	1.13	1.12
2.5	4.00	2.58	2.04	1.42	1.13	1.08	1.07

Table 9.4 *continued*

k	δ	**0.05**	**0.10**	**0.25**	**0.50**	**0.75**	**1.00**
3.0	0.00	1383.62	842.15	502.90	397.46	374.50	370.40
3.0	0.25	133.61	144.74	171.09	208.54	245.76	281.15
3.0	0.50	37.33	37.41	48.45	75.35	110.95	155.22
3.0	0.75	19.95	17.90	20.16	31.46	50.92	81.22
3.0	1.00	13.52	11.38	11.15	15.74	25.64	43.89
3.0	1.25	10.24	8.32	7.39	9.21	14.26	24.96
3.0	1.50	8.26	6.57	5.47	6.11	8.72	14.97
3.0	1.75	6.94	5.45	4.34	4.45	5.80	9.47
3.0	2.00	6.00	4.67	3.62	3.47	4.15	6.30
3.0	2.25	5.30	4.10	3.11	2.84	3.16	4.41
3.0	2.50	4.76	3.67	2.75	2.41	2.52	3.24
3.0	2.75	4.32	3.32	2.47	2.10	2.09	2.49
3.0	3.00	3.97	3.05	2.26	1.87	1.79	2.00
3.0	3.25	3.67	2.82	2.09	1.69	1.57	1.67
3.0	3.50	3.42	2.62	1.95	1.53	1.41	1.45
3.0	3.75	3.22	2.45	1.84	1.41	1.29	1.29
3.0	4.00	3.04	2.30	1.73	1.31	1.20	1.19
3.5	0.00	12851.0	4106.4	2640.16	2227.34	2157.99	2149.34
3.5	0.25	281.09	381.29	625.78	951.18	1245.90	1502.76
3.5	0.50	53.58	64.72	123.43	267.36	468.68	723.81
3.5	0.75	25.62	25.33	38.68	88.70	182.12	334.40
3.5	1.00	16.65	14.79	17.71	35.97	78.05	160.95
3.5	1.25	12.36	10.37	10.48	17.64	37.15	81.80
3.5	1.50	9.86	8.00	7.25	10.19	19.63	43.96
3.5	1.75	8.22	6.54	5.52	6.70	11.46	24.96
3.5	2.00	7.07	5.55	4.47	4.86	7.33	14.97
3.5	2.25	6.21	4.83	3.77	3.78	5.08	9.47
3.5	2.50	5.55	4.29	3.28	3.10	3.76	6.30
3.5	2.75	5.03	3.87	2.91	2.63	2.94	4.41
3.5	3.00	4.60	3.54	2.63	2.30	2.40	3.24
3.5	3.25	4.25	3.26	2.41	2.05	2.03	2.49
3.5	3.50	3.95	3.03	2.23	1.85	1.76	2.00
3.5	3.75	3.70	2.84	2.10	1.69	1.56	1.67
3.5	4.00	3.47	2.66	1.99	1.55	1.40	1.45

Output Data Sets

OUTLIMITS= Data Set

The OUTLIMITS= data set saves the control limit parameters. The following variables can be saved:

Variable	Description
ALPHA	probability (α) of exceeding limits
INDEX	optional identifier for the control limits specified with the OUTINDEX= option
LIMITN	sample size associated with the control limits
MEAN	process mean ($\overline{\overline{X}}$ or μ_0)
SIGMAS	multiple (k) of standard error of E_i
STDDEV	process standard deviation ($\hat{\sigma}$ or σ_0)
SUBGRP	*subgroup-variable* specified in the EWMACHART statement
TYPE	type (estimate or standard value) of _MEAN_ and _STDDEV_
VAR	*process* specified in the EWMACHART statement
WEIGHT	weight (r) assigned to most recent subgroup mean in computation of EWMA

The OUTLIMITS= data set does not contain the control limits; instead, it contains control limit parameters that can be used to recompute the control limits.

Notes:

1. If the control limits vary with subgroup sample size, the special missing value V is assigned to the variable _LIMITN_.
2. If the limits are defined in terms of a multiple k of the standard error of E_i, the value of _ALPHA_ is computed as $\alpha = 2(1 - \Phi(k))$, where $\Phi(\cdot)$ is the standard normal distribution function.
3. If the limits are probability limits, the value of _SIGMAS_ is computed as $k = \Phi^{-1}(1 - \alpha/2)$, where Φ^{-1} is the inverse standard normal distribution function.
4. Optional BY variables are saved in the OUTLIMITS= data set.

The OUTLIMITS= data set contains one observation for each *process* specified in the EWMACHART statement.

You can use OUTLIMITS= data sets

- to keep a permanent record of the control limit parameters
- to write reports. You may prefer to use OUTTABLE= data sets for this purpose.
- as LIMITS= data sets in subsequent runs of PROC MACONTROL

For an example of an OUTLIMITS= data set, see the section "Saving Control Limit Parameters" on page 793.

OUTHISTORY= Data Set

The OUTHISTORY= data set saves subgroup summary statistics. The following variables can be saved:

- the *subgroup-variable*
- a subgroup mean variable named by *process* suffixed with *X*
- a subgroup standard deviation variable named by *process* suffixed with *S*
- a subgroup EWMA variable named by *process* suffixed with *E*
- a subgroup sample size variable named by *process* suffixed with *N*

Given a *process* name that contains 32 characters, the procedure first shortens the name to its first 16 characters and its last 15 characters, and then it adds the suffix.

Subgroup summary variables are created for each *process* specified in the EWMACHART statement. For example, consider the following statements:

```
proc macontrol data=Clips;
   ewmachart (Gap YieldStrength)*Day /
      weight     = 0.2
      outhistory = Cliphist;
run;
```

The data set Cliphist would contain nine variables named Day, GapX, GapS, GapE, GapN, YieldStrengthX, YieldStrengthS, YieldStrengthE, and YieldStrengthN.

Additionally, the following variables, if specified, are included:

- BY variables
- *block-variables*
- *symbol-variable*
- ID variables
- _PHASE_ (if the OUTPHASE= option is specified)

For an example of an OUTHISTORY= data set, see the section "Saving Summary Statistics" on page 792.

OUTTABLE= Data Set

The OUTTABLE= data set saves subgroup summary statistics, control limits, and related information. The following variables can be saved:

Variable	Description
ALPHA	probability (α) of exceeding control limits
EXLIM	control limit exceeded on EWMA chart
EWMA	exponentially weighted moving average
LCLE	lower control limit for EWMA
LIMITN	nominal sample size associated with the control limits
MEAN	process mean
SIGMAS	multiple (k) of the standard error associated with control limits
subgroup	values of the subgroup variable
SUBN	subgroup sample size
SUBS	subgroup standard deviation
SUBX	subgroup mean
UCLE	upper control limit for EWMA
VAR	*process* specified in the EWMACHART statement
WEIGHT	weight (r) assigned to most recent subgroup mean in computation of EWMA

In addition, the following variables, if specified, are included:

- BY variables
- *block-variables*
- ID variables
- _PHASE_ (if the READPHASES= option is specified)
- *symbol-variable*

Notes:

1. Either the variable _ALPHA_ or the variable _SIGMAS_ is saved depending on how the control limits are defined (with the ALPHA= or SIGMAS= options, respectively, or with the corresponding variables in a LIMITS= data set).
2. The variables _VAR_ and _EXLIM_ are character variables of length 8. The variable _PHASE_ is a character variable of length 48. All other variables are numeric.

For an example of an OUTTABLE= data set, see "Saving Control Limit Parameters" on page 793.

ODS Tables

The following table summarizes the ODS tables that you can request with the EWMACHART statement.

Table 9.5 ODS Tables Produced with the EWMACHART Statement

Table Name	Description	Options
EWMACHART	exponentially weighted moving average chart summary statistics	TABLE, TABLEALL, TABLEC, TABLEID, TABLEOUT
Parameters	exponentially weighted moving average parameters	TABLE, TABLEALL, TABLEC, TABLEID, TABLEOUT

ODS Graphics

Before you create ODS Graphics output, ODS Graphics must be enabled (for example, by using the ODS GRAPHICS ON statement). For more information about enabling and disabling ODS Graphics, see the section "Enabling and Disabling ODS Graphics" (Chapter 21, *SAS/STAT User's Guide*). **NOTE:** In SAS/QC 13.1 the MACONTROL procedure does not support the creation of graphs that are editable with the ODS Graphics Editor.

The appearance of a graph produced with ODS Graphics is determined by the style associated with the ODS destination where the graph is produced. EWMACHART options used to control the appearance of traditional graphics are ignored for ODS Graphics output. Options for Producing Graphs Using ODS Styles lists options that can be used to control the appearance of graphs produced with ODS Graphics or with traditional graphics using ODS styles. Options for ODS Graphics lists options to be used exclusively with ODS Graphics. Detailed descriptions of these options are provided in "Dictionary of Options: SHEWHART Procedure" on page 1946.

When ODS Graphics is in effect, the EWMACHART statement assigns a name to the graph it creates. You can use this name to reference the graph when using ODS. The name is listed in Table 9.6.

Table 9.6 ODS Graphics Produced by the EWMACHART Statement

ODS Graph Name	Plot Description
EWMAChart	EWMA chart

See Chapter 3, "SAS/QC Graphics," for more information about ODS Graphics and other methods for producing charts.

Input Data Sets

DATA= Data Set

You can read raw data (process measurements) from a DATA= data set specified in the PROC MACONTROL statement. Each *process* specified in the EWMACHART statement must be a SAS variable in the DATA= data set. This variable provides measurements that must be grouped into subgroup samples indexed by the *subgroup-variable*. The *subgroup-variable*, which is specified in the EWMACHART statement, must also be a SAS variable in the DATA= data set. Each observation in a DATA= data set must contain a value for each *process* and a value for the *subgroup-variable*. If the *i*th subgroup contains n_i items, there should be n_i consecutive observations for which the value of the *subgroup-variable* is the index of the *i*th subgroup. For example, if each subgroup contains five items and there are 30 subgroup samples, the DATA= data set should contain 150 observations.

Other variables that can be read from a DATA= data set include

- _PHASE_ (if the READPHASES= option is specified)
- *block-variables*
- *symbol-variable*
- BY variables
- ID variables

By default, the MACONTROL procedure reads all the observations in a DATA= data set. However, if the data set includes the variable _PHASE_, you can read selected groups of observations (referred to as *phases*) with the READPHASES= option (for an example, see "Displaying Stratification in Phases" on page 2031.

For an example of a DATA= data set, see "Creating EWMA Charts from Raw Data" on page 786.

LIMITS= Data Set

You can read preestablished control limit parameters from a LIMITS= data set specified in the PROC MACONTROL statement. The LIMITS= data set used by the MACONTROL procedure does not contain the actual control limits, but rather it contains the parameters required to compute the limits. For example, the following statements read parameters from the data set Parms:

```
proc macontrol data=Parts limits=Parms;
   ewmachart Gap*Day;
run;
```

The LIMITS= data set can be an OUTLIMITS= data set that was created in a previous run of the MACONTROL procedure. Such data sets always contain the variables required for a LIMITS= data set; see the section "OUTLIMITS= Data Set" on page 815. The LIMITS= data set can also be created directly using a DATA step.

When you create a LIMITS= data set, you must provide the variable _WEIGHT_, which specifies the weight parameter used to compute the EWMAs. In addition, note the following:

- The variables _VAR_ and _SUBGRP_ are required. These must be character variables of length 8.
- The variable _INDEX_ is required if you specify the READINDEX= option. This must be a character variable whose length is no greater than 48.
- The variables _LIMITN_, _SIGMAS_ (or _ALPHA_), and _TYPE_ are optional, but they are recommended to maintain a complete set of control limit information. The variable _TYPE_ must be a character variable of length 8. Valid values are 'ESTIMATE', 'STANDARD', 'STDMEAN', and 'STDSIGMA'.
- BY variables are required if specified with a BY statement.

Some advantages of working with a LIMITS= data set are that

- it facilitates reusing a permanently saved set of parameters
- a distinct set of parameters can be read for each *process* specified in the EWMACHART statement
- it facilitates keeping track of multiple sets of parameters that accumulate for the same *process* as the process evolves over time

For an example, see the section "Reading Preestablished Control Limit Parameters" on page 795.

HISTORY= Data Set

You can read subgroup summary statistics from a HISTORY= data set specified in the PROC MACONTROL statement. This enables you to reuse OUTHISTORY= data sets that have been created in previous runs of the MACONTROL, SHEWHART, or CUSUM procedures or to read output data sets created with SAS summarization procedures such as PROC MEANS.

A HISTORY= data set used with the EWMACHART statement must contain the following:

- the *subgroup-variable*
- a subgroup mean variable for each *process*
- a subgroup sample size variable for each *process*
- a subgroup standard deviation variable for each *process*

The names of the subgroup mean, subgroup standard deviation, and subgroup sample size variables must be the *process* name concatenated with the suffix characters *X*, *S*, and *N*, respectively.

For example, consider the following statements:

```
proc macontrol history=Cliphist;
   ewmachart (Gap Diameter)*Day  / weight=0.2;
run;
```

The data set Cliphist must include the variables Day, GapX, GapS, GapN, DiameterX, DiameterS, and DiameterN.

Although a subgroup EWMA variable (named by the *process* name suffixed with *E*) is saved in an OUTHISTORY= data set, it is not required in a HISTORY= data set, because the subgroup mean variable is sufficient to compute the EWMAs.

Note that if you specify a *process* name that contains 32 characters, the names of the summary variables must be formed from the first 16 characters and the last 15 characters of the *process* name, suffixed with the appropriate character.

Other variables that can be read from a HISTORY= data set include

- _PHASE_ (if the READPHASES= option is specified)
- *block-variables*

- *symbol-variable*
- BY variables
- ID variables

By default, the MACONTROL procedure reads all the observations in a HISTORY= data set. However, if the HISTORY= data set includes the variable _PHASE_, you can read selected groups of observations (referred to as *phases*) by specifying the READPHASES= option (see "Displaying Stratification in Phases" on page 2031 for an example).

For an example of a HISTORY= data set, see "Creating EWMA Charts from Subgroup Summary Data" on page 789.

TABLE= Data Set

You can read summary statistics and control limits from a TABLE= data set specified in the PROC MACONTROL statement. This enables you to reuse an OUTTABLE= data set created in a previous run of the MACONTROL procedure.

The following table lists the variables required in a TABLE= data set used with the EWMACHART statement:

Variable	Description
EWMA	exponentially weighted moving average
LCLE	lower control limit for EWMA
LIMITN	nominal sample size associated with the control limits
MEAN	process mean
subgroup-variable	values of the *subgroup-variable*
SUBN	subgroup sample size
SUBS	subgroup standard deviation
SUBX	subgroup mean
UCLE	upper control limit for EWMA
WEIGHT	weight (r) assigned to most recent subgroup mean in computation of EWMA

Other variables that can be read from a TABLE= data set include

- *block-variables*
- *symbol-variable*
- BY variables
- ID variables
- _PHASE_ (if the READPHASES= option is specified). This variable must be a character variable whose length is no greater than 48.
- _VAR_. This variable is required if more than one *process* is specified or if the data set contains information for more than one *process*. This variable must be a character variable of length 8.

For an example of a TABLE= data set, see "Saving Control Limit Parameters" on page 793.

Methods for Estimating the Standard Deviation

When control limits are computed from the input data, four methods are available for estimating the process standard deviation σ. Three methods (referred to as the default, MVLUE, and RMSDF) are available with subgrouped data. A fourth method is used if the data are individual measurements (see “Default Method for Individual Measurements” on page 823).

Default Method for Subgroup Samples

This method is the default for EWMA charts using subgrouped data. The default estimate of σ is

$$\hat{\sigma} = \frac{s_1/c_4(n_1) + \ldots + s_N/c_4(n_N)}{N}$$

where N is the number of subgroups for which $n_i \geq 2$, s_i is the sample standard deviation of the ith subgroup

$$s_i = \sqrt{\frac{1}{n_i - 1} \sum_{j=1}^{n_i} (x_{ij} - \bar{X}_i)^2}$$

and

$$c_4(n_i) = \frac{\Gamma(n_i/2)\sqrt{2/(n_i - 1)}}{\Gamma((n_i - 1)/2)}$$

Here $\Gamma(\cdot)$ denotes the gamma function, and $\bar{X}_i$ denotes the ith subgroup mean. A subgroup standard deviation s_i is included in the calculation only if $n_i \geq 2$. If the observations are normally distributed, then the expected value of s_i is $c_4(n_i)\sigma$. Thus, $\hat{\sigma}$ is the unweighted average of N unbiased estimates of σ. This method is described in the American Society for Testing and Materials (1976).

MVLUE Method for Subgroup Samples

If you specify SMETHOD=MVLUE, a minimum variance linear unbiased estimate (MVLUE) is computed for σ. Refer to Burr (1969, 1976) and Nelson (1989, 1994). The MVLUE is a weighted average of N unbiased estimates of σ of the form $s_i/c_4(n_i)$, and it is computed as

$$\hat{\sigma} = \frac{h_1 s_1/c_4(n_1) + \ldots + h_N s_N/c_4(n_N)}{h_1 + \ldots + h_N}$$

where

$$h_i = \frac{[c_4(n_i)]^2}{1 - [c_4(n_i)]^2}$$

A subgroup standard deviation s_i is included in the calculation only if $n_i \geq 2$, and N is the number of subgroups for which $n_i \geq 2$. The MVLUE assigns greater weight to estimates of σ from subgroups with larger sample sizes, and it is intended for situations where the subgroup sample sizes vary. If the subgroup sample sizes are constant, the MVLUE reduces to the default estimate.

RMSDF Method for Subgroup Samples

If you specify SMETHOD=RMSDF, a weighted root-mean-square estimate is computed for σ as follows:

$$\hat{\sigma} = \frac{\sqrt{(n_1 - 1)s_1^2 + \cdots + (n_N - 1)s_N^2}}{c_4(n)\sqrt{n_1 + \cdots + n_N - N}}$$

where $n = n_1 + \cdots + n_N - (N-1)$. The weights are the degrees of freedom $n_i - 1$. A subgroup standard deviation s_i is included in the calculation only if $n_i \geq 2$, and N is the number of subgroups for which $n_i \geq 2$.

If the unknown standard deviation σ is constant across subgroups, the root-mean-square estimate is more efficient than the minimum variance linear unbiased estimate. However, in process control applications it is generally not assumed that σ is constant, and if σ varies across subgroups, the root-mean-square estimate tends to be more inflated than the MVLUE.

Default Method for Individual Measurements

When each subgroup sample contains a single observation ($n_i \equiv 1$), the process standard deviation σ is estimated as

$$\hat{\sigma} = \sqrt{\frac{1}{2(N-1)} \sum_{i=1}^{N-1} (x_{i+1} - x_i)^2}$$

where N is the number of observations, and $x_1, x_2, \ldots, x_N$ are the individual measurements. This formula is given by Wetherill (1977), who states that the estimate of the variance is biased if the measurements are autocorrelated.

Axis Labels

You can specify axis labels by assigning labels to particular variables in the input data set, as summarized in the following table:

Axis	Input Data Set	Variable
Horizontal	all	*subgroup-variable*
Vertical	DATA=	*process*
Vertical	HISTORY=	subgroup mean variable
Vertical	TABLE=	_EWMA_

For example, the following sets of statements specify the label *EWMA of Clip Gaps* for the vertical axis and the label *Day* for the horizontal axis of the EWMA chart:

```
proc macontrol data=Clips1;
   ewmachart Gap*Day / weight=0.3;
   label Gap = 'EWMA of Clip Gaps';
   label Day = 'Day';
run;

proc macontrol history=Cliphist;
   ewmachart Gap*Day / weight=0.3;
   label Gapx = 'EWMA of Clip Gaps';
   label Day  = 'Day';
run;

proc macontrol table=Cliptab;
   ewmachart Gap*Day;
   label _EWMA_  = 'EWMA of Clip Gaps';
   label Day     = 'Day';
run;
```

In this example, the label assignments are in effect only for the duration of the procedure step, and they temporarily override any permanent labels associated with the variables.

Missing Values

An observation read from a DATA=, HISTORY=, or TABLE= data set is not analyzed if the value of the subgroup variable is missing. For a particular process variable, an observation read from a DATA= data set is not analyzed if the value of the process variable is missing. Missing values of process variables generally lead to unequal subgroup sample sizes. For a particular process variable, an observation read from a HISTORY= or TABLE= data set is not analyzed if the values of any of the corresponding summary variables are missing.

Examples: EWMACHART Statement

This section provides advanced examples of the EWMACHART statement.

Example 9.1: Specifying Standard Values for the Process Mean and Process Standard Deviation

NOTE: See *Specifying Standard Values for EWMA Chart* in the SAS/QC Sample Library.

By default, the EWMACHART statement estimates the process mean (μ) and standard deviation (σ) from the data. This is illustrated in the "Getting Started" section of this chapter. However, there are applications in which standard values (μ_0 and σ_0) are available based, for instance, on previous experience or extensive sampling. You can specify these values with the MU0= and SIGMA0= options.

For example, suppose it is known that the metal clip manufacturing process (introduced in "Creating EWMA Charts from Raw Data" on page 786) has a mean of 15 and standard deviation of 0.2. The following statements specify these standard values:

```
ods graphics on;
title 'Specifying Standard Process Mean and Standard Deviation';
proc macontrol data=Clips1;
   ewmachart Gap*Day /
      odstitle = title
      mu0      = 15
      sigma0   = 0.2
      weight   = 0.3
      xsymbol  = mu0
      markers;
run;
```

The XSYMBOL= option specifies the label for the central line. The resulting chart is shown in Output 9.1.1.

Output 9.1.1 Specifying Standard Values with MU0= and SIGMA0=



The central line and control limits are determined using μ_0 and σ_0 (see the equations in Table 9.3). Output 9.1.1 indicates that the process is out-of-control, since the moving averages for Day=17, Day=19, and Day=20 lie below the lower control limit.

You can also specify μ_0 and σ_0 with the variables _MEAN_ and _STDDEV_ in a LIMITS= data set, as illustrated by the following statements:

```
data Cliplim;
   length _var_ _subgrp_ _type_ $8;
   _var_     = 'Gap';
   _subgrp_  = 'Day';
   _type_    = 'STANDARD';
   _limitn_  = 5;
   _mean_    = 15;
   _stddev_  = 0.2;
   _weight_  = 0.3;

proc macontrol data=Clips1 limits=Cliplim;
   ewmachart Gap*Day /
      odstitle = title
      xsymbol  = mu0
      markers;
run;
```

The variable _WEIGHT_ is required, and its value provides the weight parameter used to compute the EWMAs. The variables _VAR_ and _SUBGRP_ are also required, and their values must match the *process* and *subgroup-variable*, respectively, specified in the EWMACHART statement. The bookkeeping variable _TYPE_ is not required, but it is recommended to indicate that the variables _MEAN_ and _STDDEV_ provide standard values rather than estimated values.

The resulting chart (not shown here) is identical to the one shown in Output 9.1.1.

Example 9.2: Displaying Limits Based on Asymptotic Values

NOTE: See *Displaying Limits Based on Asymptotic Values* in the SAS/QC Sample Library.

The upper (lower) control limits in Output 9.1.1 are monotonically increasing (decreasing). As the number of subgroups increases, the control limits approach the following asymptotic values:

$$\text{LCL} = \overline{\overline{X}} - k\hat{\sigma}\sqrt{r/n(2-r)}$$

$$\text{UCL} = \overline{\overline{X}} + k\hat{\sigma}\sqrt{r/n(2-r)}$$

These constant limits are displayed if you specify the ASYMPTOTIC option, as illustrated by the following statements:

```
ods graphics on;
title 'Constant Control Limits Based on Asymptotic Values';
proc macontrol data=Clips1;
   ewmachart Gap*Day /
      odstitle = title
      mu0      = 15
      sigma0   = 0.2
      weight   = 0.3
      xsymbol  = mu0
      asymptotic
      markers;
run;
```

The chart is shown in Output 9.2.1.

Output 9.2.1 Asymptotic Control Limits



Note that the same three points that were outside the exact limits (displayed in Output 9.1.1) fall outside the asymptotic limits. The exact limits quickly approach the asymptotic values, so only the first few subgroups have appreciably different limits.

Example 9.3: Working with Unequal Subgroup Sample Sizes

NOTE: See *EWMA Chart with Unequal Subgroup Sample Sizes* in the SAS/QC Sample Library.

This example contains measurements from the metal clip manufacturing process (introduced in "Creating EWMA Charts from Raw Data" on page 786). The following statements create a SAS data set named Clips4, which contains additional clip gap measurements taken on a daily basis:

```
data Clips4;
   input Day @;
   length Dayc $2.;
   informat Day ddmmyy8.;
   format   Day date5.;
   Dayc=put(Day,date5.);
   Dayc=substr(Dayc,1,2);
   do i=1 to 5;
      input Gap @;
      output;
   end;
   drop i;
   label Dayc='April';
   datalines;
 1/4/86  14.93  14.65  14.87  15.11  15.18
 2/4/86  15.06  14.95  14.91  15.14  15.41
 3/4/86  14.90  14.90  14.96  15.26  15.18
 4/4/86  15.25  14.57  15.33  15.38  14.89
 7/4/86  14.68  14.63  14.72  15.32  14.86
 8/4/86  14.48  14.88  14.98  14.74  15.48
 9/4/86  14.99  15.16  15.02  15.53  14.66
10/4/86  14.88  15.44  15.04  15.10  14.89
11/4/86  15.14  15.33  14.75  15.23  14.64
14/4/86  15.46  15.30  14.92  14.58  14.68
15/4/86  15.23  14.63    .      .      .
16/4/86  15.13  15.25    .      .      .
17/4/86  15.06  15.25  15.28  15.30  15.34
18/4/86  15.22  14.77  15.12  14.82  15.29
21/4/86  14.95  14.96  14.65  14.87  14.77
22/4/86  15.01  15.11  15.11  14.79  14.88
23/4/86  14.97  15.50  14.93  15.13  15.25
24/4/86  15.23  15.21  15.31  15.07  14.97
25/4/86  15.08  14.75  14.93  15.34  14.98
28/4/86  15.07  14.86  15.42  15.47  15.24
29/4/86  15.27  15.20  14.85  15.62  14.67
30/4/86  14.97  14.73  15.09  14.98  14.46
;
```

Note that only two gap measurements were recorded on April 15 and April 16.

A partial listing of Clips4 is shown in Output 9.3.1. This data set contains three variables: Day is a numeric variable that contains the date (month, day, and year) that the measurement is taken, Dayc is a character variable that contains the day the measurement is taken, and Gap is a numeric variable that contains the measurement.

Output 9.3.1 The Data Set Clips4

The Data Set Clips4

Day	Dayc	Gap
01APR	01	14.93
01APR	01	14.65
01APR	01	14.87
01APR	01	15.11
01APR	01	15.18
02APR	02	15.06
02APR	02	14.95
02APR	02	14.91
02APR	02	15.14
02APR	02	15.41
03APR	03	14.90
03APR	03	14.90
03APR	03	14.96
03APR	03	15.26
03APR	03	15.18

The following statements request an EWMA chart, shown in Output 9.3.2, for these gap measurements:

```
ods graphics on;
title 'EWMA Chart for Gap Measurements';
proc macontrol data=Clips4;
   ewmachart Gap*Dayc / odstitle = title
                        weight   = 0.3
                        markers;
run;
```

The character variable Dayc (rather than the numeric variable Day) is specified as the *subgroup-variable* in the preceding EWMACHART statement. If Day were the *subgroup-variable*, each day during April would appear on the horizontal axis, including the weekend days of April 5 and April 6 for which no measurements were taken. To avoid this problem, the *subgroup-variable* Dayc is created from Day using the PUT and SUBSTR function. Since Dayc is a character *subgroup-variable*, a discrete axis is used for the horizontal axis, and as a result, April 5 and April 6 do not appear on the horizontal axis in Output 9.3.2. A LABEL statement is used to specify the label *April* for the horizontal axis, indicating the month that these measurements were taken.

Output 9.3.2 EWMA Chart with Varying Sample Sizes



Note that the control limits vary with the subgroup sample size. The sample size legend in the lower left corner displays the minimum and maximum subgroup sample sizes.

The EWMACHART statement provides various options for working with unequal subgroup sample sizes. For example, you can use the LIMITN= option to specify a fixed (nominal) sample size for computing control limits, as illustrated by the following statements:

```
title 'EWMA Chart for Gap Measurements';
proc macontrol data=Clips4;
   ewmachart Gap*Dayc / odstitle = title
                        weight   = 0.3
                        limitn   = 5
                        markers;
run;
```

The resulting chart is shown in Output 9.3.3.

Output 9.3.3 Control Limits Based on Fixed Sample Size



Note that the only points displayed are those corresponding to subgroups whose sample size matches the nominal sample size of five. Therefore, points are not displayed for April 15 and April 16. To plot points for all subgroups (regardless of subgroup sample size), you can specify the ALLN option, as follows:

```
title 'EWMA Chart for Gap Measurements';
proc macontrol data=Clips4;
   ewmachart Gap*Dayc/ odstitle = title
                       weight   = 0.3
                       limitn   = 5
                       alln
                       nmarkers;
run;
```

The chart is shown in Output 9.3.4. The NMARKERS option requests special symbols to identify points for which the subgroup sample size differs from the nominal sample size.

Output 9.3.4 Displaying All Subgroups Regardless of Sample Size



You can use the SMETHOD= option to determine how the process standard deviation σ is to be estimated when the subgroup sample sizes vary. The default method computes $\hat{\sigma}$ as an unweighted average of subgroup estimates of σ. Specifying SMETHOD=MVLUE requests a minimum variance linear unbiased estimate (MVLUE), which assigns greater weight to estimates of σ from subgroups with larger sample sizes. Specifying SMETHOD=RMSDF requests a weighted root-mean-square estimate. If the unknown standard deviation σ is constant across subgroups, the root-mean-square estimate is more efficient than the MVLUE. For more information, see “Methods for Estimating the Standard Deviation” on page 822.

The following statements apply all three methods:

```
proc macontrol data=Clips4;
   ewmachart Gap*Dayc / outlimits = Cliplim1
                        outindex  = 'Default'
                        weight    = 0.3
                        nochart;
   ewmachart Gap*Dayc / smethod   = mvlue
                        outlimits = Cliplim2
                        outindex  = 'MVLUE'
                        weight    = 0.3
                        nochart;
```

```
   ewmachart Gap*Dayc / smethod   = rmsdf
                        outlimits = Cliplim3
                        outindex  = 'RMSDF'
                        weight    = 0.3
                        nochart;
run;

data Climits;
   set Cliplim1 Cliplim2 Cliplim3;
run;
```

The data set Climits is listed in Output 9.3.5.

Output 9.3.5 Listing of the Data Set Climits

Estimating the Process Standard Deviation

VAR	_SUBGRP_	_INDEX_	_TYPE_	_LIMITN_	_ALPHA_	_SIGMAS_	_MEAN_	_STDDEV_	_WEIGHT_
Gap	Dayc	Default	ESTIMATE	V	.002699796	3	15.0354	0.26503	0.3
Gap	Dayc	MVLUE	ESTIMATE	V	.002699796	3	15.0354	0.26096	0.3
Gap	Dayc	RMSDF	ESTIMATE	V	.002699796	3	15.0354	0.25959	0.3

Note that the estimate of the process standard deviation (stored in the variable _STDDEV_) is slightly different depending on the estimation method. The variable _LIMITN_ is assigned the special missing value *V* in the OUTLIMITS= data set, indicating that the subgroup sample sizes vary.

Example 9.4: Displaying Individual Measurements on an EWMA Chart

NOTE: See *EWMA Chart with Individual Measurements* in the SAS/QC Sample Library.

In the manufacture of automotive tires, the diameter of the steel belts inside the tire is measured. The following data set contains these measurements for 30 tires:

```
data Tires;
   input Sample Diameter @@;
   datalines;
 1  24.05    2  23.99    3  23.95
 4  23.93    5  23.97    6  24.02
 7  24.06    8  24.10    9  23.98
10  24.03   11  23.91   12  24.06
13  24.06   14  23.96   15  23.98
16  24.06   17  24.01   18  24.00
19  23.93   20  23.92   21  24.09
22  24.11   23  24.05   24  23.98
25  23.98   26  24.06   27  24.02
28  24.06   29  23.97   30  23.96
;
```

The following statements use the IRCHART statement in the SHEWHART procedure (see "IRCHART Statement: SHEWHART Procedure" on page 1473) to create a data set containing the control limits for individual measurements and moving range charts for Diameter:

```
proc shewhart data=Tires;
   irchart Diameter*Sample / nochart outlimits=Tlimits;
run;
```

A listing of the data set Tlimits is shown in Output 9.4.1.

Output 9.4.1 Listing of the Data Set Tlimits

Control Limits for Diameter Measurements

VAR	_SUBGRP_	_TYPE_	_LIMITN_	_ALPHA_	_SIGMAS_	_LCLI_	_MEAN_	_UCLI_
Diameter	Sample	ESTIMATE	2	.002699796	3	23.8571	24.0083	24.1596

LCLR	_R_	_UCLR_	_STDDEV_
0	0.056897	0.18585	0.050423

The upper and lower control limits for the diameter measurements are 24.1596 and 23.8571, respectively.

In this example, reference lines will be used to display the control limits for the individual measurements on the EWMA chart. The following DATA step reads these control limits from Tlimits and creates a data set named Vrefdata, which contains the reference line information:

```
data Vrefdata;
   set Tlimits;
   length _reflab_ $16.;
   keep _ref_ _reflab_;
   _ref_ = _lcli_; _reflab_= 'LCL for X'; output;
   _ref_ = _ucli_; _reflab_= 'UCL for X'; output;
run;
```

A listing of the data set Vrefdata is shown in Output 9.4.2.

Output 9.4.2 Listing of the Data Set Vrefdata

Reference Line Information

reflab	_ref_
LCL for X	23.8571
UCL for X	24.1596

The following statements request an EWMA chart for these measurements:

```
ods graphics on;
title 'EWMA Chart for Steel Belt Diameters';
proc macontrol data=Tires;
   ewmachart Diameter*Sample / weight     = 0.3
                               meansymbol = square
                               lcllabel   = 'LCL for EWMA'
                               ucllabel   = 'UCL for EWMA'
                               vref       = Vrefdata
                               odstitle   = title
                               vreflabpos = 3
                               markers;
run;
```

The MEANSYMBOL= option displays the individual measurements on the EWMA chart. By default, these values are not displayed. For traditional graphics, the MEANSYMBOL= option specifies the symbol used to plot the individual measurements. For ODS Graphics, specifying a MEANSYMBOL= value causes the subgroup means to be plotted, but the symbol used is determined by the ODS style in effect. The VREF= option reads the reference line information from Vrefdata. The resulting chart is shown in Output 9.4.3.

Output 9.4.3 indicates that the process is in control. None of the diameter measurements (indicated by squares) exceed their control limits, and none of the EWMAs exceed their limits.

Output 9.4.3 Displaying Individual Measurements on EWMA Chart



Example 9.5: Computing Average Run Lengths

NOTE: See *Computing Average Run Lengths for EWMA Chart* in the SAS/QC Sample Library.

The EWMAARL DATA step function computes the average run length for an exponentially weighted moving average (EWMA) scheme (refer to Crowder 1987a,b for details). You can use this function to design a scheme by first calculating average run lengths for a range of values for the weight and then choosing the weight that yields a desired average run length.

The following statements compute the average run lengths for shifts between 0.5 and 2 and weights between 0.25 and 1. The data set ARLs is displayed in Output 9.5.1.

```
data ARLs;
   do shift=.5 to 2 by .5;
      do Weight=.25 to 1 by .25;
      arl=ewmaarl(shift,Weight,3.0);
      output;
      end;
   end;
run;
```

Output 9.5.1 Listing of the Data Set ARLs

Average Run Lengths for Various Shifts and Weights

shift=0.5

Weight	arl
0.25	48.453
0.50	75.354
0.75	110.950
1.00	155.224

shift=1

Weight	arl
0.25	11.1543
0.50	15.7378
0.75	25.6391
1.00	43.8947

shift=1.5

Weight	arl
0.25	5.4697
0.50	6.1111
0.75	8.7201
1.00	14.9677

shift=2

Weight	arl
0.25	3.61677
0.50	3.46850
0.75	4.15346
1.00	6.30296

Note that when the weight is 1.0, the EWMAARL function returns the average run length for a Shewhart chart for means. For more details, see "EWMAARL Function" on page 2180.

In addition to using the EWMAARL function to design a EWMA scheme with desired average run length properties, you can use it to evaluate an existing scheme. For example, suppose you have an EWMA chart with 3σ control limits using a weight parameter of 0.3. The following DATA step computes the average run lengths for various shifts using this scheme:

```
data ARLinfo;
   do shift=0 to 2 by .25;
      arl = ewmaarl(shift,0.3,3.0);
      output;
   end;
run;
```

The data set ARLinfo is displayed in Output 9.5.2.

Output 9.5.2 Listing of the Data Set ARLinfo

Average Run Lengths for EWMA Scheme (k=3 and r=0.3)

shift	arl
0.00	465.553
0.25	178.741
0.50	53.160
0.75	21.826
1.00	11.699
1.25	7.525
1.50	5.447
1.75	4.258
2.00	3.506

MACHART Statement: MACONTROL Procedure

Overview: MACHART Statement

The MACHART statement creates a uniformly weighted moving average control chart (commonly referred to as a moving average control chart), which is used to decide whether a process is in a state of statistical control and to detect shifts in the process average.

You can use options in the MACHART statement to

- specify the span of the moving averages (the number of terms in the moving average)
- compute control limits from the data based on a multiple of the standard error of the plotted moving averages or as probability limits
- tabulate the moving averages, subgroup sample sizes, subgroup means, subgroup standard deviations, control limits, and other information
- save control limit parameters in an output data set
- save the moving averages, subgroup sample sizes, subgroup means, and subgroup standard deviations in an output data set

- read control limit parameters from an input data set
- specify one of several methods for estimating the process standard deviation
- specify a known (standard) process mean and standard deviation for computing control limits
- display a secondary chart that plots a time trend that has been removed from the data
- add block legends and symbol markers to reveal stratification in process data
- superimpose stars at points to represent related multivariate factors
- clip extreme points to make the chart more readable
- display vertical and horizontal reference lines
- control axis values and labels
- control layout and appearance of the chart

You have three alternatives for producing moving average control charts with the MACHART statement:

- ODS Graphics output is produced if ODS Graphics is enabled, for example by specifying the ODS GRAPHICS ON statement prior to the PROC statement.
- Otherwise, traditional graphics are produced by default if SAS/GRAPH® is licensed.
- Legacy line printer charts are produced when you specify the LINEPRINTER option in the PROC statement.

See Chapter 3, "SAS/QC Graphics," for more information about producing these different kinds of graphs.

Getting Started: MACHART Statement

This section introduces the MACHART statement with simple examples that illustrate the most commonly used options. Complete syntax for the MACHART statement is presented in the section "Syntax: MACHART Statement" on page 850, and advanced examples are given in the section "Examples: MACHART Statement" on page 877.

Creating Moving Average Charts from Raw Data

NOTE: See *Uniformly Weighted Moving Average Chart* in the SAS/QC Sample Library.

In the manufacture of a metal clip, the gap between the ends of the clip is a critical dimension. To monitor the process for a change in the average gap, subgroup samples of five clips are selected daily. The data are analyzed with a uniformly weighted moving average chart. The gaps recorded during the first twenty days are saved in a SAS data set named Clips1.

```
data Clips1;
   input Day @ ;
   do i=1 to 5;
      input Gap @ ;
      output;
   end;
   drop i;
   datalines;
 1  14.76  14.82  14.88  14.83  15.23
 2  14.95  14.91  15.09  14.99  15.13
 3  14.50  15.05  15.09  14.72  14.97
 4  14.91  14.87  15.46  15.01  14.99
 5  14.73  15.36  14.87  14.91  15.25
 6  15.09  15.19  15.07  15.30  14.98
 7  15.34  15.39  14.82  15.32  15.23
 8  14.80  14.94  15.15  14.69  14.93
 9  14.67  15.08  14.88  15.14  14.78
10  15.27  14.61  15.00  14.84  14.94
11  15.34  14.84  15.32  14.81  15.17
12  14.84  15.00  15.13  14.68  14.91
13  15.40  15.03  15.05  15.03  15.18
14  14.50  14.77  15.22  14.70  14.80
15  14.81  15.01  14.65  15.13  15.12
16  14.82  15.01  14.82  14.83  15.00
17  14.89  14.90  14.60  14.40  14.88
18  14.90  15.29  15.14  15.20  14.70
19  14.77  14.60  14.45  14.78  14.91
20  14.80  14.58  14.69  15.02  14.85
;
```

The following statements produce the listing of the data set Clips1 shown in Figure 9.10:

```
title 'The Data Set Clips1';
proc print data=Clips1(obs=15) noobs;
run;
```

Figure 9.10 Partial Listing of the Data Set Clips1

The Data Set Clips1

Day	Gap
1	14.76
1	14.82
1	14.88
1	14.83
1	15.23
2	14.95
2	14.91
2	15.09
2	14.99
2	15.13
3	14.50
3	15.05
3	15.09
3	14.72
3	14.97

The data set Clips1 is said to be in “strung-out” form, since each observation contains the day and gap measurement of a single clip. The first five observations contain the gap measurements for the first day, the second five observations contain the gap measurements for the second day, and so on. Because the variable Day classifies the observations into rational subgroups, it is referred to as the *subgroup-variable*. The variable Gap contains the gap measurements and is referred to as the *process variable* (or *process* for short).

The within-subgroup variability of the gap measurements is known to be stable. You can use a uniformly weighted moving average chart to determine whether the mean level is in control. The following statements create the chart shown in Figure 9.11:

```
ods graphics off;
title 'Moving Average Chart for Gap Measurements';
proc macontrol data=Clips1;
   machart Gap*Day / span=3;
run;
```

This example illustrates the basic form of the MACHART statement. After the keyword MACHART, you specify the *process* to analyze (in this case, Gap) followed by an asterisk and the *subgroup-variable* (Day). The SPAN= option specifies the number of terms to include in the moving average. Options such as SPAN= are specified after the slash (/) in the MACHART statement. A complete list of options is presented in the section “Syntax: MACHART Statement” on page 850. You must provide the span of the moving average. As an alternative to specifying the SPAN= option, you can read the span from an input data set; see “Reading Preestablished Control Limit Parameters” on page 848.

The input data set is specified with the DATA= option in the PROC MACONTROL statement.

Figure 9.11 Uniformly Weighted Moving Average Chart for Gap Data



Each point on the chart represents the uniformly weighted moving average for a particular day. The moving average A_1 plotted at Day=1 is simply the subgroup mean for Day=1. The moving average A_2 plotted at Day=2 is the average of the subgroup means for Day=1 and Day=2. The moving average A_3 plotted at Day=3 is the average of the subgroup means for Day=1, Day=2, and Day=3.

$$
\begin{aligned}
A_1 &= \frac{14.76 + 14.82 + 14.88 + 14.83 + 15.23}{5} = 14.904 \text{ mm} \\
A_2 &= \frac{14.904 + 15.014}{2} = 14.959 \text{ mm} \\
A_3 &= \frac{14.904 + 15.014 + 14.866}{3} = 14.928 \text{ mm}
\end{aligned}
$$

For succeeding days, the moving average is similarly calculated as the average of the present and the two previous subgroup means (since a span of three is specified with the SPAN= option).

Note that the moving average for the seventh day lies above the upper control limit, signaling an out-of-control process.

By default, the control limits shown are 3σ limits estimated from the data; the formulas for the limits are given in Table 9.9.

For computational details, see "Constructing Uniformly Weighted Moving Average Charts" on page 862. For more details on reading from a DATA= data set, see "DATA= Data Set" on page 871.

Creating Moving Average Charts from Subgroup Summary Data

NOTE: See *Uniformly Weighted Moving Average Chart* in the SAS/QC Sample Library.

The previous example illustrates how you can create moving average charts using raw data (process measurements). However, in many applications the data are provided as subgroup summary statistics. This example illustrates how you can use the MACHART statement with data of this type. The following data set (Clipsum) provides the data from the preceding example in summarized form:

```
data Clipsum;
   input Day GapX GapS;
   GapN=5;
   datalines;
 1  14.904  0.18716
 2  15.014  0.09317
 3  14.866  0.25006
 4  15.048  0.23732
 5  15.024  0.26792
 6  15.126  0.12260
 7  15.220  0.23098
 8  14.902  0.17254
 9  14.910  0.19824
10  14.932  0.24035
11  15.096  0.25618
12  14.912  0.16903
13  15.138  0.15928
14  14.798  0.26329
15  14.944  0.20876
16  14.896  0.09965
17  14.734  0.22512
18  15.046  0.24141
19  14.702  0.17880
20  14.788  0.16634
;
```

A partial listing of Clipsum is shown in Figure 9.12. There is exactly one observation for each subgroup (note that the subgroups are still indexed by Day). The variable GapX contains the subgroup means, the variable GapS contains the subgroup standard deviations, and the variable GapN contains the subgroup sample sizes (these are all five).

Figure 9.12 The Summary Data Set Clipsum

The Data Set Clipsum

Day	GapX	GapS	GapN
1	14.904	0.18716	5
2	15.014	0.09317	5
3	14.866	0.25006	5
4	15.048	0.23732	5
5	15.024	0.26792	5

You can read this data set by specifying it as a HISTORY= data set in the PROC MACONTROL statement, as follows:

```
options nogstyle;
symbol color=salmon h=0.8;
title 'Moving Average Chart for Gap Measurements';
proc macontrol history=Clipsum;
   machart Gap*Day / span     = 3
                     cframe   = steel
                     cinfill  = vpab
                     cconnect = yellow
                     coutfill = salmon;
run;
options gstyle;
```

The NOGSTYLE system option causes ODS styles not to affect traditional graphics. Instead, the SYMBOL statement and MACHART statement options control the appearance of the graph. The GSTYLE system option restores the use of ODS styles for traditional graphics produced subsequently. The resulting moving average chart is shown in Figure 9.13.

Figure 9.13 Uniformly Weighted Moving Average Chart from Summary Data



Note that Gap is *not* the name of a SAS variable in the data set but is, instead, the common prefix for the names of the three SAS variables GapX, GapS, and GapN. The suffix characters *X*, *S*, and *N* indicate *mean*, *standard deviation*, and *sample size*, respectively. Thus, you can specify three subgroup summary variables in a HISTORY= data set with a single name (Gap), which is referred to as the *process*. The variables GapX, GapS, and GapN are all required. The name Day specified after the asterisk is the name of the *subgroup-variable*.

In general, a HISTORY= input data set used with the MACHART statement must contain the following variables:

- subgroup variable
- subgroup mean variable
- subgroup standard deviation variable
- subgroup sample size variable

Furthermore, the names of subgroup mean, standard deviation, and sample size variables must begin with the *process* name specified in the MACHART statement and end with the special suffix characters *X*, *S*, and *N*, respectively. If the names do not follow this convention, you can use the RENAME option in the PROC MACONTROL statement to rename the variables for the duration of the MACONTROL procedure step (see “” on page 1843 for an example).

In summary, the interpretation of *process* depends on the input data set.

- If raw data are read using the DATA= option (as in the previous example), *process* is the name of the SAS variable containing the process measurements.
- If summary data are read using the HISTORY= option (as in this example), *process* is the common prefix for the names of the variables containing the summary statistics.

For more information, see “HISTORY= Data Set” on page 872.

Saving Summary Statistics

NOTE: See *Uniformly Weighted Moving Average Chart* in the SAS/QC Sample Library.

In this example, the MACHART statement is used to create a summary data set that can be read later by the MACONTROL procedure (as in the preceding example). The following statements read measurements from the data set Clips1 and create a summary data set named Cliphist:

```
title 'Summary Data Set for Gap Measurements';
proc macontrol data=Clips1;
   machart Gap*Day / span       = 3
                     outhistory = Cliphist
                     nochart;
run;
```

The OUTHISTORY= option names the output data set, and the NOCHART option suppresses the display of the chart, which would be identical to the chart in Figure 9.11.

Figure 9.14 contains a partial listing of Cliphist.

Figure 9.14 The Summary Data Set Cliphist

Summary Data Set for Gap Measurements

Day	GapX	GapS	GapA	GapN
1	14.904	0.18716	14.9040	5
2	15.014	0.09317	14.9590	5
3	14.866	0.25006	14.9280	5
4	15.048	0.23732	14.9760	5
5	15.024	0.26792	14.9793	5

There are five variables in the data set Cliphist.

- Day contains the subgroup index.
- GapX contains the subgroup means.
- GapS contains the subgroup standard deviations.
- GapA contains the subgroup moving averages.
- GapN contains the subgroup sample sizes.

Note that the summary statistic variables are named by adding the suffix characters *X*, *S*, *A*, and *N* to the *process* Gap specified in the MACHART statement. In other words, the variable naming convention for OUTHISTORY= data sets is the same as that for HISTORY= data sets.

For more information, see "OUTHISTORY= Data Set" on page 868.

Saving Control Limit Parameters

NOTE: See *Uniformly Weighted Moving Average Chart* in the SAS/QC Sample Library.

You can save the control limit parameters used for a moving average chart in a SAS data set; this enables you to use these parameters with future data (see "Reading Preestablished Control Limit Parameters" on page 848) or modify the parameters with a DATA step program.

The following statements read measurements from the data set Clips1 (see "Creating Moving Average Charts from Raw Data" on page 839) and save the control limit parameters in a data set named Cliplim:

```
title 'Control Limit Parameters';
proc macontrol data=Clips1;
   machart Gap*Day / span      = 3
                     outlimits = Cliplim
                     nochart;
run;
```

The OUTLIMITS= option names the data set containing the control limits, and the NOCHART option suppresses the display of the chart. The data set Cliplim is listed in Figure 9.15.

Figure 9.15 The Data Set Cliplim Containing Control Limit Information

Control Limit Parameters

VAR	_SUBGRP_	_TYPE_	_LIMITN_	_ALPHA_	_SIGMAS_	_MEAN_	_STDDEV_	_SPAN_
Gap	Day	ESTIMATE	5	.002699796	3	14.95	0.21108	3

Note that the data set Cliplim does not contain the actual control limits, but rather the parameters required to compute the limits.

The data set contains one observation with the parameters for *process* Gap. The variable _SPAN_ contains the number of terms used to calculate the moving average. The value of _MEAN_ is an estimate of the process mean, and the value of _STDDEV_ is an estimate of the process standard deviation σ. The value of _LIMITN_ is the nominal sample size associated with the control limits, and the value of _SIGMAS_ is the multiple of σ associated with the control limits. The variables _VAR_ and _SUBGRP_ are bookkeeping variables that save the *process* and *subgroup-variable*. The variable _TYPE_ is a bookkeeping variable that indicates that the values of _MEAN_ and _STDDEV_ are estimates rather than standard values. For more information, see "OUTLIMITS= Data Set" on page 867.

You can create an output data set containing the control limits and summary statistics with the OUTTABLE= option, as illustrated by the following statements:

```
title 'Summary Statistics and Control Limits';
proc macontrol data=Clips1;
   machart Gap*Day / span     = 3
                     outtable = Cliptab
                     nochart;
run;
```

The data set Cliptab is listed in Figure 9.16.

This data set contains one observation for each subgroup sample. The variable _UWMA_ contains the uniformly weighted moving average. The variables _SUBX_, _SUBS_, and _SUBN_ contain the subgroup means, subgroup standard deviations, and subgroup sample sizes, respectively. The variables _LCLA_ and _UCLA_ contain the lower and upper control limits, and the variable _MEAN_ contains the central line. The variables _VAR_ and Day contain the *process* name and values of the *subgroup-variable*, respectively. For more information, see "OUTTABLE= Data Set" on page 869.

Figure 9.16 The OUTTABLE= Data Set Cliptab

Summary Statistics and Control Limits

VAR	Day	_SIGMAS_	_LIMITN_	_SPAN_	_SUBN_	_SUBX_	_SUBS_	_LCLA_	_UWMA_
Gap	1	3	5	3	5	14.904	0.18716	14.6668	14.9040
Gap	2	3	5	3	5	15.014	0.09317	14.7498	14.9590
Gap	3	3	5	3	5	14.866	0.25006	14.7865	14.9280
Gap	4	3	5	3	5	15.048	0.23732	14.7865	14.9760
Gap	5	3	5	3	5	15.024	0.26792	14.7865	14.9793
Gap	6	3	5	3	5	15.126	0.12260	14.7865	15.0660
Gap	7	3	5	3	5	15.220	0.23098	14.7865	15.1233
Gap	8	3	5	3	5	14.902	0.17254	14.7865	15.0827
Gap	9	3	5	3	5	14.910	0.19824	14.7865	15.0107
Gap	10	3	5	3	5	14.932	0.24035	14.7865	14.9147
Gap	11	3	5	3	5	15.096	0.25618	14.7865	14.9793
Gap	12	3	5	3	5	14.912	0.16903	14.7865	14.9800
Gap	13	3	5	3	5	15.138	0.15928	14.7865	15.0487
Gap	14	3	5	3	5	14.798	0.26329	14.7865	14.9493
Gap	15	3	5	3	5	14.944	0.20876	14.7865	14.9600
Gap	16	3	5	3	5	14.896	0.09965	14.7865	14.8793
Gap	17	3	5	3	5	14.734	0.22512	14.7865	14.8580
Gap	18	3	5	3	5	15.046	0.24141	14.7865	14.8920
Gap	19	3	5	3	5	14.702	0.17880	14.7865	14.8273
Gap	20	3	5	3	5	14.788	0.16634	14.7865	14.8453

MEAN	_UCLA_	_STDDEV_	_EXLIM_
14.95	15.2332	0.21108	
14.95	15.1502	0.21108	
14.95	15.1135	0.21108	
14.95	15.1135	0.21108	
14.95	15.1135	0.21108	
14.95	15.1135	0.21108	
14.95	15.1135	0.21108	UPPER
14.95	15.1135	0.21108	
14.95	15.1135	0.21108	
14.95	15.1135	0.21108	
14.95	15.1135	0.21108	
14.95	15.1135	0.21108	
14.95	15.1135	0.21108	
14.95	15.1135	0.21108	
14.95	15.1135	0.21108	
14.95	15.1135	0.21108	
14.95	15.1135	0.21108	
14.95	15.1135	0.21108	
14.95	15.1135	0.21108	
14.95	15.1135	0.21108	

An OUTTABLE= data set can be read later as a TABLE= data set. For example, the following statements read Cliptab and display a moving average chart (not shown here) identical to Figure 9.11:

```
title 'Moving Average Chart for Gap Measurements';
proc macontrol table=Cliptab;
   machart Gap*Day;
run;
```

For more information, see “TABLE= Data Set” on page 873.

Reading Preestablished Control Limit Parameters

NOTE: See *Uniformly Weighted Moving Average Chart* in the SAS/QC Sample Library.

In the previous example, the OUTLIMITS= data set saved the control limit parameters in the data set Cliplim. This example shows how to apply these parameters to new data provided in the following data set:

```
data Clips1a;
   label Gap='Gap Measurement (mm)';
   input Day @;
   do i=1 to 5;
      input Gap @;
      output;
   end;
   drop i;
   datalines;
21  14.86 15.01 14.67 14.67 15.07
22  14.93 14.53 15.07 15.10 14.98
23  15.27 14.90 15.12 15.10 14.80
24  15.02 15.21 14.93 15.11 15.20
25  14.90 14.81 15.26 14.57 14.94
26  14.78 15.29 15.13 14.62 14.54
27  14.78 15.15 14.61 14.92 15.07
28  14.92 15.31 14.82 14.74 15.26
29  15.11 15.04 14.61 15.09 14.68
30  15.00 15.04 14.36 15.20 14.65
31  14.99 14.76 15.18 15.04 14.82
32  14.90 14.78 15.19 15.06 15.06
33  14.95 15.10 14.86 15.27 15.22
34  15.03 14.71 14.75 14.99 15.02
35  15.38 14.94 14.68 14.77 14.83
36  14.95 15.43 14.87 14.90 15.34
37  15.18 14.94 15.32 14.74 15.29
38  14.91 15.15 15.06 14.78 15.42
39  15.34 15.34 15.41 15.36 14.96
40  15.12 14.75 15.05 14.70 14.74
;
```

The following statements create a moving average chart for the data in Clips1a using the control limit parameters in Cliplim:

```
ods graphics on;
title 'Moving Average Chart for Second Set of Gap Measurements';
proc macontrol data=Clips1a limits=Cliplim;
   machart Gap*Day / odstitle=title;
run;
```

The ODS GRAPHICS ON statement specified before the PROC MACONTROL statement enables ODS Graphics, so the moving average chart is created using ODS Graphics instead of traditional graphics. The chart is shown in Figure 9.17.

Figure 9.17 Using Control Limit Parameters from a LIMITS= Data Set



The LIMITS= option in the PROC MACONTROL statement specifies the data set containing the control limits parameters. By default, this information is read from the first observation in the LIMITS= data set for which

- the value of _VAR_ matches the *process* name Gap
- the value of _SUBGRP_ matches the *subgroup-variable* name Day

Note that the moving average plotted for the 39th day lies above the upper control limit, signalling an out-of-control process.

In this example, the LIMITS= data set was created in a previous run of the MACONTROL procedure. You can also create a LIMITS= data set with the DATA step. See "LIMITS= Data Set" on page 871 for details concerning the variables that you must provide, and see Example 9.6 for an illustration.

Syntax: MACHART Statement

The basic syntax for the MACHART statement is as follows:

MACHART *process* ∗ *subgroup-variable* / **SPAN=***value* < *options* > **;**

The general form of this syntax is as follows:

MACHART *processes* ∗ *subgroup-variable* < **(***block-variables***)** >
< **=***symbol-variable* | **=***'character'* > / **SPAN=***value* < *options* > **;**

Note that the SPAN= option is required unless its *value* is read from a LIMITS= data set. You can use any number of MACHART statements in the MACONTROL procedure. The components of the MACHART statement are described as follows.

process

processes

identify one or more processes to be analyzed. The specification of *process* depends on the input data set specified in the PROC MACONTROL statement.

- If raw data are read from a DATA= data set, *process* must be the name of the variable containing the raw measurements. For an example, see "Creating Moving Average Charts from Raw Data" on page 839.
- If summary data are read from a HISTORY= data set, *process* must be the common prefix of the summary variables in the HISTORY= data set. For an example, see "Creating Moving Average Charts from Subgroup Summary Data" on page 842.
- If summary data and control limits are read from a TABLE= data set, *process* must be the value of the variable _VAR_ in the TABLE= data set. For an example, see "Saving Control Limit Parameters" on page 845.

A *process* is required. If more than one *process* is specified, enclose the list in parentheses. For example, the following statements request distinct moving average charts (each with a span of 3) for Weight, Length, and Width:

```
proc macontrol data=Measures;
   machart (Weight Length Width)*Day / span=3;
run;
```

subgroup-variable

is the variable that classifies the data into subgroups. The *subgroup-variable* is required. In the preceding MACHART statement, Day is the subgroup variable. For details, see "Subgroup Variables" on page 1871.

block-variables

are optional variables that group the data into blocks of consecutive subgroups. The blocks are labeled in a legend, and each *block-variable* provides one level of labels in the legend. See "Displaying Stratification in Blocks of Observations" on page 2026 for an example.

symbol-variable

is an optional variable whose levels (unique values) determine the symbol marker or plotting character used to plot the moving averages.

- If you produce a line printer chart, an ‘A’ is displayed for points corresponding to the first level of the *symbol-variable*, a ‘B’ is displayed for points corresponding to the second level, and so on.
- If you produce traditional graphics, distinct symbol markers are displayed for points corresponding to the various levels of the *symbol-variable*. You can specify the symbol markers with SYMBOL*n* statements. See “Displaying Stratification in Levels of a Classification Variable” on page 2025 for an example.

character

specifies a plotting character for line printer charts. For example, the following statements create a moving average chart using an asterisk (*) to plot the points:

```
proc macontrol data=Values lineprinter;
   machart Weight*Hour='*' / span=3;
run;
```

options

specify chart parameters, enhance the appearance of the chart, request additional analyses, save results in data sets, and so on. The section “Summary of Options” on page 851, which follows, lists all options by function.

Summary of Options

The following tables list the MACHART statement options by function. Options unique to the MACONTROL procedure are listed in Table 9.7, and are described in detail in the section “Dictionary of Special Options” on page 859. Options that are common to both the MACONTROL and SHEWHART procedures are listed in Table 9.8. They are described in detail in “Dictionary of Options: SHEWHART Procedure” on page 1946.

Table 9.7 MACHART Statement Special Options

Option	Description
Options for Specifying Uniformly Weighted Moving Average Charts	
ALPHA=	requests probability limits for control charts
ASYMPTOTIC	requests constant control limits
LIMITN=	specifies either a fixed nominal sample size (n) for control limits or allows the control limits to vary with subgroup sample size
MU0=	specifies a standard (known) value μ_0 for the process mean
NOREADLIMITS	specifies that control limit parameters are not to be read from LIMITS= data set
READALPHA	reads _ALPHA_ instead of _SIGMAS_ from LIMITS= data set when both variables are available
READINDEX=	reads control limit parameters from the first observation in the LIMITS= data set where the variable _INDEX_ equals *value*

Table 9.7 *continued*

Option	Description
READLIMITS	reads control limit parameters from a LIMITS= data set (SAS 6.09 and earlier releases)
SIGMA0=	specifies standard (known) value σ_0 for process standard deviation
SIGMAS=	specifies width of control limits in terms of multiple k of standard error of plotted moving averages
SPAN=	specifies the number of terms in the moving average
Options for Plotting Subgroup Means	
CMEANSYMBOL=	specifies color for MEANSYMBOL= symbol
MEANCHAR=	specifies *character* to plot subgroup means on line printer charts
MEANSYMBOL=	specifies symbol to plot subgroup means in traditional graphics

Table 9.8 MACHART Statement General Options

Option	Description
Options for Displaying Control Limits	
CINFILL=	specifies color for area inside control limits
CLIMITS=	specifies color of control limits, central line, and related labels
LCLLABEL=	specifies label for lower control limit
LIMLABSUBCHAR=	specifies a substitution character for labels provided as quoted strings; the character is replaced with the value of the control limit
LLIMITS=	specifies line type for control limits
NDECIMAL=	specifies number of digits to right of decimal place in default labels for control limits and central line
NOCTL	suppresses display of central line
NOLCL	suppresses display of lower control limit
NOLIMITLABEL	suppresses labels for control limits and central line
NOLIMITS	suppresses display of control limits
NOLIMITSFRAME	suppresses default frame around control limit information when multiple sets of control limits are read from a LIMITS= data set
NOLIMITSLEGEND	suppresses legend for control limits
NOUCL	suppresses display of upper control limit
UCLLABEL=	specifies label for upper control limit
WLIMITS=	specifies width for control limits and central line
XSYMBOL=	specifies label for central line
Process Mean and Standard Deviation Options	
SMETHOD=	specifies method for estimating process standard deviation σ
TYPE=	identifies parameters as estimates or standard values and specifies value of _TYPE_ in the OUTLIMITS= data set

Table 9.8 *continued*

Option	Description
Options for Plotting and Labeling Points	
ALLLABEL=	labels every point on moving average chart
ALLLABEL2=	labels every point on trend chart
CLABEL=	specifies color for labels
CCONNECT=	specifies color for line segments that connect points on chart
CFRAMELAB=	specifies fill color for frame around labeled points
CNEEDLES=	specifies color for needles that connect points to central line
COUT=	specifies color for portions of line segments that connect points outside control limits
COUTFILL=	specifies color for shading areas between the connected points and control limits outside the limits
LABELANGLE=	specifies angle at which labels are drawn
LABELFONT=	specifies software font for labels (alias for the TEST-FONT= option)
LABELHEIGHT=	specifies height of labels (alias for the TESTHEIGHT= option)
NEEDLES	connects points to central line with vertical needles
NOCONNECT	suppresses line segments that connect points on chart
NOTRENDCONNECT	suppresses line segments that connect points on trend chart
OUTLABEL=	labels points outside control limits
SYMBOLLEGEND=	specifies LEGEND statement for levels of *symbol-variable*
SYMBOLORDER=	specifies order in which symbols are assigned for levels of *symbol-variable*
TURNALL|TURNOUT	turns point labels so that they are strung out vertically
WNEEDLES=	specifies width of needles
Axis and Axis Label Options	
CAXIS=	specifies color for axis lines and tick marks
CFRAME=	specifies fill colors for frame for plot area
CTEXT=	specifies color for tick mark values and axis labels
DISCRETE	produces horizontal axis for discrete numeric group values
HAXIS=	specifies major tick mark values for horizontal axis
HEIGHT=	specifies height of axis label and axis legend text
HMINOR=	specifies number of minor tick marks between major tick marks on horizontal axis
HOFFSET=	specifies length of offset at both ends of horizontal axis
INTSTART=	specifies first major tick mark value on horizontal axis when a date, time, or datetime format is associated with numeric subgroup variable
NOHLABEL	suppresses label for horizontal axis

Table 9.8 *continued*

Option	Description
NOTICKREP	specifies that only the first occurrence of repeated, adjacent subgroup values is to be labeled on horizontal axis
NOVANGLE	requests vertical axis labels that are strung out vertically
SKIPHLABELS=	specifies thinning factor for tick mark labels on horizontal axis
SPLIT=	specifies splitting character for axis labels
TURNHLABELS	requests horizontal axis labels that are strung out vertically
VAXIS=	specifies major tick mark values for vertical axis of moving average chart
VAXIS2=	specifies major tick mark values for vertical axis of trend chart
VFORMAT=	specifies format for primary vertical axis tick mark labels
VFORMAT2=	specifies format for secondary vertical axis tick mark labels
VMINOR=	specifies number of minor tick marks between major tick marks on vertical axis
VOFFSET=	specifies length of offset at both ends of vertical axis
VZERO	forces origin to be included in vertical axis for primary chart
VZERO2	forces origin to be included in vertical axis for secondary chart
WAXIS=	specifies width of axis lines
Plot Layout Options	
ALLN	plots means for all subgroups
BILEVEL	creates control charts using half-screens and half-pages
EXCHART	creates control charts for a process only when exceptions occur
INTERVAL=	natural time interval between consecutive subgroup positions when time, date, or datetime format is associated with a numeric subgroup variable
MAXPANELS=	maximum number of pages or screens for chart
NMARKERS	requests special markers for points corresponding to sample sizes not equal to nominal sample size for fixed control limits
NOCHART	suppresses creation of chart
NOFRAME	suppresses frame for plot area
NOLEGEND	suppresses legend for subgroup sample sizes
NPANELPOS=	specifies number of subgroup positions per panel on each chart
REPEAT	repeats last subgroup position on panel as first subgroup position of next panel
TOTPANELS=	specifies number of pages or screens to be used to display chart

Table 9.8 *continued*

Option	Description
TRENDVAR=	specifies list of trend variables
YPCT1=	specifies length of vertical axis on moving average chart as a percentage of sum of lengths of vertical axes for moving average and trend charts
ZEROSTD	displays moving average chart regardless of whether $\hat{\sigma} = 0$
Reference Line Options	
CHREF=	specifies color for lines requested by HREF= and HREF2= options
CVREF=	specifies color for lines requested by VREF= and VREF2= options
HREF=	specifies position of reference lines perpendicular to horizontal axis on moving average chart
HREF2=	specifies position of reference lines perpendicular to horizontal axis on trend chart
HREFDATA=	specifies position of reference lines perpendicular to horizontal axis on moving average chart
HREF2DATA=	specifies position of reference lines perpendicular to horizontal axis on trend chart
HREFLABELS=	specifies labels for HREF= lines
HREF2LABELS=	specifies labels for HREF2= lines
HREFLABPOS=	specifies position of HREFLABELS= and HREF2LABELS= labels
LHREF=	specifies line type for HREF= and HREF2= lines
LVREF=	specifies line type for VREF= and VREF2= lines
NOBYREF	specifies that reference line information in a data set applies uniformly to charts created for all BY groups
VREF=	specifies position of reference lines perpendicular to vertical axis on moving average chart
VREF2=	specifies position of reference lines perpendicular to vertical axis on trend chart
VREFLABELS=	specifies labels for VREF= lines
VREF2LABELS=	specifies labels for VREF2= lines
VREFLABPOS=	position of VREFLABELS= and VREF2LABELS= labels
Grid Options	
CGRID=	specifies color for grid requested with GRID or ENDGRID option
ENDGRID	adds grid after last plotted point
GRID	adds grid to control chart
LENDGRID=	specifies line type for grid requested with the ENDGRID option
LGRID=	specifies line type for grid requested with the GRID option

Table 9.8 *continued*

Option	Description
WGRID=	specifies width of grid lines
Clipping Options	
CCLIP=	specifies color for plot symbol for clipped points
CLIPFACTOR=	determines extent to which extreme points are clipped
CLIPLEGEND=	specifies text for clipping legend
CLIPLEGPOS=	specifies position of clipping legend
CLIPSUBCHAR=	specifies substitution character for CLIPLEGEND= text
CLIPSYMBOL=	specifies plot symbol for clipped points
CLIPSYMBOLHT=	specifies symbol marker height for clipped points
Graphical Enhancement Options	
ANNOTATE=	specifies annotate data set that adds features to moving average chart
ANNOTATE2=	specifies annotate data set that adds features to trend chart
DESCRIPTION=	specifies description of moving average chart's GRSEG catalog entry
FONT=	specifies software font for labels and legends on charts
NAME=	specifies name of moving average chart's GRSEG catalog entry
PAGENUM=	specifies the form of the label used in pagination
PAGENUMPOS=	specifies the position of the page number requested with the PAGENUM= option
WTREND=	specifies width of line segments connecting points on trend chart
Options for Producing Graphs Using ODS Styles	
BLOCKVAR=	specifies one or more variables whose values define colors for filling background of *block-variable* legend
CFRAMELAB	draws a frame around labeled points
COUT	draw portions of line segments that connect points outside control limits in a contrasting color
CSTAROUT	specifies that portions of stars exceeding inner or outer circles are drawn using a different color
OUTFILL	shades areas between control limits and connected points lying outside the limits
STARFILL=	specifies a variable identfying groups of stars filled with different colors
STARS=	specifies a variable identfying groups of stars whose outlines are drawn with different colors
Options for ODS Graphics	
BLOCKREFTRANSPARENCY=	specifies the wall fill transparency for blocks and phases
INFILLTRANSPARENCY=	specifies the control limit infill transparency
MARKERS	plots subgroup points with markers
NOBLOCKREF	suppresses block and phase reference lines

Table 9.8 *continued*

Option	Description
NOBLOCKREFFILL	suppresses block and phase wall fills
NOFILLLEGEND	suppresses legend for levels of a STARFILL= variable
NOPHASEREF	suppresses block and phase reference lines
NOPHASEREFFILL	suppresses block and phase wall fills
NOREF	suppresses block and phase reference lines
NOREFFILL	suppresses block and phase wall fills
NOSTARFILLLEGEND	suppresses legend for levels of a STARFILL= variable
NOTRANSPARENCY	disables transparency in ODS Graphics output
ODSFOOTNOTE=	specifies a graph footnote
ODSFOOTNOTE2=	specifies a secondary graph footnote
ODSLEGENDEXPAND	specifies that legend entries contain all levels observed in the data
ODSTITLE=	specifies a graph title
ODSTITLE2=	specifies a secondary graph title
OUTFILLTRANSPARENCY=	specifies control limit outfill transparency
OVERLAYURL=	specifies URLs to associate with overlay points
OVERLAY2URL=	specifies URLs to associate with overlay points on secondary chart
PHASEPOS=	specifies vertical position of phase legend
PHASEREFLEVEL=	associates phase and block reference lines with either innermost or the outermost level
PHASEREFTRANSPARENCY=	specifies the wall fill transparency for blocks and phases
REFFILLTRANSPARENCY=	specifies the wall fill transparency for blocks and phases
SIMULATEQCFONT	draws central line labels using a simulated software font
STARTRANSPARENCY=	specifies star fill transparency
URL=	specifies a variable whose values are URLs to be associated with subgroups
URL2=	specifies a variable whose values are URLs to be associated with subgroups on secondary chart
Input Data Set Options	
MISSBREAK	specifies that observations with missing values are not to be processed
Output Data Set Options	
OUTHISTORY=	creates output data set containing subgroup summary statistics
OUTINDEX=	specifies value of _INDEX_ in the OUTLIMITS= data set
OUTLIMITS=	creates output data set containing control limits
OUTTABLE=	creates output data set containing subgroup summary statistics and control limits
Tabulation Options	
NOTE: specifying (EXCEPTIONS) after a tabulation option creates a table for exceptional points only.	
TABLE	creates a basic table of subgroup means, subgroup sample sizes, and control limits

Table 9.8 *continued*

Option	Description
TABLEALL	is equivalent to the options TABLE, TABLECENTRAL, TABLEID, TABLELEGEND, TABLEOUTLIM, and TABLETESTS
TABLECENTRAL	augments basic table with values of central lines
TABLEID	augments basic table with columns for ID variables
TABLEOUTLIM	augments basic table with columns indicating control limits exceeded
Block Variable Legend Options	
BLOCKLABELPOS=	specifies position of label for *block-variable* legend
BLOCKLABTYPE=	specifies text size of *block-variable* legend
BLOCKPOS=	specifies vertical position of *block-variable* legend
BLOCKREP	repeats identical consecutive labels in *block-variable* legend
CBLOCKLAB=	specifies fill colors for frames enclosing variable labels in *block-variable* legend
CBLOCKVAR=	specifies one or more variables whose values are colors for filling background of *block-variable* legend
Phase Options	
CPHASELEG=	specifies text color for *phase* legend
OUTPHASE=	specifies value of _PHASE_ in the OUTHISTORY= data set
PHASEBREAK	disconnects last point in a *phase* from first point in next *phase*
PHASELABTYPE=	specifies text size of *phase* legend
PHASELEGEND	displays *phase* labels in a legend across top of chart
PHASELIMITS	labels control limits for each phase, provided they are constant within that phase
PHASEREF	delineates *phases* with vertical reference lines
READPHASES=	specifies *phases* to be read from an input data set
Star Options	
CSTARCIRCLES=	specifies color for STARCIRCLES= circles
CSTARFILL=	specifies color for filling stars
CSTAROUT=	specifies outline color for stars exceeding inner or outer circles
CSTARS=	specifies color for outlines of stars
LSTARCIRCLES=	specifies line types for STARCIRCLES= circles
LSTARS=	specifies line types for outlines of STARVERTICES= stars
STARBDRADIUS=	specifies radius of outer bound circle for vertices of stars
STARCIRCLES=	specifies reference circles for stars
STARINRADIUS=	specifies inner radius of stars
STARLABEL=	specifies vertices to be labeled
STARLEGEND=	specifies style of legend for star vertices
STARLEGENDLAB=	specifies label for STARLEGEND= legend

Table 9.8 *continued*

Option	Description
STAROUTRADIUS=	specifies outer radius of stars
STARSPECS=	specifies method used to standardize vertex variables
STARSTART=	specifies angle for first vertex
STARTYPE=	specifies graphical style of star
STARVERTICES=	superimposes star at each point on moving average chart
WSTARCIRCLES=	specifies width of STARCIRCLES= circles
WSTARS=	specifies width of STARVERTICES= stars
Options for Interactive Control Charts	
HTML=	specifies a variable whose values create links to be associated with subgroups
HTML2=	specifies variable whose values create links to be associated with subgroups on secondary chart
HTML_LEGEND=	specifies a variable whose values create links to be associated with symbols in the symbol legend
WEBOUT=	creates an OUTTABLE= data set with additional graphics coordinate data
Options for Line Printer Charts	
CLIPCHAR=	specifies plot character for clipped points
CONNECTCHAR=	specifies character used to form line segments that connect points on chart
HREFCHAR=	specifies line character for HREF= and HREF2= lines
SYMBOLCHARS=	specifies characters indicating *symbol-variable*
VREFCHAR=	specifies line character for VREF= and VREF2= lines

Dictionary of Special Options

ALPHA=*value*

requests *probability limits*. If you specify ALPHA=α, the control limits are computed so that the probability is α that a single moving average exceeds its control limits. The value of α can range between 0 and 1. This assumes that the process is in statistical control and that the data follow a normal distribution. For the equations used to compute probability limits, see "Control Limits" on page 863.

Note the following:

- As an alternative to specifying ALPHA=α, you can read α from the variable _ALPHA_ in a LIMITS= data set by specifying the READALPHA option.
- As an alternative to specifying ALPHA=α (or reading _ALPHA_ from a LIMITS= data set), you can request "$k\sigma$ control limits" by specifying SIGMAS=k (or reading _SIGMAS_ from a LIMITS= data set).

If you specify neither the ALPHA= option nor the SIGMAS= option, the procedure computes 3σ control limits by default.

ASYMPTOTIC

requests constant upper and lower control limits for all subgroups having the following values:

$$\text{LCL} = \overline{\overline{X}} - \frac{k\hat{\sigma}}{\sqrt{nw}}$$
$$\text{UCL} = \overline{\overline{X}} + \frac{k\hat{\sigma}}{\sqrt{nw}}$$

Here w is the span of the moving average, and n is the nominal sample size associated with the control limits. Substitute $\Phi^{-1}(1-\alpha/2)$ for k if you specify probability limits with the ALPHA= option. When you do not specify the ASYMPTOTIC option, the control limits are computed using the exact formulas in Table 9.9. Use the ASYMPTOTIC option only if all the subgroup sample sizes are the same or if you specify LIMITN=*n*.

CMEANSYMBOL=*color*

specifies the *color* used for the symbol requested with the MEANSYMBOL= option in traditional graphics. This option is ignored unless you are producing traditional graphics.

LIMITN=*n*

LIMITN=VARYING

specifies either a fixed or varying nominal sample size for the control limits.

If you specify LIMITN=*n*, moving averages are calculated and displayed only for those subgroups with a sample size equal to *n*, unless you also specify the ALLN option, which causes all the moving averages to be calculated and displayed. By default (or if you specify LIMITN=VARYING), moving averages are calculated and displayed for all subgroups, regardless of sample size.

MEANCHAR='*character*'

specifies a *character* used in legacy line printer charts to plot the subgroup mean for each subgroup. By default, subgroup means are not plotted. This option is ignored unless you specify the LINEPRINTER option in the PROC MACONTROL statement.

MEANSYMBOL=*keyword*

specifies a symbol used to plot the subgroup mean for each subgroup in traditional graphics. By default, subgroup means are not plotted. This option is ignored unless you are producing traditional graphics.

MU0=*value*

specifies a known (standard) value μ_0 for the process mean μ. By default, μ is estimated from the data.

NOTE: As an alternative to specifying MU0=μ_0, you can read a predetermined value for μ_0 from the variable _MEAN_ in a LIMITS= data set.

See Example 9.6.

NOREADLIMITS

specifies that control limit parameters for each *process* listed in the MACHART statement are *not* to be read from the LIMITS= data set specified in the PROC MACONTROL statement.

The following example illustrates the NOREADLIMITS option:

```
proc macontrol data=Pistons limits=Diamlim;
   machart Diameter*Hour;
   machart Diameter*Hour / noreadlimits span=3;
run;
```

The first MACHART statement reads the control limits from the first observation in the data set Diamlim for which the variable _VAR_ is equal to 'Diameter' and the variable _SUBGRP_ is equal to 'Hour'. The second MACHART statement computes estimates of the process mean and standard deviation for the control limits from the measurements in the data set Pistons. Note that the second MACHART statement is equivalent to the following statements, which would be more commonly used:

```
proc macontrol data=Pistons;
   machart Diameter*Hour / span=3;
run;
```

For more information about reading control limit parameters from a LIMITS= data set, see the READLIMITS option later in this list.

READALPHA

specifies that the variable _ALPHA_, rather than the variable _SIGMAS_, is to be read from a LIMITS= data set when both variables are available in the data set. Thus the limits displayed are probability limits. If you do not specify the READALPHA option, then _SIGMAS_ is read by default.

READINDEX=*'value'*

reads control limit parameters from a LIMITS= data set (specified in the PROC MACONTROL statement) for each *process* listed in the MACHART statement. The control limit parameters for a particular *process* are read from the first observation in the LIMITS= data set for which

- the value of _VAR_ matches *process*
- the value of _SUBGRP_ matches the *subgroup-variable*
- the value of _INDEX_ matches *value*

The *value* can be up to 48 characters and must be enclosed in quotes.

READLIMITS

specifies that control limit parameters are to be read from a LIMITS= data set specified in the PROC MACONTROL statement. The parameters for a particular *process* are read from the first observation in the LIMITS= data set for which

- the value of _VAR_ matches *process*
- the value of _SUBGRP_ matches the *subgroup variable*

NOTE: In SAS 6.10 and later releases, the READLIMITS option is not necessary.

SIGMA0=*value*

specifies a known (standard) value σ_0 for the process standard deviation σ. The *value* must be positive. By default, the MACONTROL procedure estimates σ from the data using the formulas given in "Methods for Estimating the Standard Deviation" on page 874.

NOTE: As an alternative to specifying SIGMA0=σ_0, you can read a predetermined value for σ_0 from the variable _STDDEV_ in a LIMITS= data set.

SIGMAS=*value*

specifies the width of the control limits in terms of the multiple k of the standard error of the plotted moving averages on the chart. The value of k must be positive. By default, k = 3 and the control limits are 3σ limits.

SPAN=*value*

specifies the number of terms used to calculate the moving average (*value* is an integer greater than 1). The SPAN= option is required unless you read control limit parameters from a LIMITS= data set or a TABLE= data set. See "Plotted Points" on page 862 and "Choosing the Span of the Moving Average" on page 864 for details.

Details: MACHART Statement

Constructing Uniformly Weighted Moving Average Charts

The following notation is used in this section:

A_i	uniformly weighted moving average for the *i*th subgroup
w	span parameter (number of terms in moving average)
μ	process mean (expected value of the population of measurements)
σ	process standard deviation (standard deviation of the population of measurements)
x_{ij}	*j*th measurement in *i*th subgroup, with j =1, 2, 3, ..., n_i
n_i	sample size of *i*th subgroup
$\overline{X}_i$	mean of measurements in *i*th subgroup. If $n_i = 1$, then the subgroup mean reduces to the single observation in the subgroup.
$\overline{\overline{X}}$	weighted average of subgroup means
$\Phi^{-1}(\cdot)$	inverse standard normal function

Plotted Points

Each point on the chart indicates the value of the uniformly weighted moving average for that subgroup. The moving average for the *i*th subgroup (A_i) is defined as

$$A_i = (\overline{X}_1 + \ldots + \overline{X}_i)/i \qquad \text{if } i < w$$

$$A_i = (\overline{X}_i + \ldots + \overline{X}_{i-w+1})/w \quad \text{if } i \geq w$$

where w is the span, or number of terms, of the moving average. You can specify the span with the SPAN= option in the MACHART statement or with the value of _SPAN_ in a LIMITS= data set.

Central Line

By default, the central line on a moving average chart indicates an estimate for μ, which is computed as

$$\hat{\mu} = \overline{\overline{X}} = \frac{n_1\bar{X}_1 + \cdots + n_N\bar{X}_N}{n_1 + \cdots + n_N}$$

If you specify a known value (μ_0) for μ, the central line indicates the value of μ_0.

Control Limits

You can compute the limits in the following ways:

- as a specified multiple (k) of the standard error of A_i above and below the central line. The default limits are computed with $k = 3$ (these are referred to as 3σ *limits*).
- as probability limits defined in terms of α, a specified probability that A_i exceeds the limits

The following table presents the formulas for the limits:

Table 9.9 Limits for Moving Average Chart

Control Limits
$\text{LCL} = \overline{\overline{X}} - k(\hat{\sigma}/\min(i,w))\sqrt{(1/n_i) + (1/n_{i-1}) + \ldots + (1/n_{1+\max(i-w,0)})}$
$\text{UCL} = \overline{\overline{X}} + k(\hat{\sigma}/\min(i,w))\sqrt{(1/n_i) + (1/n_{i-1}) + \ldots + (1/n_{1+\max(i-w,0)})}$

Probability Limits
$\text{LCL} = \overline{\overline{X}} - \Phi^{-1}(1-\alpha/2)(\hat{\sigma}/\min(i,w))\sqrt{(1/n_i) + (1/n_{i-1}) + \ldots + (1/n_{1+\max(i-w,0)})}$
$\text{UCL} = \overline{\overline{X}} + \Phi^{-1}(1-\alpha/2)(\hat{\sigma}/\min(i,w))\sqrt{(1/n_i) + (1/n_{i-1}) + \ldots + (1/n_{1+\max(i-w,0)})}$

These formulas assume that the data are normally distributed. If standard values μ_0 and σ_0 are available for μ and σ, respectively, replace $\overline{\overline{X}}$ with μ_0 and replace $\hat{\sigma}$ with σ_0 in Table 9.9. Note that the limits vary with both n_i and i.

If the subgroup sample sizes are constant ($n_i = n$), the formulas for the control limits simplify to

$$
\begin{aligned}
\text{LCL} &= \overline{\overline{X}} - \frac{k\hat{\sigma}}{\sqrt{n \min(i,w)}} \\
\text{UCL} &= \overline{\overline{X}} + \frac{k\hat{\sigma}}{\sqrt{n \min(i,w)}}
\end{aligned}
$$

Refer to Montgomery (1996) for more details. When the subgroup sample sizes are constant, the width of the control limits for the first w moving averages decreases monotonically because each of the first w moving averages includes one more term than the preceding moving average.

If you specify the ASYMPTOTIC option, constant control limits with the following values are displayed:

$$
\begin{aligned}
\text{LCL} &= \overline{\overline{X}} - \frac{k\hat{\sigma}}{\sqrt{nw}} \\
\text{UCL} &= \overline{\overline{X}} + \frac{k\hat{\sigma}}{\sqrt{nw}}
\end{aligned}
$$

For asymptotic probability limits, replace k with $\Phi^{-1}(1-\alpha/2)$ in these equations. You can display asymptotic limits by specifying the ASYMPTOTIC option.

You can specify parameters for the moving average limits as follows:

- Specify k with the SIGMAS= option or with the variable _SIGMAS_ in a LIMITS= data set.
- Specify α with the ALPHA= option or with the variable _ALPHA_ in a LIMITS= data set.
- Specify a constant nominal sample size $n_i \equiv n$ for the control limits with the LIMITN= option or with the variable _LIMITN_ in a LIMITS= data set.
- Specify w with the SPAN= option or with the variable _SPAN_ in a LIMITS= data set.
- Specify μ_0 with the MU0= option or with the variable _MEAN_ in a LIMITS= data set.
- Specify σ_0 with the SIGMA0= option or with the variable _STDDEV_ in a LIMITS= data set.

Choosing the Span of the Moving Average

There are few published guidelines for choosing the span w. In some applications, practical experience may dictate the choice of w. A more systematic approach is to choose w by considering its effect on the average run length (the expected number of points plotted before a shift is detected). This effect was studied by Roberts (1959), who used simulation methods.

You can use Table 9.10 and Table 9.11 to find a combination of k and w that yields a desired ARL for an in-control process ($\delta = 0$) and for a specified shift of δ.

Table 9.10 Average Run Lengths for One-Sided Uniformly Weighted Moving Average Charts

		w (span)						
k	δ	**2**	**3**	**4**	**5**	**6**	**8**	**10**
2.0	0.00	51.58	60.97	70.58	80.18	89.78	108.65	127.47
2.0	0.25	25.01	26.47	28.00	29.33	30.76	33.08	35.18
2.0	0.50	13.41	13.31	13.40	13.69	14.01	14.66	15.17
2.0	0.75	8.00	7.75	7.78	7.97	8.15	8.60	9.06
2.0	1.00	5.27	5.20	5.29	5.45	5.67	6.15	6.69
2.0	1.50	2.90	3.03	3.24	3.50	3.73	4.23	4.66
2.0	2.00	2.04	2.27	2.51	2.73	2.95	3.32	3.65
2.0	2.50	1.68	1.91	2.11	2.31	2.48	2.78	3.04
2.0	3.00	1.46	1.68	1.85	2.01	2.16	2.40	2.63
2.0	4.00	1.20	1.38	1.52	1.64	1.75	1.94	2.10
2.0	5.00	1.06	1.18	1.31	1.41	1.50	1.65	1.79
2.5	0.00	179.92	204.43	230.32	259.32	287.08	339.71	394.43
2.5	0.25	72.62	71.56	72.48	72.93	73.40	75.54	77.47
2.5	0.50	33.67	30.13	28.54	27.49	26.93	26.29	26.03
2.5	0.75	17.28	15.01	13.91	13.42	13.13	13.00	13.10
2.5	1.00	9.94	8.66	8.20	8.01	7.96	8.24	8.63
2.5	1.50	4.43	4.13	4.21	4.39	4.64	5.17	5.69
2.5	2.00	2.65	2.77	3.03	3.29	3.54	4.01	4.43

Table 9.10 (continued)

k	δ	**2**	**3**	**4**	**5**	**6**	**8**	**10**
2.5	2.50	1.98	2.24	2.50	2.74	2.95	3.32	3.67
2.5	3.00	1.70	1.95	2.17	2.37	2.55	2.86	3.14
2.5	4.00	1.37	1.59	1.76	1.90	2.03	2.28	2.49
2.5	5.00	1.15	1.35	1.51	1.62	1.73	1.92	2.08
3.0	0.00	792.24	867.57	963.95	1051.77	1150.79	1345.96	1539.75
3.0	0.25	269.28	244.26	231.50	226.25	220.89	209.87	204.74
3.0	0.50	104.18	83.86	72.84	65.43	60.85	54.62	50.34
3.0	0.75	45.69	34.45	28.79	25.69	23.66	21.24	20.15
3.0	1.00	22.73	16.74	14.20	12.89	12.12	11.52	11.45
3.0	1.50	7.65	6.16	5.70	5.64	5.75	6.23	6.78
3.0	2.00	3.77	3.49	3.63	3.89	4.17	4.71	5.20
3.0	2.50	2.46	2.63	2.90	3.18	3.43	3.88	4.28
3.0	3.00	1.96	2.23	2.50	2.74	2.95	3.33	3.65
3.0	4.00	1.57	1.81	2.00	2.18	2.34	2.62	2.87
3.0	5.00	1.30	1.55	1.72	1.85	1.97	2.20	2.40
3.5	0.00	4275.15	4536.99	4853.63	5168.75	5485.97	6088.03	6613.01
3.5	0.25	1281.12	1078.59	964.86	886.26	830.03	751.66	684.98
3.5	0.50	413.30	294.47	235.00	197.27	169.50	136.01	115.48
3.5	0.75	153.50	98.31	73.49	59.29	50.49	40.45	34.53
3.5	1.00	63.68	39.34	29.37	24.06	20.88	17.70	16.12
3.5	1.50	15.84	10.44	8.50	7.78	7.47	7.51	7.97
3.5	2.00	6.06	4.73	4.49	4.61	4.86	5.43	6.01
3.5	2.50	3.27	3.13	3.34	3.63	3.92	4.45	4.91
3.5	3.00	2.31	2.54	2.83	3.11	3.36	3.80	4.19
3.5	4.00	1.77	2.02	2.25	2.45	2.64	2.97	3.27
3.5	5.00	1.48	1.74	1.91	2.06	2.21	2.48	2.71

Table 9.11 Average Run Lengths for Two-Sided Uniformly Weighted Moving Average Charts

		***w* (span)**						
k	δ	2	3	4	5	6	8	10
2.0	0.00	25.46	29.62	33.94	38.08	42.35	51.20	59.48
2.0	0.25	20.43	22.38	24.21	25.87	27.35	30.08	32.33
2.0	0.50	12.73	12.80	13.02	13.29	13.57	14.19	14.84
2.0	0.75	7.87	7.68	7.71	7.86	8.03	8.44	8.90
2.0	1.00	5.24	5.14	5.22	5.40	5.59	6.09	6.60
2.0	1.50	2.90	3.02	3.24	3.48	3.71	4.19	4.63
2.0	2.00	2.04	2.26	2.51	2.73	2.94	3.31	3.63
2.0	2.50	1.67	1.91	2.12	2.30	2.47	2.77	3.03
2.0	3.00	1.46	1.67	1.85	2.01	2.15	2.40	2.63
2.0	4.00	1.20	1.38	1.52	1.64	1.75	1.94	2.10

Table 9.11 *continued*

k	δ	2	3	4	5	6	8	10
2.0	5.00	1.06	1.19	1.31	1.41	1.50	1.65	1.79
2.5	0.00	89.48	101.24	114.35	127.74	140.88	166.98	192.93
2.5	0.25	63.12	64.91	67.00	68.75	69.84	72.22	74.49
2.5	0.50	32.46	29.54	28.20	27.33	26.72	25.92	25.72
2.5	0.75	17.28	14.97	13.85	13.29	13.02	12.81	12.98
2.5	1.00	9.94	8.61	8.16	7.99	8.01	8.23	8.63
2.5	1.50	4.42	4.14	4.20	4.38	4.62	5.16	5.67
2.5	2.00	2.65	2.77	3.03	3.29	3.54	4.00	4.43
2.5	2.50	1.99	2.24	2.50	2.73	2.95	3.33	3.65
2.5	3.00	1.69	1.95	2.17	2.37	2.54	2.86	3.14
2.5	4.00	1.37	1.59	1.76	1.90	2.04	2.27	2.49
2.5	5.00	1.15	1.35	1.51	1.63	1.73	1.92	2.09
3.0	0.00	397.12	436.27	481.16	527.14	574.05	667.68	762.89
3.0	0.25	245.51	228.67	222.75	216.07	213.79	207.03	201.71
3.0	0.50	103.15	83.49	72.47	65.67	60.67	53.93	50.30
3.0	0.75	45.56	34.25	29.01	25.72	23.59	21.12	19.93
3.0	1.00	22.68	16.81	14.19	12.92	12.18	11.54	11.48
3.0	1.50	7.68	6.14	5.71	5.65	5.77	6.23	6.77
3.0	2.00	3.74	3.49	3.63	3.88	4.17	4.71	5.21
3.0	2.50	2.46	2.63	2.90	3.18	3.43	3.89	4.29
3.0	3.00	1.96	2.23	2.50	2.73	2.95	3.32	3.66
3.0	4.00	1.57	1.81	2.00	2.18	2.34	2.62	2.88
3.0	5.00	1.30	1.55	1.72	1.85	1.97	2.20	2.40
3.5	0.00	2217.61	2372.09	2567.27	2775.06	2983.70	3398.08	3810.50
3.5	0.25	1186.27	1027.67	940.30	875.91	826.53	744.59	676.61
3.5	0.50	411.69	295.62	232.68	195.65	169.21	135.73	116.06
3.5	0.75	152.52	97.33	72.30	58.98	50.59	40.22	34.71
3.5	1.00	64.03	39.46	29.18	24.08	20.80	17.54	16.16
3.5	1.50	15.83	10.36	8.47	7.73	7.46	7.56	8.00
3.5	2.00	6.05	4.71	4.49	4.61	4.85	5.44	6.00
3.5	2.50	3.27	3.12	3.34	3.64	3.92	4.44	4.91
3.5	3.00	2.32	2.54	2.83	3.11	3.36	3.80	4.19
3.5	4.00	1.77	2.02	2.25	2.46	2.65	2.97	3.26
3.5	5.00	1.49	1.74	1.91	2.06	2.21	2.48	2.71

For example, suppose you want to construct a two-sided moving average chart with an in-control ARL of 100 and an ARL of 9 for detecting a shift of $\delta = 1$. Table 9.11 shows that the combination $w = 3$ and $k = 2.5$ yields an in-control ARL of 101.24 and an ARL of 8.61 for $\delta = 1$.

Note that you can also use Table 9.10 and Table 9.11 to evaluate an existing moving average chart (see Example 9.7).

The following SAS program computes the average run length for a two-sided moving average chart for various shifts in the mean. This program can be adapted to compute averages run lengths for various combinations of k and w.

```
data sim;
   drop span delta time j y x;
   span=4;
   do shift=0,.25,.5,.75,1,1.5,2,2.5,3,4,5;
      do j=1 to 50000;
         do time=1 to 15000;
            if time<=100 then
               delta=0;
            else
               delta=shift;
            y=delta+rannor(234);
            if time<span then
               x=.;
            else
               x=(y+lag1(y)+lag2(y)+lag3(y))/span;
            if time>=101 and abs(x)>3/sqrt(span)
              then leave;
         end;
         arl=time-100;
         output;
      end;
   end;

proc means;
   class shift;
run;
```

In the preceding program, the size of the span w (SPAN) is 4 and the shifts in the mean are introduced to the variable (Y) $y \sim N(0, 1)$ after the first 100 observations. The first DO loop specifies shifts of various magnitude, the second DO loop performs 50000 simulations for each shift, and the third DO loop counts the run length (TIME), that is, the number of samples observed before the control chart signals. A large upper bound (15000) for TIME is specified so that the run length is uncensored.

The program can be generalized for various span sizes by assigning a different value for the variable SPAN and changing the expression for X appropriately. Optionally, you can compute the ARL for a one-sided chart by changing the limits, that is, `x>3/sqrt(span)`. This was the technique used to construct Table 9.10 and Table 9.11.

Output Data Sets

OUTLIMITS= Data Set

The OUTLIMITS= data set saves the control limit parameters. The following variables can be saved:

Variable	Description
ALPHA	probability (α) of exceeding limits
INDEX	optional identifier for the control limits specified with the OUTINDEX= option
LIMITN	sample size associated with the control limits
MEAN	process mean ($\overline{\overline{X}}$ or μ_0)
SIGMAS	multiple (k) of standard error of A_i
SPAN	number of terms in the moving average
STDDEV	process standard deviation ($\hat{\sigma}$ or σ_0)
SUBGRP	*subgroup-variable* specified in the MACHART statement
TYPE	type (estimate or standard value) of _MEAN_ and _STDDEV_
VAR	*process* specified in the MACHART statement

The OUTLIMITS= data set does not contain the control limits; instead, it contains control limit parameters that can be used to recompute the control limits.

Notes:

1. If the control limits vary with subgroup sample size, the special missing value V is assigned to the variable _LIMITN_.
2. If the limits are defined in terms of a multiple k of the standard error of A_i, the value of _ALPHA_ is computed as $\alpha = 2(1 - \Phi(k))$, where $\Phi(\cdot)$ is the standard normal distribution function.
3. If the limits are probability limits, the value of _SIGMAS_ is computed as $k = \Phi^{-1}(1 - \alpha/2)$, where Φ^{-1} is the inverse standard normal distribution function.
4. Optional BY variables are saved in the OUTLIMITS= data set.

The OUTLIMITS= data set contains one observation for each *process* specified in the MACHART statement.

You can use OUTLIMITS= data sets

- to keep a permanent record of the control limit parameters
- to write reports. You may prefer to use OUTTABLE= data sets for this purpose.
- as LIMITS= data sets in subsequent runs of PROC MACONTROL

For an example of an OUTLIMITS= data set, see "Saving Control Limit Parameters" on page 845.

OUTHISTORY= Data Set

The OUTHISTORY= data set saves subgroup summary statistics. The following variables can be saved:

- the *subgroup-variable*
- a subgroup mean variable named by *process* suffixed with *X*
- a subgroup standard deviation variable named by *process* suffixed with *S*

- a subgroup moving average variable named by *process* suffixed with *A*
- a subgroup sample size variable named by *process* suffixed with *N*

Given a *process* name that contains 32 characters, the procedure first shortens the name to its first 16 characters and its last 15 characters, and then it adds the suffix.

Subgroup summary variables are created for each *process* specified in the MACHART statement. For example, consider the following statements:

```
proc macontrol data=Clips;
   machart (Gap Yieldstrength)*Day / span      =3
                                     outhistory=Cliphist;
run;
```

The data set Cliphist would contain nine variables named Day, GapX, GapS, GapA, GapN, YieldstrengthX, YieldstrengthS, YieldstrengthA, and YieldstrengthN.

Additionally, the following variables, if specified, are included:

- BY variables
- *block-variables*
- *symbol-variable*
- ID variables
- _PHASE_ (if the OUTPHASE= option is specified)

For an example of an OUTHISTORY= data set, see "Saving Summary Statistics" on page 844.

OUTTABLE= Data Set

The OUTTABLE= data set saves subgroup summary statistics, control limits, and related information. The following variables can be saved:

Variable	Description
ALPHA	probability (α) of exceeding control limits
EXLIM	control limit exceeded on moving average chart
LCLA	lower control limit for moving average
LIMITN	nominal sample size associated with the control limits
MEAN	process mean
SIGMAS	multiple (k) of the standard error associated with control limits
SPAN	number of terms in the moving average
subgroup	values of the subgroup variable
SUBN	subgroup sample size
SUBS	subgroup standard deviation
SUBX	subgroup mean
UCLA	upper control limit for moving average
UWMA	uniformly weighted moving average
VAR	*process* specified in MACHART statement

In addition, the following variables, if specified, are included:

- BY variables
- *block-variables*
- ID variables
- _PHASE_ (if the READPHASES= option is specified)
- *symbol-variable*

Notes:

1. Either the variable _ALPHA_ or the variable _SIGMAS_ is saved depending on how the control limits are defined (with the ALPHA= or SIGMAS= options, respectively; or with the corresponding variables in a LIMITS= data set).
2. The variables _VAR_ and _EXLIM_ are character variables of length 8. The variable _PHASE_ is a character variable of length 48. All other variables are numeric.

For an example of an OUTTABLE= data set, see "Saving Control Limit Parameters" on page 845.

ODS Tables

The following table summarizes the ODS tables that you can request with the MACHART statement.

Table 9.12 ODS Tables Produced with the MACHART Statement

Table Name	Description	Options
MACHART	uniformly weighted moving average chart summary statistics	TABLE, TABLEALL, TABLEC, TABLEID, TABLEOUT
Parameters	uniformly weighted moving average parameters	TABLE, TABLEALL, TABLEC, TABLEID, TABLEOUT

ODS Graphics

Before you create ODS Graphics output, ODS Graphics must be enabled (for example, by using the ODS GRAPHICS ON statement). For more information about enabling and disabling ODS Graphics, see the section "Enabling and Disabling ODS Graphics" (Chapter 21, *SAS/STAT User's Guide*). **NOTE:** In SAS/QC 13.1 the MACONTROL procedure does not support the creation of graphs that are editable with the ODS Graphics Editor.

The appearance of a graph produced with ODS Graphics is determined by the style associated with the ODS destination where the graph is produced. MACHART options used to control the appearance of traditional graphics are ignored for ODS Graphics output. Options for Producing Graphs Using ODS Styles lists options that can be used to control the appearance of graphs produced with ODS Graphics or with traditional graphics using ODS styles. Options for ODS Graphics lists options to be used exclusively with ODS Graphics.

Detailed descriptions of these options are provided in "Dictionary of Options: SHEWHART Procedure" on page 1946

When ODS Graphics is in effect, the MACHART statement assigns a name to the graph it creates. You can use this name to reference the graph when using ODS. The name is listed in Table 9.13.

Table 9.13 ODS Graphics Produced by the MACHART Statement

ODS Graph Name	Plot Description
MAChart	moving average chart

See Chapter 3, "SAS/QC Graphics," for more information about ODS Graphics and other methods for producing charts.

Input Data Sets

DATA= Data Set

You can read raw data (process measurements) from a DATA= data set specified in the PROC MACONTROL statement. Each *process* specified in the MACHART statement must be a SAS variable in the DATA= data set. This variable provides measurements that must be grouped into subgroup samples indexed by the *subgroup-variable*. The *subgroup-variable*, which is specified in the MACHART statement, must also be a SAS variable in the DATA= data set. Each observation in a DATA= data set must contain a value for each *process* and a value for the *subgroup-variable*. If the ith subgroup contains n_i items, there should be n_i consecutive observations for which the value of the *subgroup-variable* is the index of the ith subgroup. For example, if each subgroup contains five items and there are 30 subgroup samples, the DATA= data set should contain 150 observations.

Other variables that can be read from a DATA= data set include

- _PHASE_ (if the READPHASES= option is specified)
- *block-variables*
- *symbol-variable*
- BY variables
- ID variables

By default, the MACONTROL procedure reads all of the observations in a DATA= data set. However, if the data set includes the variable _PHASE_, you can read selected groups of observations (referred to as *phases*) with the READPHASES= option (for an example, see "Displaying Stratification in Phases" on page 2031.

For an example of a DATA= data set, see "Creating Moving Average Charts from Raw Data" on page 839.

LIMITS= Data Set

You can read preestablished control limits parameters from a LIMITS= data set specified in the PROC MACONTROL statement. The LIMITS= data set used by the MACONTROL procedure does not contain the actual control limits, but rather it contains the parameters required to compute the limits. For example, the following statements read control limit parameters from the data set Parms:

```
proc macontrol data=Parts limits=Parms;
   machart Gap*Day;
run;
```

The LIMITS= data set can be an OUTLIMITS= data set that was created in a previous run of the MACONTROL procedure. Such data sets always contain the variables required for a LIMITS= data set; see the section "OUTLIMITS= Data Set" on page 867. The LIMITS= data set can also be created directly using a DATA step.

When you create a LIMITS= data set, you must provide the variable _SPAN_, which specifies the number of terms to use in the moving average. In addition, note the following:

- The variables _VAR_ and _SUBGRP_ are required. These must be character variables of length 8.
- The variable _INDEX_ is required if you specify the READINDEX= option. This must be a character variable whose length is no greater than 48.
- The variables _LIMITN_, _SIGMAS_ (or _ALPHA_), and _TYPE_ are optional, but they are recommended to maintain a complete set of control limit information. The variable _TYPE_ must be a character variable of length 8. Valid values are 'ESTIMATE', 'STANDARD', 'STDMEAN', and 'STDSIGMA'.
- BY variables are required if specified with a BY statement.

Some advantages of working with a LIMITS= data set are that

- it facilitates reusing a permanently saved set of parameters
- a distinct set of parameters can be read for each *process* specified in the MACHART statement
- it facilitates keeping track of multiple sets of parameters that accumulate for the same *process* as the process evolves over time

For an example, see "Reading Preestablished Control Limit Parameters" on page 848.

HISTORY= Data Set

You can read subgroup summary statistics from a HISTORY= data set specified in the PROC MACONTROL statement. This enables you to reuse OUTHISTORY= data sets that have been created in previous runs of the MACONTROL, SHEWHART, or CUSUM procedures or to read output data sets created with SAS summarization procedures such as PROC MEANS.

A HISTORY= data set used with the MACHART statement must contain the following:

- the *subgroup-variable*
- a subgroup mean variable for each *process*

- a subgroup sample size variable for each *process*
- a subgroup standard deviation variable for each *process*

The names of the subgroup mean, subgroup standard deviation, and subgroup sample size variables must be the *process* name concatenated with the suffix characters *X*, *S*, and *N*, respectively.

For example, consider the following statements:

```
proc macontrol history=Cliphist;
   machart (Gap Diameter)*Day  / span=3;
run;
```

The data set Cliphist must include the variables Day, GapX, GapS, GapN, DiameterX, DiameterS, and DiameterN.

Although a moving average variable (named by the *process* name suffixed with *A*) is saved in an OUTHISTORY= data set, it is not required in a HISTORY= data set, because the subgroup mean variable is sufficient to compute the moving averages.

Note that if you specify a *process* name that contains 32 characters, the names of the summary variables must be formed from the first 16 characters and the last 15 characters of the *process* name, suffixed with the appropriate character.

Other variables that can be read from a HISTORY= data set include

- _PHASE_ (if the READPHASES= option is specified)
- *block-variables*
- *symbol-variable*
- BY variables
- ID variables

By default, the MACONTROL procedure reads all the observations in a HISTORY= data set. However, if the HISTORY= data set includes the variable _PHASE_, you can read selected groups of observations (referred to as *phases*) by specifying the READPHASES= option (see "Displaying Stratification in Phases" on page 2031 for an example).

For an example of a HISTORY= data set, see "Creating Moving Average Charts from Subgroup Summary Data" on page 842.

TABLE= Data Set

You can read summary statistics and control limits from a TABLE= data set specified in the PROC MACONTROL statement. This enables you to reuse an OUTTABLE= data set created in a previous run of the MACONTROL procedure.

The following table lists the variables required in a TABLE= data set used with the MACHART statement:

Variable	Description
LCLE	lower control limit for Moving Average
LIMITN	nominal sample size associated with the control limits
MEAN	process mean
SPAN	number of terms in the moving average
subgroup-variable	values of the *subgroup-variable*
SUBN	subgroup sample size
SUBS	subgroup standard deviation
SUBX	subgroup mean
UCLA	upper control limit for moving average
UWMA	uniformly weighted moving average

Other variables that can be read from a TABLE= data set include

- *block-variables*
- *symbol-variable*
- BY variables
- ID variables
- _PHASE_ (if the READPHASES= option is specified). This variable must be a character variable whose length is no greater than 48.
- _VAR_. This variable is required if more than one *process* is specified or if the data set contains information for more than one *process*. This variable must be a character variable of length 8.

For an example of a TABLE= data set, see "Saving Control Limit Parameters" on page 845.

Methods for Estimating the Standard Deviation

When control limits are computed from the input data, four methods are available for estimating the process standard deviation σ. Three methods (referred to as the default, MVLUE, and RMSDF) are available with subgrouped data. A fourth method is used if the data are individual measurements (see "Default Method for Individual Measurements" on page 875).

Default Method for Subgroup Samples

This method is the default for moving average charts using subgrouped data. The default estimate of σ is

$$\hat{\sigma} = \frac{s_1/c_4(n_1) + \ldots + s_N/c_4(n_N)}{N}$$

where N is the number of subgroups for which $n_i \geq 2$, s_i is the sample standard deviation of the ith subgroup

$$s_i = \sqrt{\frac{1}{n_i - 1} \sum_{j=1}^{n_i} (x_{ij} - \bar{X}_i)^2}$$

and

$$c_4(n_i) = \frac{\Gamma(n_i/2)\sqrt{2/(n_i - 1)}}{\Gamma((n_i - 1)/2)}$$

Here $\Gamma(\cdot)$ denotes the gamma function, and $\bar{X}_i$ denotes the ith subgroup mean. A subgroup standard deviation s_i is included in the calculation only if $n_i \geq 2$. If the observations are normally distributed, then the expected value of s_i is $c_4(n_i)\sigma$. Thus, $\hat{\sigma}$ is the unweighted average of N unbiased estimates of σ. This method is described in the American Society for Testing and Materials (1976).

MVLUE Method for Subgroup Samples

If you specify SMETHOD=MVLUE, a minimum variance linear unbiased estimate (MVLUE) is computed for σ. Refer to Burr (1969, 1976) and Nelson (1989, 1994). The MVLUE is a weighted average of N unbiased estimates of σ of the form $s_i/c_4(n_i)$, and it is computed as

$$\hat{\sigma} = \frac{h_1 s_1/c_4(n_1) + \ldots + h_N s_N/c_4(n_N)}{h_1 + \ldots + h_N}$$

where

$$h_i = \frac{[c_4(n_i)]^2}{1 - [c_4(n_i)]^2}$$

A subgroup standard deviation s_i is included in the calculation only if $n_i \geq 2$, and N is the number of subgroups for which $n_i \geq 2$. The MVLUE assigns greater weight to estimates of σ from subgroups with larger sample sizes, and it is intended for situations where the subgroup sample sizes vary. If the subgroup sample sizes are constant, the MVLUE reduces to the default estimate.

RMSDF Method for Subgroup Samples

If you specify SMETHOD=RMSDF, a weighted root-mean-square estimate is computed for σ as follows:

$$\hat{\sigma} = \frac{\sqrt{(n_1 - 1)s_1^2 + \cdots + (n_N - 1)s_N^2}}{c_4(n)\sqrt{n_1 + \cdots + n_N - N}}$$

where $n = n_1 + \cdots + n_N - (N - 1)$. The weights are the degrees of freedom $n_i - 1$. A subgroup standard deviation s_i is included in the calculation only if $n_i \geq 2$, and N is the number of subgroups for which $n_i \geq 2$.

If the unknown standard deviation σ is constant across subgroups, the root-mean-square estimate is more efficient than the minimum variance linear unbiased estimate. However, in process control applications it is generally not assumed that σ is constant, and if σ varies across subgroups, the root-mean-square estimate tends to be more inflated than the MVLUE.

Default Method for Individual Measurements

When each subgroup sample contains a single observation ($n_i \equiv 1$), the process standard deviation σ is estimated as

$$\hat{\sigma} = \sqrt{\frac{1}{2(N-1)} \sum_{i=1}^{N-1} (x_{i+1} - x_i)^2}$$

where N is the number of observations, and $x_1, x_2, \ldots, x_N$ are the individual measurements. This formula is given by Wetherill (1977), who states that the estimate of the variance is biased if the measurements are autocorrelated.

Axis Labels

You can specify axis labels by assigning labels to particular variables in the input data set, as summarized in the following table:

Axis	Input Data Set	Variable
Horizontal	all	*subgroup-variable*
Vertical	DATA=	*process*
Vertical	HISTORY=	subgroup mean variable
Vertical	TABLE=	_UWMA_

For example, the following sets of statements specify the label *Moving Average of Clip Gaps* for the vertical axis and the label *Day* for the horizontal axis of the moving average chart:

```
proc macontrol data=Clips1;
   machart Gap*Day / span=4;
   label Gap = 'Moving Average of Clip Gaps';
   label Day = 'Day';
run;

proc macontrol history=cliphist;
   machart Gap*Day / span=4;
   label GapX = 'Moving Average of Clip Gaps';
   label Day  = 'Day';
run;

proc macontrol table=cliptab;
   machart Gap*Day;
   label _uwma_   = 'Moving Average of Clip Gaps';
   label Day      = 'Day';
run;
```

In this example, the label assignments are in effect only for the duration of the procedure step, and they temporarily override any permanent labels associated with the variables.

Missing Values

An observation read from a DATA=, HISTORY=, or TABLE= data set is not analyzed if the value of the subgroup variable is missing. For a particular process variable, an observation read from a DATA= data set is not analyzed if the value of the process variable is missing. Missing values of process variables generally lead to unequal subgroup sample sizes. For a particular process variable, an observation read from a HISTORY= or TABLE= data set is not analyzed if the values of any of the corresponding summary variables are missing.

Examples: MACHART Statement

This section provides advanced examples of the MACHART statement.

Example 9.6: Specifying Standard Values for the Process Mean and Process Standard Deviation

NOTE: See *Standard Values for Moving Average Charts* in the SAS/QC Sample Library.

By default, the MACHART statement estimates the process mean (μ) and standard deviation (σ) from the data. This is illustrated in "Getting Started: MACHART Statement" on page 838. However, there are applications in which standard values (μ_0 and σ_0) are available based, for instance, on previous experience or extensive sampling. You can specify these values with the MU0= and SIGMA0= options.

For example, suppose it is known that the metal clip manufacturing process (introduced in "Creating Moving Average Charts from Raw Data" on page 839) has a mean of 15 and standard deviation of 0.2. The following statements specify these standard values:

```
ods graphics on;
title 'Specifying Standard Process Mean and Standard Deviation';
proc macontrol data=Clips1;
   machart Gap*Day /
       odstitle = title
       mu0      = 15
       sigma0   = 0.2
       span     = 4
       xsymbol  = mu0
       markers;
run;
```

The XSYMBOL= option specifies the label for the central line. The resulting chart is shown in Output 9.6.1.

Output 9.6.1 Specifying Standard Values with MU0= and SIGMA0=



The central line and control limits are determined using μ_0 and σ_0 (see the equations in Table 9.9). Output 9.6.1 indicates that the process is out-of-control since the moving averages for Day=17, Day=19, and Day=20 lie below the lower control limit.

You can also specify μ_0 and σ_0 with the variables _MEAN_ and _STDDEV_ in a LIMITS= data set, as illustrated by the following statements:

```
data Cliplim;
   length _var_ _subgrp_ _type_ $8;
   _var_     = 'Gap';
   _subgrp_  = 'Day';
   _type_    = 'STANDARD';
   _limitn_  = 5;
   _mean_    = 15;
   _stddev_  = 0.2;
   _span_    = 4;
run;

proc macontrol data=Clips1 limits=Cliplim;
   machart Gap*Day / xsymbol=mu0
                     odstitle = title
                     markers;
run;
```

The variable _SPAN_ is required, and its value provides the number of terms in the moving average. The variables _VAR_ and _SUBGRP_ are also required, and their values must match the *process* and *subgroup-variable*, respectively, specified in the MACHART statement. The bookkeeping variable _TYPE_ is not required, but it is recommended to indicate that the variables _MEAN_ and _STDDEV_ provide standard values rather than estimated values.

The resulting chart (not shown here) is identical to the one shown in Output 9.6.1.

Example 9.7: Annotating Average Run Lengths on the Chart

NOTE: See *ARLs Shown on a Moving Average Chart* in the SAS/QC Sample Library.

You can use Table 9.10 and Table 9.11 to find a moving average chart scheme with the desired average run length properties. Specifically, you can find a combination of k and w that yields a desired ARL for an in-control process ($\delta = 0$) and for a specified shift of δ.

You can also use these tables to evaluate an existing moving average chart scheme. For example, the moving average chart shown in Output 9.6.1 has a two-sided scheme with $w = 4$ and $k = 3$. Suppose you want to detect a shift of $\delta = .5$. From Table 9.11, the average run length with $w = 4$, $k = 3$, and $\delta = .5$ is 72.47. The in-control average run length ($\delta = 0$) for this scheme is 481.16.

The following statements create an inset data set that can be read to display these ARL values on the moving average chart:

```
data ARLinset;
   length _label_ $ 8;
   _label_ = 'ARL In';
   _value_ = 481.16;
   output;
   _label_ = 'ARL Out';
   _value_ = 72.47;
   output;
run;
```

The following statements create the moving average chart shown in Output 9.7.1.

```
title 'Moving Average Chart to Detect a Shift of 0.5';
ods graphics on;
proc macontrol data=Clips1;
   machart Gap*Day / mu0      = 15
                     sigma0   = 0.2
                     span     = 4
                     xsymbol  = mu0
                     odstitle = title;
   inset data = ARLinset / header = 'ARL Evaluation'
                           pos    = ne;
run;
```

The average run lengths in this example (481.16 and 72.27) are simply copied from Table 9.11. You can generalize the preceding program so that it computes the average run lengths by incorporating the simulation program from the section “Choosing the Span of the Moving Average” on page 864.

Output 9.7.1 Displaying Average Run Lengths on Chart



For more information on annotating charts with insets, refer to "INSET Statement: MACONTROL Procedure" on page 880.

INSET Statement: MACONTROL Procedure

Overview: INSET Statement

The INSET statement enables you to enhance a moving average control chart by adding a box or table (referred to as an *inset*) of summary statistics directly to the graph. A possible application of an inset is to present moving average parameters on the chart rather than displaying them in a legend. An inset can also display arbitrary values provided in a SAS data set.

Note that the INSET statement by itself does not produce a display but must be used in conjunction with an MACHART or EWMACHART statement. Insets are not available with line printer charts, so the INSET statement is not applicable when the LINEPRINTER option is specified in the PROC MACONTROL statement.

You can use options in the INSET statement to

- specify the position of the inset
- specify a header for the inset table
- specify graphical enhancements, such as background colors, text colors, text height, text font, and drop shadows

Getting Started: INSET Statement

This section introduces the INSET statement with a basic example showing how it is used. See "INSET and INSET2 Statements: SHEWHART Procedure" on page 1928 for a complete description of the INSET statement.

This example is based on the same scenario as the first example in the "Getting Started" section of "EWMACHART Statement: MACONTROL Procedure" on page 785. An EWMA chart is used to analyze data from the manufacture of metal clips. The following statements create a data set containing measurements to be analyzed and the EWMA chart shown in Figure 9.18.

```
data Clips1;
   input Day @ ;
   do i=1 to 5;
      input Gap @ ;
      output;
   end;
   drop i;
   datalines;
 1  14.76  14.82  14.88  14.83  15.23
 2  14.95  14.91  15.09  14.99  15.13
 3  14.50  15.05  15.09  14.72  14.97
 4  14.91  14.87  15.46  15.01  14.99
 5  14.73  15.36  14.87  14.91  15.25
 6  15.09  15.19  15.07  15.30  14.98
 7  15.34  15.39  14.82  15.32  15.23
 8  14.80  14.94  15.15  14.69  14.93
 9  14.67  15.08  14.88  15.14  14.78
10  15.27  14.61  15.00  14.84  14.94
11  15.34  14.84  15.32  14.81  15.17
12  14.84  15.00  15.13  14.68  14.91
13  15.40  15.03  15.05  15.03  15.18
14  14.50  14.77  15.22  14.70  14.80
15  14.81  15.01  14.65  15.13  15.12
16  14.82  15.01  14.82  14.83  15.00
17  14.89  14.90  14.60  14.40  14.88
18  14.90  15.29  15.14  15.20  14.70
19  14.77  14.60  14.45  14.78  14.91
20  14.80  14.58  14.69  15.02  14.85
;
```

```
ods graphics on;
proc macontrol data=Clips1;
   ewmachart Gap*Day / weight   = 0.3
                       odstitle = title
                       markers
                       nolegend;
   inset stddev weight;
run;
```

Figure 9.18 Exponentially Weighted Moving Average Chart with an Inset



Syntax: INSET Statement

The syntax for the INSET statement is as follows:

INSET *keyword-list* < / *options* > **;**

You can use any number of INSET statements in the MACONTROL procedure. However, when ODS Graphics is enabled, at most two insets are displayed inside the plot area and at most two are displayed in the chart margins. Each INSET statement produces a separate inset and must follow an EWMACHART or MACHART statement. The inset appears on every panel (page) produced by the last chart statement preceding it.

Keywords specify the statistics to be displayed in an inset; options control the inset's location and appearance. A complete description of the INSET statement syntax is given in the section "Syntax: INSET and INSET2 Statements" on page 1934 of Chapter 17, "The SHEWHART Procedure." The INSET statement options are identical in the MACONTROL and SHEWHART procedures, but the available keywords are different. The options are listed in Table 17.87. The keywords available with the MACONTROL procedure are listed in Table 9.14 to Table 9.17.

Table 9.14 Summary Statistics

Keyword	Description
MEAN	estimated or specified process mean
N	nominal subgroup size
NMIN	minimum subgroup size
NMAX	maximum subgroup size
NOUT	number of subgroups outside control limits
NLOW	number of subgroups below lower control limit
NHIGH	number of subgroups above upper control limit
STDDEV	estimated or specified process standard deviation
DATA=	arbitrary values from *SAS-data-set*

Table 9.15 Parameter for Uniformly Weighted Moving Average Charts

Keyword	Description
SPAN	number of terms used to calculate moving average

Table 9.16 Parameter for Exponentially Weighted Moving Average Charts

Keyword	Description
WEIGHT	weight assigned to most recent subgroup mean in computation of the EWMA

You can use the keywords in Table 9.17 only when producing ODS Graphics output. The labels for the statistics use Greek letters.

Table 9.17 Keywords Specific to ODS Graphics Output

Keyword	Description
UMU	estimated or specified process mean
USIGMA	estimated or specified process standard deviation

References

American Society for Quality Control (1983), *ASQC Glossary and Tables for Statistical Quality Control*, Milwaukee: ASQC.

American Society for Testing and Materials (1976), *ASTM Manual on Presentation of Data and Control Chart Analysis*, Philadelphia: ASTM.

Burr, I. W. (1969), "Control Charts for Measurements with Varying Sample Sizes," *Journal of Quality Technology*, 1, 163–167.

Burr, I. W. (1976), *Statistical Quality Control Methods*, New York: Marcel Dekker.

Crowder, S. V. (1987a), "Average Run Lengths of Exponentially Weighted Moving Average Charts," *Journal of Quality Technology*, 19, 161–164.

Crowder, S. V. (1987b), "A Simple Method for Studying Run-Length Distributions of Exponentially Weighted Moving Average Charts," *Technometrics*, 29, 401–408.

Hunter, J. S. (1986), "The Exponentially Weighted Moving Average," *Journal of Quality Technology*, 18, 203–210.

Kume, H. (1985), *Statistical Methods for Quality Improvement*, Tokyo: AOTS Chosakai.

Montgomery, D. C. (1996), *Introduction to Statistical Quality Control*, 3rd Edition, New York: John Wiley & Sons.

Nelson, L. S. (1983), "The Deceptiveness of Moving Averages," *Journal of Quality Technology*, 15, 99–100.

Nelson, L. S. (1989), "Standardization of Shewhart Control Charts," *Journal of Quality Technology*, 21, 287–289.

Nelson, L. S. (1994), "Shewhart Control Charts with Unequal Subgroup Sizes," *Journal of Quality Technology*, 26, 64–67.

Roberts, S. W. (1959), "Control Chart Tests Based on Geometric Moving Averages," *Technometrics*, 1, 239–250.

Robinson, P. B. and Ho, T. Y. (1978), "Average Run Lengths of Geometric Moving Average Charts by Numerical Methods," *Technometrics*, 20, 85–93.

Wadsworth, H. M., Stephens, K. S., and Godfrey, A. B. (1986), *Modern Methods for Quality Control and Improvement*, New York: John Wiley & Sons.

Wetherill, G. B. (1977), *Sampling Inspection and Quality Control*, 2nd Edition, New York: Chapman & Hall.

Wortham, A. W. and Heinrich, G. F. (1972), "Control Charts Using Exponential Smoothing Techniques," *ASQC Annual Conference Transactions*, 26, 451–458.

Wortham, A. W. and Ringer, L. J. (1971), "Control via Exponential Smoothing," *Logistics Review*, 7, 33–40.

Subject Index

annotating
  example, 879

EWMA charts
  asymptotic control limits, displaying, 807
  asymptotic control limits, example, 826
  average run lengths, computing, 835
  axis labels, 823
  central line, 810
  control limit equations, 810
  control limits, computing, 806, 810
  displaying subgroup means, example, 833
  examples, advanced, 824
  examples, introductory, 786
  missing values, 824
  notation, 809
  ODS tables, 818
  options summarized by function, 798
  overview, 785
  plotted points, 809
  plotting character, 798
  plotting subgroup means, 807
  probability limits, 806
  process mean, specifying, 807
  process standard deviation, specifying, 809
  reading preestablished control limit parameters, 795–797, 819, 820
  reading probability limits, 808
  reading raw measurements, 786, 787, 789, 818, 819
  reading subgroup summary statistics, 789–792, 820, 821
  reading summary statistics and control limits, 795, 821
  saving control limit parameters, 793, 815, 816
  saving subgroup summary statistics, 792, 793, 816
  saving summary statistics and control limits, 794, 795, 817
  specifying parameters for, 824, 826
  standard deviation, estimating, 822, 823
  syntax, 797
  varying subgroup sample sizes, 828
  weight parameter, choosing, 812
  weight parameter, specifying, 809
exponentially weighted moving average charts, *see* EWMA charts

geometric moving average charts, *see* EWMA charts

getting started, MACONTROL procedure
  adding insets to plots, 881

insets
  getting started, 881
  overview, 880

missing values
  MACONTROL procedure, 824
moving average control charts, *see* EWMA charts, *see* uniformly weighted moving average charts
  adding features to, 781
  average run lengths, displaying, 879
  graphics catalog, specifying, 782
  introduction, 778
  learning about, 780
  line printer features, 781, 782
  lineprinter charts, creating, 783
  reading control limit parameters, 783
  reading raw measurements, 781
  reading subgroup summary statistics, 782, 783
  syntax, 781

uniformly weighted moving average charts
  adding features to, 879
  annotating charts, 879
  asymptotic control limits, displaying, 860
  axis labels, 876
  central line, 862
  control limit equations, 863, 864
  control limits, computing, 859
  examples, advanced, 877
  examples, introductory, 838
  missing values, 876
  notation, 862
  ODS tables, 870
  options summarized by function, 851
  overview, 837
  plotted points, 862
  plotting character, 851
  plotting subgroup means, 860
  probability limits, 859
  process mean, specifying, 860
  process standard deviation, specifying, 861
  reading preestablished control limit parameters, 848, 849, 871, 872
  reading probability limits, 861
  reading raw measurements, 839–841, 871

reading subgroup summary statistics, 842–844, 872, 873
reading summary statistics and control limits, 848, 873, 874
saving control limit parameters, 845, 846, 867, 868
saving subgroup summary statistics, 844, 845, 868, 869
saving summary statistics and control limits, 846, 847, 869, 870
span of moving average, choosing, 864
span parameter, specifying, 862
specifying parameters for, 877, 879
standard deviation, estimating, 874, 875
syntax, 850

Syntax Index

ALPHA= option
  MACONTROL procedure, 806
ASYMPTOTIC option
  MACONTROL procedure, 807, 860

block-variables, MACONTROL procedure
  MACHART statement, 850
block-variables, MACONTROL procedure
  EWMACHART statement, 798
BY statement
  MACONTROL procedure, 783

CMEANSYMBOL= option
  MACONTROL procedure, 807, 860

EWMACHART statement, *see also* MACONTROL procedure, EWMACHART statement
  examples, advanced, 824
  examples, introductory, 786
  overview, 785
  syntax, 797

INSET and INSET2 statements, *see* MACONTROL procedure, INSET statement
INSET statement
  getting started, 881
  overview, 880
  syntax, 882

LIMITN= option
  MACONTROL procedure, 807, 860

MACHART statement, *see also* MACONTROL procedure, MACHART statement
  examples, advanced, 877
  examples, introductory, 838
  overview, 837
  syntax, 850
macontrol, 778
MACONTROL procedure, 781
  ANNOTATE2= option, 781
  ANNOTATE= option, 781
  DATA= data set, 781
  FORMCHAR= option, 781, 782
  GOUT= option, 782
  HISTORY= data set, 782, 783
  INSET statement, 879
  introduction, 778
  LIMITS= data set, 783
  LINEPRINTER option, 783
  overview, 780
  syntax, 781
  TABLE= data set, 783
MACONTROL procedure, BY statement, 783
MACONTROL procedure, EWMACHART statement
  ALLN option, 831
  ALPHA= option, 806
  ASYMPTOTIC option, 807, 826
  CMEANSYMBOL= option, 807
  DATA= data set, 818, 819
  HISTORY= data set, 789–792, 820, 821
  LIMITN= option, 807, 830
  LIMITS= data set, 795–797, 819, 820, 825
  MEANCHAR= option, 807
  MEANSYMBOL= option, 807, 835
  missing values, 824
  MU0= option, 807, 824, 826
  NMARKERS option, 831
  NOREADLIMITS option, 808
  OUTHISTORY= data set, 792, 793, 816
  OUTLIMITS= data set, 793, 815, 816
  OUTTABLE= data set, 794, 795, 817
  READALPHA option, 808
  READINDEX= option, 808
  READLIMITS option, 808
  RESET option, 809
  SIGMA0= option, 809, 824, 826
  SIGMAS= option, 809
  SMETHOD= option, 822, 832
  TABLE= data set, 795, 821
  VREF= option, 835
  WEIGHT= option, 787, 797, 809
  XSYMBOL= option, 824
MACONTROL procedure, MACHART statement
  ALPHA= option, 859
  ASYMPTOTIC option, 860
  CMEANSYMBOL= option, 860
  DATA= data set, 871
  HISTORY= data set, 842–844, 872, 873
  LIMITN= option, 860
  LIMITS= data set, 826, 848, 849, 871, 872, 878, 879
  MEANCHAR= option, 860
  MEANSYMBOL= option, 860
  missing values, 876
  MU0= option, 860, 877, 879
  NOREADLIMITS option, 860

OUTHISTORY= data set, 844, 845, 868, 869
OUTLIMITS= data set, 845, 846, 867, 868
OUTTABLE= data set, 846, 847, 869, 870
READALPHA option, 861
READINDEX= option, 861
READLIMITS option, 861
SIGMA0= option, 861, 877, 879
SIGMAS= option, 862
SMETHOD= option, 875
SPAN= option, 840, 850, 862
TABLE= data set, 848, 873, 874
XSYMBOL= option, 877
MEANCHAR= option
MACONTROL procedure, 807, 860
MEANSYMBOL= option
MACONTROL procedure, 807, 860
MU0= option
MACONTROL procedure, 807, 860

NOREADLIMITS option
MACONTROL procedure, 808, 860

processes, MACONTROL procedure
MACHART statement, 850
processes, MACONTROL procedure
EWMACHART statement, 797

READALPHA option
MACONTROL procedure, 808, 861
READINDEX= option
MACONTROL procedure, 808, 861
READLIMITS option
MACONTROL procedure, 808, 861
RESET option
MACONTROL procedure, 809

SIGMA0= option
MACONTROL procedure, 809, 861
SIGMAS= option
MACONTROL procedure, 809, 862
SPAN= option
MACONTROL procedure, 862
subgroup-variable, MACONTROL procedure
MACHART statement, 850
subgroup-variable, MACONTROL procedure
EWMACHART statement, 797
symbol-variable, MACONTROL procedure
MACHART statement, 851
symbol-variable, MACONTROL procedure
EWMACHART statement, 798

WEIGHT= option
MACONTROL procedure, 809

XSYMBOL= option
MACONTROL procedure, 824, 877